
Лабораторная работа №1 

Исследование функций пакета Scilab 

для обработки и визуализации данных 
 

1. Цель работы 

Изучение среды численного моделирования элементов информационных 
систем Scilab и ее базовых функций,  приобретение практических навыков мо-

делирования в среде Scilab. 

2. Основные теоретические положения 

Scilab – это кроссплатформенный свободно распространяемый математи-

ческий пакет, обладающий сходным с Matlab синтаксисом встроенного языка. 
Пакет программ Scilab предназначен для выполнения математических вычис-

лений и численного моделирования широкого спектра информационных и 
управляющих систем.   

Пакет можно скачать по адресу https://www.scilab.org/. После установки 
пакета Scilab для получения справочных сведений о его возможностях и коман-
дах необходимо выбрать пункт меню «Справка» или нажать кнопку F1. Откро-

ется окно «Справочная система» (рисунок 2.1), в котором можно выполнять 
поиск необходимых сведений как с использованием разделов содержания 

справки, так и по ключевым словам. 

 
Рисунок 2.1 – Справочная система Scilab 

 

http://www.scilab.org/
https://www.scilab.org/

Пакет Scilab позволяет выполнять команды в интерактивном режиме. По-
рядок ввода команд аналогичен пакету Матлаб. При этом используются сле-

дующие специфические обозначения: 
 
// - начало однострочного комментария 

% - начальный символ имени системных переменных, например: 
   %i _мнимая единица (√−1); 

%pi _число π = 3.141592653589793; 
%e _число e = 2.7182818; 

%inf _машинный символ бесконечности (∞); 
%NaN _неопределенный результат (0/0,∞/∞ и т. п.); 

%eps _условный ноль %eps=2.220Е-16; 
%t, %f – логические константы True и False. 

 
Системные переменные используются в качестве констант в выражениях 

и не могут быть изменены пользователем. Ниже представлен пример ввода вы-
ражения, содержащего системные переменные и встроенные функции, и вы-

числения его значения в командном окне Scilab: 
 

--> у= sin(%pi/4)+%e^2 

 у  =  
   8.0961629 

 
Scilab имеет расширенный перечень встроенных основных функций (см. 

п. меню «Справка»—> основные функции).  Клавиша Tab может использовать-

ся для автозавершения ввода имени функции или переменной (рисунок 2.2).  
 

 
Рисунок 2.2 – Пример автозавершения имени функции 

 

 Ввод в программу исходных данных с клавиатуры  выполняется вызо-
вом функции x = input(message [, "string"]), где message – строка, которая 

отображается на экране перед выполнением ввода, необязательный параметр 
"string" указывается при вводе строки, x – вещественное число или строка (если 

указан параметр "string"). 
Для вывода на экран результатов вычислений может использоваться 

функция disp. Если требуется вывести комбинацию строковых значений и чи-
сел, то требуется предварительно преобразовать числа в строки с помощью 

функции string: 
 
--> t=19 
--> disp("Средняя температура "+string(t)+" градусов") 

  Средняя температура 19 градусов 

Для форматированного вывода на экран значений переменных может ис-

пользоваться функция, аналогичная функции printf языка Си.  Например: 
 
--> printf("Результат: exp(3) = %f",exp(3)) 
Результат: exp(3) = 20.085537 

 

Ввод-вывод данных из файлов и в файлы выполняется  в Scilab c исполь-
зованием функций mopen, mfscanf, mfprintf, которые полностью аналогичны 

функциям языка Си: fopen, fscanf, fprintf.  
При программировании в Scilab необходимо различать понятия сценарий 

и функция, определяемая пользователем. 
Сценарием называют любую программу Scilab. Любой сценарий состоит 

из последовательности инструкций (команд), которые описывают конкретные 
действия с объектами Scilab. Сценарии создаются в окне редактора SciNotes, 

который может быть вызван из пункта меню «Инструменты». Сценарии сохра-
няются  в *.sce файлах. 

Функция, определяемая пользователем, также является сценарием, но 
характеризуется наличием имени, благодаря которому к ней можно обратиться 
из любого места сценария. Имеется два способа определения  функций: func-
tion и deff.  Определение функции c использованием конструкции function  

аналогично большинству языков программирования и не требует пояснений. 
Например: 

 
function z=f(x,y) 
 z=1/(exp(-y)+exp(-x)) 

endfunction 

 

Обычно такое определение функции сохраняется в файле *.sce на диске 
(например, f.sce). Чтобы использовать функцию, необходимо загрузить её из 

файла. Для этого необходимо выполнить команду exec. Её единственным ар-
гументом является строка с указанием положения сохраненного файла в файло-

вой системе. Например, если файл f.sce был сохранен в домашнем каталоге 
пользователя, то команда загрузки будет выглядеть следующим образом: 

exec('~/f.sce'). После этого функция, описанная в файле, может быть вызвана в 
текущей сессии, например f(1,2).  

Определение функции с использованием оператора deff реализуется в со-
ответствии с шаблоном: 

 
deff('[имя1,...,имяN]=имя_функции(переменная_1,...,переменная_M)',  

     'имя1=выражение1;...;имяN=выражениеN'), 
 

где список [имя1,...,имяN] представляет имена выходных значений функции, 

выражение1,…, выражениеN – правила вычисления этих значений. Листинг 

ниже демонстрирует определение функции z=f(x,y) с помощью оператора deff  
и использование f для вычисления значения функции: 

--> deff('z=f(x,y)','z=1/(exp(-y)+exp(-x))'); 
--> a=1; b=2; c=f(a,b) 

 c  =  
   1.9872232 

 
Scilab хранит установки, функции и переменные в рабочей области, назы-

ваемой Workspace. Переменные рабочей области отображаются в окне обозре-
вателя переменных. Для очистки рабочей области от всех пользовательских пе-

ременных используется команда  clear all. Если требуется удалить одну пере-
менную, то применяется команда clear <имя переменной>.  

Система Scilab предназначена для выполнения математических вычисле-
ний. Обозначение математических операций и управляющих конструкций (if, 
for, whilе и др.) такое же, как и в системе Matlab. Рассмотрим примеры некото-
рых операций с одномерными (векторы) и двумерными массивами (матрицы). 

Вектор-строка задается своими элементами, которые записываются  
внутри квадратных скобок и разделяются пробелами или запятыми. Для зада-

ния вектора-столбца его элементы следует разделять точкой с запятой (;). На-
пример: 

--> xvec=[1 2 3] // вектор-строка  

 xvec  =  

   1.   2.   3. 

--> yvec=[1;2;3] // вектор-столбец 
 yvec  =  
   1. 

   2. 
   3. 

 

Векторы удобно создавать с помощью оператора “ : ”. Например, если 

требуется создать вектор из нечетных чисел от 1до 10, то можно применить 
оператор: 

--> x=1:2:10  // Хнач:Шаг:Хкон 

 x  =  

1. 3.   5.   7.   9. 

 

Переменную заданную как массив можно использовать в арифметических 

выражениях в качестве аргумента стандартных математических функций. Ре-
зультатом выполнения таких операторов являются массивы. 

Для определения длины вектора используется функция length. Для полу-
чения i-го компонента вектора  применяется команда x(i), а чтобы получить 

компоненты вектора со 2-го по 4-й, используется инструкция 
 
--> x(2:4) 
 ans  = 

   3.   5.   7. 

 
Матрица представляется в виде набора векторов-строк, разделяемых точкой с 

запятой: 
 

--> m=[1 2 3;4 5 6] 
 m  =  

   1.   2.   3. 
   4.   5.   6. 
 

Размер матрицы определяется с помощью функции size 
 

--> size(m) 
 ans  = 

   2.   3. 
 

Для выделения строки или столбца матрицы используют оператор “:”, напри-

мер: 
 

--> m(2,:) 
 ans  = 
   4.   5.   6. 

--> m(:,3) 
ans  = 

   3. 
   6. 

 

Матрицы и векторы можно формировать, составляя их из ранее заданных мат-
риц и векторов: 

 
--> v1=[1 2 3]; v2=[4 5 6]; v3=[7 8 9]; 

--> //Горизонтальная конкатенация векторов–строк: 
--> V=[v1 v2 v3] 
  V = 1 2 3 4 5 6 7 8 9 

-->//Вертикальная конкатенация векторов–строк, 
-->//результат матрица: 

--> V=[v1; v2; v3] 
  V = 
  1 2 3 

  4 5 6 
  7 8 9 

-->//Горизонтальная конкатенация матриц: 
--> M=[V V V] 
  M = 

  1 2 3 1 2 3 1 2 3 
  4 5 6 4 5 6 4 5 6 

  7 8 9 7 8 9 7 8 9 
-->//Вертикальная конкатенация матриц: 
--> M=[V;V] 

  M = 

  1 2 3 
  4 5 6 
  7 8 9 

  1 2 3 
  4 5 6 

  7 8 9 
 
Для работы с матрицами в Scilab применяются матричные операции: “+” ― 

сложение; “–“ ― вычитание; “ ‘ “― транспонирование; “* “― матричное ум-
ножение; “^”― возведение в степень; “\”― левое деление; “/ “― правое деле-

ние; “.*” ― поэлементное умножение матриц; “.^”―_ поэлементное возведение 
в степень; “.\”― поэлементное левое деление; “./”― поэлементное правое деле-

ние. Эти операции  аналогичны операциям Matlab. 
Примеры действий над матрицами: 

 
-->A=[1 2 0;-1 3 1;4 -2 5]; 

-->B=[-1 0 1;2 1 1;3 -1 -1]; 
-->//Вычислить (AT+B)2 - 2A(0.5BT-A) 
-->(A'+B)^2-2*A*(1/2*B'-A) 

ans = 
  10. 8. 24. 

  11. 20. 35. 
  63. - 30. 68. 
--> //Решить матричные уравнения А∙Х=В и Х∙A=B. 

-->A=[3 2;4 3]; 
-->B=[-1 7;3 5]; 

-->//Решение матричного уравнения AX=B: решение X=A-1 B 
-->X=A\B 
X = 

  - 9. 11. 
  13. - 13. 

-->//Решение матричного уравнения XA=B: решение X=B A-1 
-->X=B/A 
X = 

  - 31. 23. 
  - 11. 9. 
-->X*A-B // проверка 

ans = 

  0. 0. 

  0. 0. 
 

Для работы с матрицами и векторами в Scilab существуют специальные 
функции: ones(m,n) – создает матрицу из единиц; zeros(m,n) – создает матрицу 
из нулей; eуе(m,n) - создает единичную матрицу. Функции для вычисления раз-

личных числовых характеристик матриц:  
sum – сумма элементов массива; 

prod – произведение элементов массива; 
max, min – максимальное и минимальное значение массива; 

mean, median – среднее и медианное значение массива; 

det – определитель квадратной матрицы; 
rank – ранг матрицы; 

norm – норма квадратной матрицы; 
cond – число обусловленности матрицы (произведение нормы исходной 

матрицы на норму обратной матрицы); 

spec – собственные значения и собственные векторы квадратной матри-
цы; 

inv – вычисляет обратную матрицу; 
pinv – вычисляет  псевдообратную матрицу; 

linsolv – решает  систему линейных алгебраических уравнений; 
svd – выполняет  сингулярное разложение матрицы и др. 

 
Построение двумерного графика  в отдельном графическом окне в коор-

динатах x,y осуществляется по команде plot(x,y).  Команда позволяет также за-
давать цвет  и стиль изображения точек на графике, например команда 

plot(1,2,"*r") отобразит звездочку красным цветом в позиции с координатами 
(1,2). Для отображения графика функции вида y=f(x) удобно значения x зада-
вать в виде вектора, который формируется с помощью вызова функции 

x=linspace(a,b,n) , где параметры a и b задают интервал определения функция, а n 
– число точек разбиения этого интервала. На рисунке 2.3 приведен пример про-

граммы, определяющей функции f(x) и g(x) и отображающей  их графики в од-
ной системе координат с помощью plot, соответственно красным и зеленым 

цветом: 
 

function y=f(x) 
y=(x 2̂+2*x)*exp(-x/2) 

endfunction 

function y=g(x) 
y=cos(2*x)*exp(-x/2) 

endfunction  

x=linspace(-2,10,100); 
plot(x,f,"r",x,g,"g") 
xgrid //отображение сетки 

// отображение надписей  
xtitle('Функции f(x) и g(x)',  

'Позиция','Ускорение') 

//легенда: 4 – правый нижний угол, %t - рамка 
legend(‘f(x)’,’g(x)’,4,%t)  

 

 
 

Рисунок 2.3 – Построение двумерных графиков с помощью plot 
 

Функция subplot(m,n,p) разбивает графическое окно на m окон по вертика-
ли и n окон по горизонтали, текущим окном становится окно с номером p. При-

меняется для отображения нескольких графиков – каждый в своем подокне.  
Команда figure(i) делает активным i-ое окно. Для очистки текущего гра-

фического окна применяется функция clf. Функция xdel(i) закрывает графиче-
ские окна, где i – вектор номеров окон.  

Scilab позволяет отображать поверхности и кривые в 3-х мерном про-
странстве с помощью функции surf. Эта функция имеет три параметра: x, y и z. 

Векторы x и y, размерами m и n, определяют координатные точки по осям Ox и 
Oy. Матрица z, размером nxm,  хранит высоты в координатных узлах. Для ото-

бражения поверхности, определяемой функцией z=f(x,y), необходимо предва-
рительно определить z во всех координатных узлах с помощью вызова функции 
feval(x,y,f). Функция feval возвращает транспонированную матрицу, размером 

mxn. Поэтому при вызове функции surf матрицу z требуется транспонировать. 
Удобнее 3-d поверхности строить с использованием функций plot3d и plot3d1. 

Сравнительный пример построения 3-D поверхности с помощью указанных 
функций представлен на рисунке 2.4. 

 
function z=f(x, y) 

 z=3*x^2+y^2; 
endfunction 
x=linspace(-1,1,50); 

y=linspace(-2,2,100); 
z=feval(x,y,f); 

clf 
figure(1) 
subplot(1,3,1) 

surf(x,y,z') 
subplot(1,3,2) 

plot3d(x,y,z) 
subplot(1,3,3) 
plot3d1(x,y,z) 

 

 
 

Рисунок 2.4 – 3-D поверхности, построенные функциями surf, plot3d,plot3d1 
 

В Scilab имеются и другие функции для построения 2-х и 3-х мерных 
графиков, например: plot2d, mesh, contour, plot3d2, plot3d3, раram3d и др. 

 Scilab позволяет выполнять статистическую обработку данных в масси-
вах. Функция mean(M) вычисляет среднее всех элементов матрицы М, 

mean(M,’r’) – среднее по столбцам, а mean(M,’c’)- среднее по строкам. Функция 

stdev(M) вычисляет стандартное отклонение, а функция variance(M) – диспер-
сию всех элементов массива M. Специальная функция tabul(M) вычисляет час-

тоты появления значений в массиве M. Для генерации псевдослучайных чисел 
могут использоваться функции grand и rand. Функция rand(n1,n2,…nn[, p]) - 
формирует многомерную матрицу случайных чисел, необязательный параметр p 

- это символьная переменная, с помощью которой можно задать тип распреде-
ления случайной величины ('uniform' - равномерное , 'normal'- гауссовское). На 

рисунке 2.4 изображен  листинг программы для генерации матрицы размером 
10х10 из нормально распределенных случайных чисел с нулевым средним и 

единичной дисперсией, а также построена гистограмма распределения. 
 

--> x=rand(10,10,'normal'); 
--> mean(x) 

 ans  = 
  -0.0218062 
--> variance(x) 

 ans  = 
   1.1118125 

--> histplot(12,x); 

. 
 

 
 

 
Рисунок  2.4 – Пример использования функций rand и histplot 

 
В функции grand доступны более лучшие генераторы случайных чисел в 

том смысле, что они имеют как более длинный период, так и более лучшие ста-
тистические свойства. Кроме этого, функция grand обладает большими функ-

циональными возможностями, в частности, позволяет генерировать случайные 
числа по различным законам распределения. Ниже приведены некоторые при-
меры вызова функции grand: 

x=grand(400,500,’nor’,0,1); //нормальный распределение 
x=grand(1,1000,’unf’,-1,1); // равномерное распред., числа в диапазоне [-1,1] 

x=grand(1,1000,’exp’,3); // экспоненциальное распределение, лямбда=3 
x=grand(1,5000,’chi’,10); // распределение хи квадрат с 10-ю степенями свободы 

x=grand(1,5000,’poi’,2); // распределение Пуассона с мат. ожиданием 2 

about:blankgrand.html


13 
 

3 Описание лабораторной установки 

В качестве лабораторной установки используется персональный компью-
тер с инсталлированной системой Scilab и модулем моделирования нейронных 

сетей Neuralnetworks 2.0. 
При запуске Scilab появляется главное окно, показанное на рисунке 3.1. 
  

 
 

Рисунок 3.1 – Главное окно системы Scilab 
 

Оно содержит окно обозревателя файлов, окно обозревателя переменных, 

журнал команд, окно новостей и командное окно. При выборе одного из окон 
его рамка окрашивается синим цветом.  

Управление Scilab осуществляется через систему меню и командное окно. 
Появление стрелки вида  --> (prompt) свидетельствует о готовности Scilab вос-

принимать команду. 
Для установки модуля моделирования нейронных сетей необходимо в 

командном окне ввести команду инсталляции модуля: 
 
atomsInstall("neuralnetwork") 

 
В этом случае из сети Интернет автоматически будет загружен и устав-

новлен Neural Network Module 2.0. При необходимости модуль может быть ус-
тановлен вручную. Необходимые компоненты располагаются по адресу 

https://atoms.scilab.org/toolboxes/neuralnetwork/2.0 

 

https://atoms.scilab.org/toolboxes/neuralnetwork/2.0

4 Варианты заданий и программа работы  

4.1  Ознакомиться с особенностями построения системы Scilab, основны-

ми командами и функциями, а также правилами построения сценариев [2]. 

4.2  Выбрать вариант задания, взяв остаток от деления номера студента в 

списке группы на 15. Выполнить задания 4.3-4.6 в соответствии с выбранным 

вариантом. 

4.3  Определить функцию f(x) и вычислить N её значений на заданном от-

резке. На экран вывести значения аргумента и значения функции. Построить 

график функции f(x), снабдив его всеми необходимыми надписями и коорди-

натной сеткой. 

4.4 Если возможно, то вычислить матрицу обратную D 

4.5. Вычислить 3 значения функции на заданном отрезке. Вывести значе-

ния  аргумента и значения функции на экран. Построить 3-D график функции.  

4.6 Сгенерировать  матрицу из (N+10)x(100*N)  одинаково распределен-

ных случайных чисел, где N – номер  студента в списке группы. Построить гис-

тограмму распределения этих чисел. Вычислить среднее, дисперсию и стан-

дартное отклонение, медиану. Варианты распределений случайных чисел: 

1. Равновероятное распределение на интервале (0,1). 

2. Нормальное распределение с мат. ожиданием 1 и дисперсией 2. 

3. Равновероятное распределение на интервале (-5,5). 

4. Нормальное распределение с мат.ожиданием 5 и дисперсией 0.25. 

5. Экспоненциальное распределение с параметром лямбда=5. 

6. Экспоненциальное распределение с параметром лямбда=10. 

7. Распределение Пирсона хи-квадрат с 5-ю степенями свободы. 

8. Распределение Пирсона хи-квадрат с 20-ю степенями свободы. 

9. Распределение Пирсона хи-квадрат с 1, 10, 30-ю степенями свободы. 

10.  Распределение Пирсона хи-квадрат с 1, 5, 20-ю степенями свободы. 

11.  Распределение Пуассона с мат. ожижанием 2. 

12.  Распределение Пуассона с мат. ожижанием 10. 

13.  Нормальное распределение с мат. ожиданием -5 и дисперсией 5. 

14.  Равновероятное распределение на интервале (-3,10). 

15.  Экспоненциальное распределение с параметром лямбда=1. 

5  Методические указания по выполнению работы 

5.1. При выполнении задания 4.3 рекомендуется воспользоваться функци-

ей  plot2d.  Синтаксис функции (параметры в квадратных скобках не обяза-
тельны): 
 plot2d([logflag,][x,],y[,style[,strf[,leg[,rect[,nax]]]]]) 
 plot2d([logflag,][x,],y,<opt_args>) 

 

Здесь  x –действительная матрица или вектор; если х отсутствует, то предпола-

гается, что вектор изменяется от 1:n, где n - число пунктов кривой, определяе-

мых параметром y; 

у - действительная матрица или вектор; 

<opt_args> - последовательность ключей key1=value1, key2=value2,..., где 

key1, key2,... могут иметь значения: 

 logflag - устанавливает масштаб вдоль осей (линейный или логарифмиче-

ский), возможные значения: "nn", "nl", "ln" и "ll"; 

style - устанавливает стиль для каждой кривой, целочисленное положи-

тельное или отрицательное значение; 

strf - управляет отображением заголовка, задается в виде последователь-

ности из трех цифр "xyz" (по умолчанию strf = "081"). 

Значения этих цифр и остальных ключей можно посмотреть в справке к 

функции. Ниже приведен пример сценария для построения графиков 3-х  сину-

соидальных функций с использованием plot2d: 

clf;  
x=[0:0.1:2*%pi]'; 

plot2d(x,[sin(x) sin(2*x) sin(3*x)]) 

 5.3 В ходе выполнения задания 4.4. используйте встроенные функции для 

работы с матрицами, описанные на стр. 9  
 5.4. При построении 3-D графиков (задание 4.5) рекомендуется изучить 

возможности пункта меню  «Инструменты» графического окна, в частности, 
вращения изображения и увеличения области, а также возможностей контекст-

about:blankclf.html
about:blanksin.html
about:blanksin.html
about:blanksin.html

ного меню, которое появляется при нажатии правой кнопки мышки. Исполь-
зуйте эти инструменты для более детального анализа свойств функции и редак-

тирования изображения и надписей. 
 5.5 При построении гистограмм в соответствии с заданием 4.6 восполь-
зуйтесь расширенными возможностями функции histplot, приведенными в при-

мере: 
 

d=rand(1,10000,'normal');   
clf; histplot(20,d) 

xgrid 
clf; histplot(20,d,normalization=%f) 
clf; histplot(20,d,leg='rand(1,10000,''normal'')',style=5) 

clf; histplot(20,d,leg='rand(1,10000,''normal'')',style=16, rect=[-3,0,3,0.5]); 

 

 

6 Содержание отчета 

 
6.1 Цель работы. 

6.2 Вариант задания, описание используемых формул.   
6.3 Листинги программ с комментариями. 
6.4 Результаты вычислений и графики функций. 

6.5  Выводы по результатам исследований. 
 

7 Контрольные вопросы 
 

7.1 Дайте общую характеристику системы моделирования Scilab и приве-

дите основные отличия от системы Matlab. 
7.2 Какие типы данных, используются в Scilab  и каким образом задаются 

константы в программах для Scilab? 
7.3 Как в Scilab определяются пользовательские функции? 

7.4 Приведите пример задания вектора-строки, вектора столбца и матри-
цы. 

7.5 Поясните, как осуществляются операции конкатенации  матриц и как 
они реализуются инструкциями Scilab? 

7.6 Как математически записываются комплексные числа и какие опера-
ции существуют в Scilab для работы с такими числами? 

7.7 Какие основные статистические функции имеются в Scilab для обра-
ботки матриц? 

7.8 Каким образом можно сохранить данные в файле или ввести их в 

Scilab из файла? 
7.9 Какие основные команды Scilab используются для построения дву-

мерных графиков? 

about:blankrand.html
about:blankclf.html
about:blankclf.html
about:blankclf.html
about:blankclf.html

7.10 Как сделать, чтобы вывод нового графика не стирал предыдущий 
график? 

7.11 Как отобразить на графике координатную сетку, подписать рисунок 
и названия осей координат? 

7.12 Как построить в Scilab 3-D поверхность? 

7.13 Напишите сценарий генерирования случайной последовательности 
с нормальным законом распределения и построения гистограммы 

распределения. 
7.14 Как можно разделить графическое окно на отдельные подокна? 

7.15 Как можно разделить переменные на равномерные интервалы? 
Приведите пример. 

7.16 Напишите сценарий генерирования смеси гармонического сигнала с 
шумом и графического отображения этого процесса. 

7.17 Напишите сценарий генерации случайных чисел с экспоненциаль-
ным распределением и построения гистограммы такого распределе-

ния. 

Лабораторная работа №2 

Исследование активационных функций нейронных  
элементов 

1  Цель работы 

Углубление теоретических знаний в области архитектуры нейронных се-
тей, исследование свойств активационных функций  нейронных элементов, 

приобретение практических навыков моделирования простейшей нейронной 
сети прямого распространения. 

2  Основные теоретические положения 

2.1 Структура нейрона с одним входом 

Элементарной ячейкой нейронной сети является простейший нейрон. 
Структурная схема простейшего нейрона с единственным скалярным входом 

показана на рисунке 2.1а. На рисунке приняты обозначения, используемые в 
книге [3]. 

 

 

 

 
а) б) 

Рисунок 2.1 – Структура нейрона с единственным входом 

 
Скалярный входной сигнал р умножается на скалярный весовой коэффи-

циент w, и результирующий взвешенный вход w*p является аргументом функ-
ции активации нейрона f, которая порождает скалярный выход а. 

Нейрон, показанный на рисунке 2.1,б, дополнен скалярным смещением b. 
Смещение суммируется со взвешенным входом w*p и приводит к сдвигу аргу-

мента функции f на величину b. Действие смещения можно свести к схеме 
взвешивания, если представить, что нейрон имеет второй входной сигнал со 

значением равным 1, а весовой коэффициент равен b. Вход n функции актива-
ции нейрона по-прежнему остается скалярным и равным сумме взвешенного 

входа и смещения b. Эта сумма является аргументом функции активации f; вы-
ходом функции активации является сигнал a. Константы w и b являются ска-
лярными параметрами нейрона. Основной принцип работы нейронной сети со-

стоит в настройке параметров нейрона таким образом, чтобы поведение сети 
соответствовало некоторому желаемому поведению. Регулируя веса и парамет-

ры смещения, можно обучить сеть реализовать определенную функцию; воз-
можно также, что сеть сама будет корректировать свои параметры, чтобы дос-

тичь требуемого результата. 
Уравнение простейшего нейронного элемента со смещением можно запи-

сывать в виде 

.                                    (2.1) 

Здесь константа 1 рассматривается как входное значение и может быть учтена в 
линейной комбинации расширенного вектора весов x=[w b] и расширенного 

вектора входа z=[p 1]
T
 : 

         
 
 
     x z).                                      (2.2) 

 
2.2 Активационные функции 

 

Рассмотрим основные функции активации, используемые в нейронных 
сетях и реализованные  в модуле Neuralnetworks 2.0 пакета Scilab. 

Единичная функция активации с жестким ограничением (hardlim) соот-

ветствует функции Хевисайда. Она равна 0, если n<0, и 1, если n  0. Пример 

вызова функции в пакете Scilab:   

 

n =  [-5:0.1:5]; 

a = ann_hardlim_activ(n);  // вызов функции hardlim 

plot(n,a,'.'); 
 

 В результате получим график функции hardlim в диапазоне значений входа от –

5 до + 5 (рисунок 2.2а). 
 

 

 
а) б) 

Рисунок 2.2 – Функции активации нейронных элементов 
 

Линейная функция активации (purelin). Линейная функция описывается 

соотношением a=n . График функции изображен на рисунке 2.2б. Пример вызо-
ва функции в пакете Scilab: a = ann_purelin_activ(n). 

Логистическая (сигмовидная) функция активации (logsig). Эта функция 
описывается соотношением a = 1/(1 + exp(–n)). График функции изображен  на 

рисунке 2.3. Аргумент функции может принимать любое значение в диапазоне 

от – до +, а выход изменяется в диапазоне от 0 до 1. Пример вызова в пакете 

Scilab:  a = ann_logsig_activ(n). Благодаря свойству дифференцируемости эта 

)1**( bpwfa 

функция часто используется в сетях с обучением на градиентных методов опти-
мизации.  

 
Рисунок 2.3 – График сигмовидной функции активации 

 
Тангенциальная (сигмовидная) функция активации (tansig). Эта функция 

описывается соотношением a = (exp(n)-(exp(-n))/(exp(n) + exp(–n)). График 
функции подобен графику функции logsig. Её аргумент также может принимать 

любое значение в диапазоне от – до +, но значение изменяется в диапазоне 

от -1 до 1. Пример вызова функции в пакете Scilab:  a = ann_tansig_activ(n). 

 В таблице 2.1 представлены дополнительные функции активации, часто 

используемые в моделях ИНС. При необходимости они легко могут быть за-
программированы самостоятельно. 

 
Таблица 2.1 – Дополнительные функции активации 

Символы, изображенные в квадрате в правом верхнем углу графиков на 

рисунках 2.2–2.3, а также в 3-ем столбце таблицы 2.1, представляют условное 
обозначение функции активации. Эти обозначения используются на структур-

ных схемах нейронных сетей. 

2.3 Нейрон с векторным входом и сеть прямого распространения 
 

Нейрон с одним вектором входа p с R элементами p1, p2, …, pR  изображен 
на рисунке 2.4. Здесь каждый элемент вектора p умножается на веса w11, w12, … 

, w1R . Взвешенные значения вектора входа поступают на сумматор. Их сумма 
равна скалярному произведению вектора-строки W на вектор входа p. 

 

 
Рисунок 2.4 – Схема простейшего нейрона с  векторонным входа 

 
Смещение b суммируется с взвешенной суммой входов. Результирующая 

сумма n, называемая сетевым значением, равна 

n = w11 p1 + w12 p2  + … + w1R pR + b                                (2.3) 

Сетевое значение n служит аргументом функции активации f. Выражение (2.3) 

удобно записывать в векторно-матричной форме: 

n = Wp + b.                                                     (2.4) 

При рассмотрении нейросетей, содержащих нейроны с векторными вхо-

дами, часто используется обобщенная векторно-матричная схема нейрона, изо-
браженная на рисунке 2.5 и структурно соответствующая выражению (2.4). 

Вход нейрона изображается в виде темной вертикальной черты, под кото-
рой указывается количество элементов входа R. Размер вектора входа p указы-

вается ниже символа p и равен R1. Вектор входа умножается на вектор-строку 

W длины R. Как и прежде, константа 1 рассматривается как вход, который ум-
ножается на скалярное смещение b.  

 
Рисунок 2.5 – Обобщенная векторно-матричная схема нейронного элемента 

 
Входом n функции активации нейрона служит сумма смещения b и про-

изведения W*p. Эта сумма преобразуется функцией активации  f, на выходе ко-

торой формируется выход нейрона a, который в данном случае является ска-
лярной величиной. Структурная схема, изображенная на рисунке 2.5, соответ-

ствует слою сети. Если слой содержит S нейронов, то он характеризуется мат-
рицей весов W, размером SxR, и вектором смещений b, размером Sx1.  

Каждый раз, когда используется обобщенное обозначение сети, размер-

ность матриц указывается под именами векторно-матричных переменных. Эта 
система обозначений поясняет строение сети и связанную с ней матричную ма-

тематику. 
С помощью схемы однослойной сети, изображенной на рисунке 2.5., 

можно построить многослойную сеть. В качестве примера на рисунке 2.6 изо-
бражена трехслойная сеть прямого распространения (feed forward – FF). Для 

этой сети выход предыдущего слоя является входом следующего слоя. Входом 
сети является вход первого слоя, т.е. вектор  p, а выходом – выход последнего 

слоя, т.е. вектор a
3
. Соответственно прямое распространение входного вектора 

по сети опишется уравнением: 
 

 .                     (2.5) 
 

Рисунок 2.6 – Трехслойная сеть прямого распространения 
 

Внутренние слои сети называются скрытыми слоями (hidden layers). Вход 

сети, представленный вектором p, называют входным слоем. Сокращенно 
структуру многослойной сети обозначают, указывая последовательно размер-

ность входного слоя и количество нейронов в последующих слоях. Например, 
структура сети, изображенная на рисунке 2.6, обозначается в виде: R-S

1
-S

2
-S

3
.
 

 
2.4. Производные активационных функций  

и инициализация весов нейросети 
 

В ходе обучения нейронной сети выполняют минимизацию некоторой 
функции потерь (целевой функции) сети, которая определяется целью функ-
ционирования сети. Поиск минимума функции потерь сопряжен с её диффе-

ренцированием, что приводит к необходимости вычисления производных и 
распространению их значений  в обратном направлении для определения чув-

ствительности функции потерь к настраиваемым параметрам сети. Важно, что-
бы при обратном распространении значения производных не становились нуле-

выми из-за зон насыщения функций активации, т.к. это приводит к потере эф-
фективности рассматриваемых далее градиентных алгоритмов обучения.  

Производные рассмотренных функций активации, использующие линей-
ные зависимости, равны 0 в зоне насыщения и равны 1 на линейном участке 
(satlin, satlins, poslin, purelin). Производные сигмовидных функций вычисляются 

на основе следующих выражений [3]: 
- производная логистической сигмовидной функции      (logsig):  

     

  
               ;                                          (2.6) 

- производная тангенциальной сигмовидной функции (гиперболический тан-
генс)       (tansig):  

      

  
                                                         (2.7) 

Под инициализацией нейронной сети подразумевается задание начальных 
значений матрицы весов W. Обычно начальным весам сети присваивают не-

большие случайные значения, а смещениям – нули. Для этого могут непосред-
ственно использоваться генераторы случайных чисел (rand, grand), рассмотрен-

ные в лабораторной работе 1. В модуле NeuralNetworks 2.0 пакета Scilab имеет-
ся встроенная функция ann_ffbp_init для инициализации многослойной сети с 

прямыми связями. Вызов этой функции осуществляется следующим образом: 
 

W = ann_ffbp_init(N,r), 
 

где  N – вектор, задающий количество нейронов в каждом слое сети, включая 
входной и выходной слои; r– вектор, задающий желаемый диапазон изменения 

значений весов; W – список, элементами которого являются расширенные мат-
рицы весов W(i) слоя i. Функция инициализирует веса сети путём вызова гене-

ратора случайных чисел grand c равномерным законом распределения. При 
этом векторы смещений b(i) каждого слоя получают нулевые значения и вклю-

чаются в матрицу W(i) в виде её последнего столбца. Пример вызова функции  
инициализации для сети со структурой  3-3-2-1: 

 
N = [3 3 2 1]; 

r = [-1 1]; 
W = ann_ffbp_init(N, r) 

W  =  
       W(1) 

   0.9297771   0.4516779   0.9143339   0. 

   0.9353899   0.9411856  -0.7802765   0. 
  -0.6847738   0.9622194  -0.0292487   0. 

      W(2) 
   0.5962117  -0.4059411  -0.990433    0. 
   0.6005609  -0.7162273  -0.1564774   0. 

       W(3) 
  -0.775071   0.831471   0. 

 
3 Варианты заданий и программа работы 

 

3.1. Повторить теоретический материал по источникам [1, 2, 5]. 

3.2. Выбрать вариант в соответствии с таблицей 2.2. Правила определения 
номера варианта изложены в лабораторной работе 1. 

 
Таблица 2.2 – Варианты  заданий  

Вариант Структура 

сети 

Активаци-

онная 
функция 1 

Активаци-

онная 
функция 2 

Входной сигнал для i-го 

входа 

1 100-2-1 logsig purelin sin(2*%pi*0.01*i*t) 

2 100-4-1 logsig satlin sin(2*%pi*0.02*i*t) 

3 100-6-1 logsig satlins sin(2*%pi*0.03*i*t) 

4 100-8-1 logsig poslin sin(2*%pi*0.04*i*t) 

5 100-10-1 tansig purelin sin(2*%pi*0.01*i*t) 

6 100-12-1 tansig satlin sin(2*%pi*0.02*i*t) 

7 100-14-1 tansig satlins sin(2*%pi*0.03*i*t) 

8 100-16-1 tansig poslin sin(2*%pi*0.04*i*t) 

9 50-4-1 logsig purelin sin(2*%pi*0.01*i*t) 

10 50-8-1 logsig satlin sin(2*%pi*0.02*i*t) 

11 50-16-1 logsig satlins sin(2*%pi*0.03*i*t) 

12 50-32-1 logsig poslin sin(2*%pi*0.04*i*t) 

13 50-16-1 tansig purelin sin(2*%pi*0.01*i*t) 

14 50-8-1 tansig satlin sin(2*%pi*0.02*i*t) 

15 50-4-1 tansig satlins sin(2*%pi*0.03*i*t) 

 

3.3. Построить графики активационных функций и их производных в со-
ответствии с вариантом задания, используя соответствующие встроенные 

функции модуля NeuralNetworks 2.0 пакета Scilab. В случае отсутствия указан-
ных встроенных функции определить их самостоятельно на основе выражений, 
приведенных в таблице 2.1. 

3.4. Реализовать две сети прямого распространения в соответствии с за-
данной структурой, запрограммировав вычисления в соответствии с выражени-

ем (2.5) для двух видов активационных функций, заданных по варианту. 
3.5. Выполнить моделирование двух нейросетей, сгенерировав входные 

сигналы в соответствии с вариантом задания на интервале времени от 0 до 2/F, 
где F – минимальная частота входного гармонического сигнала (например, для 

варианта 1 F=0.01Гц). При этом инициализацию сетей выполнить двумя спосо-
бами: 

1) случайными значениями из диапазона от -10 до +10; 
2) случайными значениями из диапазона от -1/sqrt(R) до +1/sqrt(R), где R 

число входов нейрона. 
3.6. Вычислить выходные значения всех нейронов сети. Построить гра-

фики активностей одного из нейронов скрытого слоя и нейрона выходного 
слоя, а также гистограмму общей активности всех нейронов скрытого слоя и 
отдельно гистограмму активности выходного нейрона для двух способов ини-

циализации для каждой из двух реализованных сетей. Вычислить значение 

средней активности и стандартного отклонения на выходах нейронов скрытого 
и выходного слоёв. 

3.7. Вычислить значения производных нелинейностей для всех нейронов. 
Построить графики значений производных нелинейности для одного из нейро-
нов скрытого слоя и нейрона выходного слоя, а также гистограмму производ-

ных функции активации скрытых нейронов и выходного (для двух способов 
инициализации и для каждой из двух реализованных сетей). Вычислить среднее 

значение и стандартное отклонение производных для нейронов скрытого и вы-
ходного слоёв. 

3.8. Выполнить анализ полученных результатов, обратив внимание на ха-
рактер гистограмм. Сделать выводы о характере распределения значений ак-

тивностей и производных для заданных активационных функций и каждого из 
способов инициализации. Сравнить результаты для каждой из двух реализо-

ванных сетей. 
3.9. Подготовить и защитить отчет по работе. 

 
4 Методические рекомендации по выполнению работы 

 

4.1. Модуль NeuralNetwork 2.0 пакета Scilab имеет ограниченный набор 
встроенных функций активации и их производных: 

 

a = ann_hardlim_activ(n) – единичная с жестким ограничением; 

a = ann_logsig_activ(n) – логистическая, однополярная сигмоидальная; 

a = ann_purelin_activ(n) – линейная; 

a = ann_tansig_activ(n) – биполярная тангенциальная сигмоидальная; 

d_a= ann_d_hardlim_activ(y) – производная единичной функции; 

d_a = ann_d_logsig_activ(y) – производная логистической функции; 

d_a = ann_d_purelin_activ(y) – производная линейной функции; 

d_a = ann_d_tansig_activ(y) – производная биполярной сигмоидальной функции. 
 

Остальные функции активации и их производные следует определять са-

мостоятельно на основе выражений, указанных в таблице 2.1.  
Обратите внимание на то, что функции ann_d_logsig_activ(y) и 

ann_d_tansig_activ(y) вычисляют производные на основе выражений (2.6) и 
(2.7). При этом входные аргументы этих функций должны представлять собой 

предварительно вычисленные значения однополярной сигмовидной функции 
ann_logsig_activ(n) или тангенциальной биполярной сигмовидной функции 

ann_tansig_activ(n).  
Указанные выше встроенные функции допускают обработку входных ар-

гументов, заданных векторами либо матрицами. В этом случае они применяют-
ся к векторам и матрицам поэлементно. Однако в силу специфики определения 
встроенной функции a = ann_logsig_activ(n) она не позволяет обрабатывать ар-

гумент n, заданный прямоугольной матрицей. Поэтому рекомендуется переоп-
ределить эту функцию самостоятельно, воспользовавшись следующим выраже-

нием: 
 

a=1 ./ (1+exp(-n)). 

4.2. При выполнении задания 3.4 для реализации выражения (2.5) удобно 
входные векторы представлять матрицей P, в которой каждый столбец соответ-

ствует очередному входному вектору. В этом случае произведение WP будет 
матрицей, что потребует преобразования вектора смещений b в матрицу для 
обеспечения выполнения операции сложения при вычислении n. Преобразова-

ние b матрицу можно выполнить с помощью встроенной функции repmat пакета 
Scilab. Пример вызова этой функции см. ниже в п.4.3. 

4.3. Для генерации матрицы P рекомендуется воспользоваться нижесле-
дующим примером, в котором также показано как выполнить моделирование 

сети (задание 3.5) на примере одного слоя, построить графики активностей и 
производных, вычислить необходимые статистики (задание 3.6 и 3.7): 

 

//задание исходных данных 
F=0.01 //минимальная частота входного сигнала 

R=5; //число входов слоя 
S=3; // число нейронов слоя 

 
//формирование матрицы p входных сигналов 
t=0:0.1:2/F; 

fi=(2*%pi*F).*t; 
p=[sin(fi)]; 

for i=2:R 
    p=[p; sin(fi*i)]; 
end 

 
//инициализация весов и смещений  

scale=1/sqrt(R);  
w=grand(S,R,'unf',-scale,scale) 
b=zeros(S,1); 

 
//моделирование слоя c tansig нелинейностью 

a=ann_tansig_activ(w*p+repmat(b,1,size(p,2))); 
 
//вычисление производных tansig нелинейности  

d_a=ann_d_tansig_activ(a); 
 

//вычисление статистик 
mean_a=mean(a) 
stdev_a=stdev(a) 

mean_d_a=mean(d_a) 
stdev_d_a=stdev(d_a) 

 
//построение графиков активностей, производных и гистограмм 
clf(1); 

figure(1); 
subplot(2,2,1) 

plot(t,a(1,:),t,a(2,:), t,a(3,:)); 
title('Активность нейронов слоя') 
subplot(2,2,2) 

plot(t,d_a(1,:),t,d_a(2,:), t,d_a(3,:)); 
title('Производные функции активации  слоя') 

subplot(2,2,3) 
histplot(20,a); 
title('Гистограмма активности нейронов слоя') 

subplot(2,2,4) 
histplot(20,d_a); 

title('Гистограмма производных функции активации') 
 

Результаты вычислений представлены на рисунке 2.7. 

 
Рисунок 2.7 – Графики активности нейронов и производных функции 

активации, гистограммы 
 

5 Содержание отчета 
 

5.1.  Цель работы. 
5.2.  Вариант задания.  

5.3.  Схема нейронной сети, формулы нелинейностей и их производных в соот-
ветствии с вариантом задания.  

5.4.  Листинги программ с комментариями.  

5.5.  Графики активности на выходе нейрона скрытого и выходного слоёв. Гис-
тограммы, значения основных статистик активности. 

5.6.  Графики значений производных нелинейности на выходе нейронов скры-
того и выходного слоёв. Гистограммы, значения основных статистик про-

изводных 
5.7.  Выводы по результатам исследований. 

6 Контрольные вопросы 

6.1.  Нарисуйте схему нейронного элемента с векторным входом и запишите 
выражение для вычисления выходного значения. 

6.2.  Приведите выражение функции активации с жестким ограничением 

(функция Хевисайда) и ее графическое представление. 
6.3.  Приведите выражения для линейных функций активации (satlin, satlins) с 

насыщением и их графическое представление. 
6.4.  Приведите выражение для знаковой функции активации и ее графическое 

представление. 

6.5.  Приведите выражения для униполярной и биполярной сигмовидных 
функций активации и их графическое представление. 

6.6.  Приведите выражение для выходных сигналов НЭ с квадратической ра-
диальной функцией. 

6.7.  Нарисуйте схему и приведите аналитическое выражение функции преоб-
разования для линейного порогового элемента. 

6.8.  Продемонстрируйте на компьютере, каким образом проводились иссле-
дования функций активации? 

6.9.  Нарисуйте структурные схемы однослойной и многослойной искусствен-
ных нейронных сетей с прямыми связями.  

6.10.  Поясните понятия входного, скрытого и выходного слоев. Как сокращен-
но обозначают структуру сетей прямого распространения? 

6.11.  Приведите выражения для вычисления производных активационных 

функций, указанных в таблице 2.1. 
6.12.  Используя правила дифференцирования, выведите выражения для вычис-

ления производных логистической и тагенциальной сигмовидных актива-
ционных функций. 

6.13.  Что такое иницализация нейронной сети и как её выполняют? Приведите 
примеры кода на языке Scilab. 

6.14.  Нарисуйте структурную схему 3-х слойной сети прямого распростране-
ния и запишите на языке Scilab фрагмент кода для вычисления её выход-

ного значения, если последовательность входных векторов p представля-
ется в виде матрицы.  

6.15.  Какой желательный характер распределений должны иметь выходные 
значения нейронов нейросети и почему? Как этого добиваются? 

6.16.  Почему нежелательно, чтобы производные активационных функций при-
нимали нулевые значения? 

Лабораторная работа №3 

Исследование однослойного персептрона 

1 Цель работы 

Углубление теоретических знаний в области архитектуры нейронных се-

тей с пороговыми активационными функциями, исследование свойств одно-
слойного персептрона и правила его обучения, приобретение практических на-

выков обучения и моделирования однослойной сети при решении простых за-
дач классификации. 

2 Основные теоретические положения 

2.1 Структура однослойного персептрона 

Общая структурная схема персептрона изображена на рисунке 3.1. Пер-

септрон представляет собой однослойную сеть, которая состоит из S нейронов с 
R входами, характеризуется матрицей весов WSxR  , и вектором смещений bSx1, 

использует пороговую активационную функцию с жестким ограничением 
hardlim.  

 
Рисунок 3.1 ─ Структурная схема однослойного персептрона 

 

Каждая срока матрицы W соответствует набору весов одного из нейронов 
персептрона. Если обозначить i-ую строку матрицы как i w

T
 , то выход отдель-

ного нейрона сети запишется в виде: 
 

                             (3.1) 
Если скалярное произведение i-ой строки матрицы весов  iw

T
 на входной 

вектор p , будет больше или равно  –bi , то выход  нейрона будет равен 1, иначе  

─ 0 . Таким образом, каждый нейрон персептрона делит пространство входных 

данных на две области. Персептрон, состоящий из S нейронов, способен (при 

определенных условиях) распознать  классов. 

 
2.2 Граница решения 

 

Рассмотрим персептрон из одного нейрона (простой персептрон) с 2-мя 
входами (рисунок 3.2.) 

 

 
Рисунок 3.2 ─ Простой персептрон и его граница решения 

 
С учетом только 2-х входов выходной сигнал нейрона запишется в виде:  
 

     (3.2) 
 

Так как значение функции hardlim(n) переходит из 0 в 1 при n=0, то грани-

ца решения определится из условия: 
 

                (3.3) 
Пусть 

 
Тогда граница решения между областями определится из уравнения: 
 

 
 

Это уравнение задаёт границу решения в виде линии (рисунок 3.2.) Чтобы изо-
бразить её, были найдены точки пересечения линии с осями (р1, p2): 

 
 
 
                                                                                ,                                                                                                                                . 

 
 
 

С одной стороны границы выход персептрона равен a=1, а с другой ─ a=0. Что-

бы определить область, где выход персептрона будет равен 1, рассмотрим точ-
ку  

,  
для которой a=1, действительно  

. 
Таким образом, часть входного пространства, где a=1, будет находиться выше 

границы решения (на рисунке 3.2. ─ затененная область).  
 

2.3 Графическое построение границы решения 
 

Границу решения можно также построить графически. Отметим, что 
граница всегда ортогональна вектору весов w, как изображено на рисунке 3.2. 

Действительно, граница решения определяется условием: 
 

                                          (3.4) 
 

Для всех точек, принадлежащих границе, скалярное произведение вход-

ного вектора p на вектор весов w имеет одно и то же значение, равное -b. Это 
означает, что соответствующие входные векторы p имеют одно и то же значе-

ние проекции на вектор весов w, т.е. их концы должны лежать на линии орто-
гональной вектору весов. Отметим, что вектор весов всегда направлен в сторо-

ну области, где выход нейрона равен 1. 
После того как определено направление вектора весов w, значение сме-

щения определяется путем выбора точки на границе и решения уравнения 
(3.4).  

Пример. 
Построим границу для случая реализации функции AND на основе простого 

персептрона.  
1. Зададим пары входных и выходных значений логического элемента AND 
в соответствии с рисунком 3.3. 

 
Рисунок 3.3 ─ Входные и выходные значения логического элемента AND 

(зачерненная точка соответствует 1) 

 
2. Выберем возможную границу решения (рисунок 3.4). 

Рисунок 3.4 ─ Граница решения и вектор весов элемента AND 
 

3. Построим вектор весов w, ортогональный границе. Вектор может иметь лю-
бую длину, например 

. 
4. Найдем смещение b. Для этого выберем произвольную точку на границе и 
решим уравнение:  

 
 
 
 

 
 

5. Протестируем решение. Например, для          на выходе персептрона должен 

быть 0. Действительно  
 

 
 

2.4. Правило обучения персептрона 

 
Правило обучения персептрона представляет собой процедуру обучения 

с учителем и заключается в поиске параметров (весов и смещений), которые 

обеспечивают совпадения желаемой t и действительной реакции персептрона a 
на заданный входной вектор p. Обучение осуществляется на основе обучающе-

го множества примеров «вход-выход»: 

 
В ходе обучения персептрона на его вход подается каждый входной век-

тор p множества примеров, вычисляется реакция a, которая сравнивается с тре-

буемым выходом t и вычисляется ошибка e. Например, для простого персеп-
трона с одним выходом ошибка равна 

 
В зависимости от значения ошибки корректируется вектор весов персептрона  в 

соответствии с правилом:  

                       (3.5) 



36 
 

Для коррекции (обновления смещений) (смещение эквивалентно некоторому 
дополнительному входу с весом b, для которого p=1) используется упрощенное 

правило 3.5:  

                                               (3.6) 
В том случае, когда персептрон содержит несколько нейронов приведен-

ные правила обучения можно записать в матричной форме:   

                                          (3.7) 
В соответствии с правилами (3.7) для каждого очередного вектора p из 

множества примеров новые значения матрицы весов W
new

 или вектора смеще-

ний b
new

 вычисляются на основе их предыдущих значений W
old

 и b
old

. Обычно 
начальные значения весов и смещений инициализируют небольшими слу-

чайными значениями. Как следует из (3.5), вектор весов отдельного нейрона 
персептрона  входе обучения изменяется на величину очередного входного век-

тора p c учетом ошибки, которая равна -1, 0 или 1.  
Можно доказать, что правило обучения персептрона обеспечивает 

схождение весов к требуемым значениям, обеспечивающим правильную 
классификацию, за конечное число шагов при условии существования реше-

ния (теорема сходимости) [1, 5]. 

Сетевая функция n персептрона играет роль дискриминантной функции. 

В общем случае эта функция описывает уравнение гиперплоскости. Поэтому 
простой персептрон может классифицировать только такие образы входного 

пространства, которые разделимы с помощью гиперплоскости. Иными словами, 
задачи классификации, решаемые простым персептроном, – это линейные се-
парабельные задачи.  

 
3. Варианты заданий и программа работы 

 

3.1. Повторить теоретический материал, относящийся к архитектуре пер-

септрона и его обучению [1, 4, 5]. 
3.2. Для заданной матрицы входных данных P и заданного вектора вы-

ходных значений T (таблица 3.1), разработать простой персептрон, решающий 

задачу классификации 2-х классов. Решение найти с помощью графического 
построения границы решения и вычисления весов и смещения вручную. Про-

тестировать с помощью компьютера полученное решение для всех входных 
векторов (столбцов матрицы P).  

Правила определения номера варианта изложены в лабораторной работе 
1. 

Таблица 3.1 – Варианты задания п. 3.2 

Вариант Матрица P Вектор T 
1 [0 1 1 0; 

 0 0 1 1] 
[0 1 0 1] 

2 [0 1 1 0; 

 0 0 1 1] 

[0 0 1 1] 

3 [0 1 1 0; 

 0 0 1 1] 

[1 1 0 0] 

4 [0 1 1 0; 
 0 0 1 1] 

[1 0 0 1] 

5 [0 1 1 0; 

 0 0 1 1] 

[0 0 1 0] 

6 [0 1 1 0; 
 0 0 1 1] 

[0 0 0 1] 

7 [0 1 1 0; 

 0 0 1 1] 

[1 0 0 0] 

8 [0 1 1 0; 
 0 0 1 1] 

[0 1 0 0] 

9 [0 1 1 0; 
 0 0 1 1] 

[1 1 0 1] 

10 [0 1 1 0; 
 0 0 1 1] 

[1 1 1 0] 

11 [0 1 1 0; 
 0 0 1 1] 

[1 0 1 1] 

12 [0 1 1 0; 

 0 0 1 1] 

[0 1 1 1] 

13 [0 1 1 0 2; 
 0 0 1 1 0] 

[0 0 1 0 1] 

14 [0 1 1 0 2; 

 0 0 1 1 0] 

[0 1 0 0 1] 

15 [0 1 1 0 2; 
 0 0 1 1 0] 

[1 1 0 0 1] 

 

3.3. Даны четыре класса, каждый из которых представлен 2-мя точками 
(столбцами матрицы P), указанными в таблице 3.3. Необходимо: 

- разработать структурную схему персептрона, распознающего эти 4 
класса; 

- выполнить предварительный анализ задачи, изобразив все точки четы-
рех классов и построив графически возможные границы решений персептрона; 

- анализируя границы решений, задать допустимые целевые значения вы-
ходов персептрона (определить матрицу T) для всех входных точек, представ-
ленных столбцами матрицы данных P;  

- изучить функцию ann_PERCEPTRON и используя её, обучить персептрон 
правильному распознаванию входных классов; 

- написать программу, которая: 
- отображает диаграмму размещения входных точек из P на плоско-

сти с координатами (p1, p2); 
- обучает персептрон; 

- накладывает на диаграмму входных точек границы решения после 
обучения персептрона; 

- выполняет тестирование полученного решения для всех заданных 
входных данных, а также для дополнительно выбранных точек из облас-

тей принадлежности входных классов.  

Таблица 3.3 – Варианты задания п.3.3 

Вари-

ант 

Класс 1 

P(:,1), 
P(:,2), 

Класс 2 

P(:,3), 
P(:,4), 

Класс 3 

P(:,5), 
P(:,6), 

Класс 4 

P(:,7), 
P(:,8), 

1 [1; 1], 

[2; 2] 

[-1; 1], 

[-2; 2] 

[-1; -1], 

[-2; -2] 

[1; -1], 

[2; -2] 

2 [2; 1], 
[2; 2] 

[-2; 1], 
[-2; 2] 

[-2; -1], 
[-2; -2] 

[2; -1], 
[2; -2] 

3 [1; 2], 

[2; 2] 

[-1; 2], 

[-2; 2] 

[-1; -2], 

[-2; -2] 

[1; -2], 

[2; -2] 

4 [1; 1], 
[2; 2] 

[-1; 1], 
[-2; 2] 

[-1; 0], 
[-2; -1] 

[1; 0], 
[2; -1] 

5 [2; 1], 

[2; 2] 

[-2; 1], 

[-2; 2] 

[-2; 0], 

[-2; -1] 

[2; 0], 

[2; -1] 

6 [1; 2], 
[2; 2] 

[-1; 2], 
[-2; 2] 

[-1; -1], 
[-2; -1] 

[1; -1], 
[2; -1] 

7 [1; 0], 
[2; 1] 

[-1; 0], 
[-2; 1] 

[-1; -1], 
[-2; -2] 

[1; -1], 
[2; -2] 

8 [2; 0], 
[2; 1] 

[-2; 0], 
[-2; 1] 

[-2; -1], 
[-2; -2] 

[2; -1], 
[2; -2] 

9 [1; 1], 
[2; 1] 

[-1; 1], 
[-2; 1] 

[-1; -2], 
[-2; -2] 

[1; -2], 
[2; -2] 

10 [2; 1], 

[3; 2] 

[-1; 1], 

[-2; 2] 

[-1; -1], 

[-2; -2] 

[2; -1], 

[3; -2] 

11 [3; 1], 
[3; 2] 

[-2; 1], 
[-2; 2] 

[-2; -1], 
[-2; -2] 

[3; -1], 
[3; -2] 

12 [2; 2], 

[3; 2] 

[-1; 2], 

[-2; 2] 

[-1; -2], 

[-2; -2] 

[3; -2], 

[3; -2] 

13 [1; 1], 
[2; 2] 

[-2; 1], 
[-3; 2] 

[-2; -1], 
[-3; -2] 

[1; -1], 
[2; -2] 

14 [2; 1], 

[2; 2] 

[-3; 1], 

[-3; 2] 

[-3; -1], 

[-3; -2] 

[2; -1], 

[2; -2] 

15 [1; 2], 
[2; 2] 

[-2; 2], 
[-3; 2] 

[-2; -2], 
[-3; -2] 

[1; -2], 
[2; -2] 

 

3.4. Выполнить анализ полученных результатов, обратив внимание на то, 
что при каждом новом запуске функции обучения границы решения меняются. 

Сравнить результаты нескольких запусков программы. Сохранит ли программа 
работоспособность, если матрицу целевых значений выходов T формировать 

произвольным образом (например, используя в качестве целевого выходного 
вектора номер его класса в двузначном двоичном коде)?  

3.5. Подготовить и защитить отчет по работе. 

 
4. Методические рекомендации по выполнению работы 

 

4.1. Модуль NeuralNetwork 2.0 пакета Scilab содержит встроенные функ-

ции для обучения и моделирования персептрона:  

[w,b]= ann_PERCEPTRON(P,T) – функция обучения персептрона, код которой с ком-
ментариями приведен в приложении A; 
y= ann_PERCEPTRON_run(P,w,b) – функция моделирования персептрона. 

 

 Ниже приведен пример использования этих функций для обучения и 

моделирования простого персептрона, реализующего логическую операцию 
AND:  
 
//пример обучающей выборки для лог. эл-та AND  
P = [0 0 1 1 ; 0 1 0 1]; 

 T = [0 0 0 1]; 
 // вызов функции обучения  

[w,b] = ann_PERCEPTRON(P,T); 
 Epoch: 1 
 Epoch: 2 

 Epoch: 3 
 Epoch: 4 

 Epoch: 5 
 Epoch: 6 
 // отображение рез-тов 

 --> w 
  w = 2.2113249 1.7560439 

 --> b 
  b = -2.9997789 
 // тестирование элемента  AND 

 // вызов функции моделирования персептрона: 
 // y = ann_hardlim_activ(w*P+repmat(b,1,size(P,2)); 

 y = ann_PERCEPTRON_run(P,w,b) 
  y = 0. 0. 0. 1. 
 

4.2. При выполнении задания 3.2 для графического построения границы ре-
шения простого персептрона и вычисления его весов и смещения следует руко-

водствоваться примером, рассмотренным в п. 2.3 настоящей лабораторной ра-
боты. Для тестирования полученного решения необходимо использовать функ-
цию моделирования персептрона:    
 

y = ann_PERCEPTRON_run(P,w,b). 
 

4.3  При выполнении задания 3.3 необходимо провести предварительный ана-
лиз расположения точек, представляющих классы, на плоскости (p1, p2) и на-

рисовать вручную возможные границы решения персептрона. Поскольку зада-
но 4 класса, то персептрон должен состоять из 2-х нейронов, каждый из кото-

рых создаёт границу, которая делит входное пространство на 2 области. При 
этом первая граница разделяет множество входных примеров на 2 подмножест-

ва, а вторая граница разделяет эти подмножества еще раз на две части. Это по-
зволяет персептрону успешно распознавать 4 заданных класса, которые должны 

быть линейно-разделимыми. Чтобы обучить персептрон этому, необходимо 
корректно сформировать матрицу целевых выходов персептрона T. 

Рассмотрим пример. На рисунке 3.5 изображены 

«точки», представляющие входные классы 
(1,2,3,4) и возможные границы решений персеп-
трона. Необходимо сформировать целевые значе-

ния выходов двух нейронов персептрона. Для 
этого примем решение относительно того, с какой 

стороны границы на выходе нейрона будет 1, а с 
какой – 0. Пусть для границы, отделяющей зачер-

ненные «точки» от не зачернённых, 1 будет со 
стороны зачерненных точек. Пусть для границы, 

            Рисунок 3.5                  отделяющей кружочки от квадратов, на выходе 
нейрона 1 будет со стороны, где изображены квадраты. Тогда векторы целевых 

значений выходов будут следующими:  
класс 1: t = [0;0]; 

класс 2: t = [0;1]; 
класс 3: t = [1;0]; 

класс 4: t = [1;1]. 
Отсюда для заданной матрицы входных данных P (каждый столбец соот-

ветствует «точке» на рисунке)  
 

P= [   1   1   2   2   -1   -2   -1   -2; 

          1   2  -1  0     2   1    -1   -2] 
 

матрица T целевых значений выходов будет равна: 
 

T= [   0   0   0   0   1   1   1   1; 

          0   0   1   1   0   0   1   1]. 
 

Теперь, используя значения матриц P и Т, можно обучить персептрон, вызвав 

функцию обучения:  
 

[w,b]=ann_PERCEPTRON(P,T) 
 

Ниже приведен пример кода программы, обеспечивающей обучение пер-

септрона для рассматриваемого примера, отображение диаграммы расположе-
ния точек и границ решений (рисунок 3.6): 
 

clear all; 
K=4; //количество классов 

N=2; //количество точек в классе 
S=2; //число нейронов в слое  
 

P=[ 1 1 2 2 -1 -2 -1 -2; 
       1 2 -1 0  2  1 -1 -2]; 
 

T=[0 0 0 0 1 1 1 1; 
     0 0 1 1 0 0 1 1]; 

// вызов функции обучения 
[w,b] = ann_PERCEPTRON(P,T); 
 

//построение диаграммы расположения входных точек  
clf; 
j=-3; 

for i=1:K*N //цикл для всех точек данных из P 
    // 3-й  аргумент plot2d ( j )  определяет отображаемый символ 

   //  его допустимые значения -1,-2,…,-15 
    plot2d(P(1,i),P(2,i),j); 
    if modulo(i,2)==0 // меняем j для при выборе нового класса из P 

        j=j-1; 
    end 

end 
xgrid; 
xtitle("Границы решений", "p1", "p2"); 

mtlb_hold('on')  // запрет очистки граф. окна 
 

//отображение границ решений 
x = min(P)-0.1:0.1:max(P)+0.1; 
l_х=length(x); 

y=zeros(S,l_х); 
for i=1:S 
    y(i,:) = -w(i,1)/w(i,2).*x - b(i)./w(i,2); //уравнение линии 

    plot(x,y(i,:)); 
end 
 

// тестирование персептрона 
// вызов функции моделирования персептрона: 

y = ann_PERCEPTRON_run(P,w,b) 
y  =  

   0.   0.   0.   0.   1.   1.   1.   1. 
   0.   0.   1.   1.   0.   0.   1.   1. 
 

 
Рисунок 3.6 – Диаграмма расположения входных точек и границы решений 

5. Содержание отчета 

 

5.1.  Цель работы. 
5.2.  Вариант задания.  

5.3.  Схема простого персептрона, решающего задачу классификации 2-х клас-
сов; графические построения границы решения и вычисления его весов и 

смещения, выполненные вручную; листинг программы с результатами тес-
тирования простого персептрона.  

5.4.  Схема однослойного персептрона, решающего задачу классификации 4-х 
классов; пояснение выбора значений матрицы T; результаты обучения пер-

септрона, правила обучения; диаграмма расположения классов с граница-
ми решений, соответствующими обученному персептрону; листинг про-
граммы с комментариями. 

5.5.  Выводы по результатам исследований. 
 

6. Контрольные вопросы 
 

6.1 Нарисуйте схему однослойного персептрона, объясните все обозначения, 
запишите формулу вычисления вектора выходных значений и объясните её.  

6.2 Сколько линейно-разделимых классов способен распознать персептрон, со-
держащий S нейронов? Объясните почему? 

6.3. Изобразите простой персептрон. Запишите уравнение границы решения. 
6.4. Как построить границу решения простого персептрона, если известны его 

веса и смещения? Как определить область входного пространства, где выход 
персептрона будет равен 1? 
6.5. Как можно графически построить границу решения персептрона. Приведи-

те пример построения. 
6.6. Как связана ориентация вектора весов с границей решения нейрона персеп-

трона?  
6.7. Запишите правило обучения персептрона в векторной и матричной форме? 

6.8. Чему равны значения ошибки персептрона? 
6.9. В каком направлении корректируются значения векторов весов нейронов 

персептрона? 
6.10. Сформулируйте теорему сходимости персептрона. 

6.11. Какой класс задач распознавания способен решать персептрон? 
6.12. Как при известном расположении распознаваемых классов корректно 

сформировать матрицу целевых значений выходов персептрона? Приведите 
пример. 

6.13. Какие встроенные функции модуля Neuralnetwork 2.0 используются для 
обучения и моделирования персептрона? Приведите примеры их вызова. 
6.14. Объясните алгоритм, реализуемый функцией ann_PERCEPTRON. 

Приложение A. Код функции  ann_PERCEPTRON(P, T) 

 
function [w,b] = ann_PERCEPTRON(P, T)  

//инициализация матрицы весов и вектора смещений  
w = rand(size(T,1),size(P,1)); 

b = rand(size(T,1),1); 
iter = %t; // присвоение флагу iter=True, контроль эпох  
itercnt = 0; // счетчик итераций  

while iter == %t  // цикл по эпохам, пока iter=True 
no_err = 0          // счетчик безошибочных классификаций  

    for cnt = 1:size(P,2) // цикл по всем входным примерам (одна эпоха)  
        e = T(:,cnt) - ann_hardlim_activ(w*P(:,cnt)+b);  // вектор ошибки текущего примера  
        w = (w + e*P(:,cnt)'); // обучение матрицы весов  

        b = b + e; // обучение вектора смещений  
        if sum(e) == 0 then  // сумма текущих ошибок равна нулю  

            no_err = no_err + 1;  // число безошибочных классификаций  
        end 
        if  no_err == size(P,2) then //число безошиб. классификаций = числу примеров  

            iter = %f; // сбрасываем флаг цикла эпох  
        end 

    end 
    itercnt = itercnt + 1; // счетчик эпох  
    disp('Epoch: ' + string(itercnt)); 

end 
endfunction 

Лабораторная работа №4 

Исследование адаптивного линейного элемента  

1 Цель работы 

Углубление теоретических знаний в области архитектуры нейронных се-

тей с линейной активационной функцией, исследование свойств квадратичной 
целевой функции и LMS-алгоритма обучения, приобретение практических на-

выков обучения однослойной сети линейных адаптивных элементов при реше-
нии задачи классификации и адаптивной фильтрации. 

2 Основные теоретические положения 

2.1 Структура однослойной сети из адаптивных линейных элементов 

Адаптивный линейный элемент (АЛЭ) (ADALINE - Adaptive Linear Ele-

ment) представляет собой нейрон с линейной функцией преобразования purelin. 
Общая структурная схема однослойной сети из АЛЭ изображена на рисунке 

4.1.  

 
Рисунок 4.1 ─ Схема однослойной сети  из адаптивных линейных элементов 

 

Выходное значение отдельного нейрона сети вычисляется в соответствии с вы-

ражением 

                          (4.1) 
 

2.2 Классификация с использованием АЛЭ 
 

Рассмотрим  АЛЭ с 2-мя входами и 1 выходом (рисунок 4.2.). 

  

Рисунок 4.2 ─ АЛЭ и его граница решения 

С учетом только 2-х входов выходной сигнал нейрона будет равен  
 

                      (4.2) 
 

Граница решения определится из условия  
 

                   (4.3) 
 

АЛЭ разделяет входное пространство на две области, где a>0 и a<0 (см. рису-
нок 4.2). Таким образом, АЛЭ (как и персептрон) может использоваться для 

классификации линейно сепарабельных объектов. 
 

2.3 Целевая функция нейронной сети 
 

Задача обучения нейронных сетей сводится к поиску весов и смещений, 
которые обеспечивают необходимую эффективность сети. Количественная ме-

ра эффективности сети определяется целевой функцией или критерием эф-
фективности. Целевая функция зависит от параметров сети и имеет малые зна-

чения, когда сеть функционирует хорошо, и большие значения, когда сеть 
функционирует плохо. Поэтому целью обучения является поиск параметров 

сети, которые обеспечивают минимум целевой функции.  
Пусть F(x) непрерывная целевая функция, где x = [x1,x2,…, xn]

T
 ─ вектор  

столбец параметров сети в n-мерном евклидовом пространстве. Необходимо 
найти минимум F(x) по x. В точке x=x* будет минимум целевой функции, если 
градиент функции (вектор первых частных производных F(x) по x) будет равен 

нулю в этой точке (условие первого порядка) 
 

    .                                          (4.4) 
 

Точки x, удовлетворяющие этому условию, называются стационарными. В 
точке x* будет существовать строгий минимум, если (условие2-го порядка) 
 

   ,                                     (4.4)  

где       – квадратная матрица Гессе целевой функции (матрица вторых ча-
стных производных F(x) по x, называемая гессианом),  Δx=x - x* – окрестность 

точки минимума. Чтобы условие (4.4) выполнялось для любых ||Δx||>0 доста-
точно, чтобы матрица Гессе была положительно определена. Для того чтобы в 

точке x* находился минимум (строгий или слабый) достаточно, чтобы матрица 
Гессе была положительно полуопределена. Напомним, что матрица положи-

тельно определена, если все собственные числа λ матрицы положительны. Если 
все собственные числа матрицы не отрицательны, то матрица положительно 

полуопределена. 
При обучении нейронных сетей часто используют квадратичную целевую 

функцию. Общая форма квадратичной целевой функции  

                                     (4.5) 
где A – симметричная матрица. Градиент и гессиан квадратичной целевой 
функции F(x) соответственно равны: 
 

                             (4.6) 
 

2.4 Алгоритм наискорейшего спуска 
 

Цель оптимизации заключается в поиске вектора параметров сети  x*, ко-

торый минимизирует целевую функцию F(x). Для поиска x* применяют алго-
ритмы последовательного приближения – итеративные алгоритмы. В соответ-
ствии с такими алгоритмами поиск начинается с начального значения x0  и на 

каждом шаге k вектор параметров корректируют в соответствии с формулой 
 

                                        (4.7) 
 или  

                                  (4.8) 
где pk – вектор, определяющий направление поиска;  αk – скорость обучения, 

определяющая длину шага Δx k. 
Итеративные алгоритмы оптимизации отличаются выбором вектора на-

правления поиска pk  и способами вычисления значений скорости обучения. 
Так, в алгоритме наискорейшего спуска (SDA– steepest descent algorithm) на-

правление поиска обратно направлению вектора градиента, т.е. pk  = - gk , где 
           –  значение вектора градиента целевой функции на k–ой итерации 

алгоритма. Соответственно процедура SDA определяется выражением:  
 

                                            (4.9) 

Если квадратичная целевая функция F(x) имеет сильный минимум, то все соб-
ственные значения матрицы Гессе A должны быть положительными и условие 

схождения алгоритма SDA задается неравенством: 

,                                                    (4.10) 
где λmax — максимальное собственное число матрицы Гессе. Так как собствен-
ные числа матрицы Гессе квадратичной целевой функции определяют кривизну 

поверхности целевой функции, то в соответствии с (4.10) скорость обучения 
устойчивого SDA обратно пропорциональна максимальной кривизне целевой 

функции.  
 

2.3 Решение минимума среднего квадрата ошибки 
 

Рассмотрим АЛЭ с одним выходом. Для упрощения анализа введем сле-
дующие обозначения: x

T
 =[1w b] и z

T
=[p 1] . Тогда выходной сигнал АЛЭ мож-

но представить в виде скалярного произведения a= x
T 

z. 

В ходе обучения с учителем параметры (веса и смещения) АЛЭ корректи-
руются таким образом, чтобы выходной сигнал a имел наилучшее приближение 

к желаемой реакции t. Для этого вычисляют ошибку приближения e=(t-a) и ми-
нимизирует средний квадрат ошибки (СКО, MSE – mean square error). С уче-
том введенных обозначений целевая функция F(x), заданная в виде СКО, будет 

равна  

                (4.11) 
 

где Е – символ математического ожидания.  Раскроем (4.11):  
 

 
 

СКО целевая функция может быть переписана в квадратичной форме 
 

   (4.12) 
 

Здесь h – вектор кросс-корреляции, R – корреляционная матрица.  

Сравним целевую функцию (4.12) с квадратичной целевой функцией об-
щего вида (4.5). Получим, что  

 

    d=-2h,  A=2R ,                                                    (4.13) 
 

т.е. матрица Гессе целевой функции (4.12) равна 2R. Точка минимума целевой 
функции F(x) определится из условия 1-го порядка (4.4): 
 

.              (4.14) 
 

Если корреляционная матрица положительно определена, то будет существо-

вать единственная точка минимума целевой функции (4.12) 
 

  x*=R
-1

 h.                                                     (4.15) 
 

Решение (4.15), позволяющее вычислять параметры АЛЭ через матрицу 

R и вектор h, называют  решением минимума СКО (оптимальным решени-
ем Винера). На практике точные значения h и R обычно неизвестны. Однако, 

если входные данные, представляемые вектором p, соответствуют стационар-
ному эргодическому процессу, то, используя усреднение по времени, можно 
оценить h и R.  Если на k-м шаге имеются оценки матрицы R и вектора h, обо-

значаемые как Rk и hk , то для поиска винеровского решения можно использо-
вать алгоритм наискорейшего спуска.  Подставив значение градиента из (4.14) в 

выражение (4.9), можно получить алгоритм наискорейшего спуска, сходящийся 
к решению (4.15): 

gk   =-2hk+2Rkx k   ,                                             (4.16) 
x k+1 = x k - α k  g k ,                                              (4.17) 

Отметим, что существование решения (4.15) целиком зависит от R. 
Следовательно, характеристики входного вектора p определяют возможность 

существования единственного решения. 
 

2.4. LMS алгоритм 

 
На практике вычислять оценки матрицы R и вектора h затратно. Поэтому 

выполняют аппроксимацию алгоритма наискорейшего спуска,  в которой ис-
пользуют приближенные оценки градиента. Идея заключается в аппроксимации 

среднего квадрата ошибки  (4.11)  более простым выражением 
 

                                (4.18) 
 

в котором вместо математического ожидания E квадрата ошибки используется 
просто квадрат мгновенной ошибки. Тогда оценка градиента, называемая в 

этом случае  стохастическим (случайным) градиентом, будет равна 
 

                                          (4.19) 
 

Так как  a= xT z  и e=(t-a), то стохастический градиент будет равен 
 

                               (4.20) 
 

Подставив стохастический градиент (4.19)  в алгоритм SDA, получим  
получим  LMS-алгоритм (LMS- least mean square, алгоритм  наименьшего 

среднего квадрата, также называемый правилом обучения Уидроу-Хоффа)  
 

,                                (4.21) 

или  

  (4.22) 
 

Если используется слой из АЛЭ, то LMS-алгоритм записывается в мат-

ричной форме: 
 

       (4.21) 
 

Поскольку в выражении  (4.21) используются неточные оценки градиента, 
то в процессе обучения возникает шум, в результате чего траектория движения 

вектора параметров x k  к вектору x* в LMS-алгоритме  не совпадает с траекто-
рией движения в алгоритме наискорейшего спуска. LMS-алгоритм привлекает 

своей простотой. 
Условие сходимости SDA (4.9) определялось значением собственных чи-

сел матрицы Гессе A квадратичной целевой функции. Для  LMS справедливо, 
что A=2R . Поэтому по аналогии можно записать условие сходимости LMS-
алгоритма  

                                      (4.22) 

или    

                                                                                         (4.23) 
где  λi - собственные числа  корреляционной матрицы R. Если это условие вы-

полняется, то можно показать, что LMS алгоритм, обрабатывающий на каждой 
итерации один входной вектор, сходится к решению, совпадающему с решени-
ем минимума СКО (4.15) . 

 
2.5 Адаптивная фильтрация 

 

На основе АЛЭ реализуют адаптивные фильтры (рисунок 4.3). Для этого 

на входе АЛЭ размещают линию задержки (tapped delay line – TDL). Один эле-
мент линии задержки, обозначенный на рисунке 4.3 буквой D, представляет со-

бой регистр, который запоминает входное значение на время, равное шагу дис-
кретизации. В этом случае вектор p , поступающий на вход АЛЭ, соответствует 
совокупности значений на выходе линии задержки – y(k),y(k-1),…,y(k-R+1). 

 
 
Рисунок 4.3 – Схема адаптивного фильтра на основе АЛЭ 

 
С учетом принятых обозначений выход адаптивного фильтра запишется в 

виде: 

……(4.24) 
 

Уравнение (4.24) соответствует фильтру с конечной импульсной характе-

ристикой (КИХ), который также называют трансверсальным фильтром.  
Адаптивные фильтры используются для решения задач моделирования, 

идентификации, подавления шумов, компенсации эхо-сигналов, построения эк-
валайзеров каналов связи, кодирования речи, управления адаптивными антен-

ными решетками и др. Рассмотрим применение адаптивного фильтра для пред-
сказания сигналов. На рисунке 4.4 изображена схема адаптивного предсказате-

ля, построенного на основе АЛЭ, обеспечивающего предсказание текущего от-
счета входного сигнала y(k) по 2-м предыдущим отсчетам: y(k-1) и y(k-2). 

Рисунок 4.4 – Адаптивный предсказатель 

 

Отметим, что адаптивный линейный предсказатель в ходе обучения обес-

печивает поиск параметров, которые минимизируют СКО в виде целевой функ-
ции (4.12). Так как желаемое значение на выходе предсказателя должно совпа-

дать с текущим входным значением, т.е. t(k)=y(k) и входной вектор предсказа-
теля равен 

z(k)=p(k)=[y(k-1) y(k-2)]
 T 

 ,                                     (4.25) 
 

то составляющие целевой функции (4.12) запишутся в виде: 
 

             
                   

                   
 ,           (4.26) 

 

            
          

          
 ,                     .         (4.27) 

 

При известных значениях матрицы R и вектора h оптимальные веса пред-
сказателя определятся в виде решения минимума СКО (4.15). Например, пусть  

 

   
        
        

   ,          
    
    

 . 
 

Тогда  в соответствии с (4.15) оптимальный вектор параметров предсказа-
теля будет равен x*=[w1,1 w1,2]=[1 0]. 

При исследовании свойств алгоритмов, базирующихся на СКО целевой 
функции полезно использовать графики линий контуров постоянного уровня 

e=const. Эти графики для поверхности квадратичной целевой функции, являю-
щейся параболоидом, представляют собой гиперэллипсы (рисунок 4.5). Глав-

ные оси этих гиперэллипсов ориентированы в направлении собственных векто-
ров матрицы R, длина осей обратно пропорциональна собственным числам R. 

Чем больше соответствующее собственное число матрицы R, тем больше гра-
диент F(x) вдоль соответствующей оси эллипса и, соответственно, линии кон-

туров равных уровней располагаются ближе друг к другу. Для рассматриваемо-
го примера R собственные числа равны λ1=0,17 и  λ2=1,97. Соответственно, эл-

липс вытянут вдоль собственного вектора с меньшим собственным значением, 
т.е λ1=0,17. 

Так как вычисление оптимальных параметров предсказателя в соответст-

вии с (4.15) требует вычисления обратной матрицы, то на практике поиск опти-
мальных параметров выполняют на основе LMS-алгоритма. При этом скорость 

схождения, точность и устойчивость LMS-алгоритма сильно зависит от скоро-
сти обучения α. Максимальное устойчивое значение α определяется на основе 

соотношения (4.23). Для рисунка 4.5 αmax=1/ λ2=0,507. На практике при исполь-

зовании адаптивных фильтров определять собственные числа R не представля-
ется возможным. Однако, учитывая свойство корреляционной матрицы, со-

гласно которому след корреляционной матрицы равен сумме её собственных 

чисел, т.е.             
      

 
   , можно заключить, что для СКО целевой 

функции           . C учетом этого соотношения условие (4.23) можно 

представить в виде более жесткого условия, гарантирующего устойчивую ра-

боту LMS-алгоритма для трансверсального фильтра: 
 

  
 

     
 
   

 
 

    
                                             (4.28) 

 

где   
  — дисперсия  входного сигнала адаптивного фильтра. В соответствии с 

(4.28) для рассматриваемого примера должно быть                     . 

 Так как LMS-алгоритм аппроксимирует алгоритм наискорейшего спуска, 
то при малых скоростях обучения   траектория движения вектора параметров 

LMS-алгоритма будет менее хаотична и в среднем будет соответствовать дви-

жению вектора параметров алгоритма наискорейшего спуска, которая направ-
лена перпендикулярно линиям равных контуров, как показано на рисунке 4.5. 

Однако при значениях α близких к αmax продвижение к некоторому малому 
уровню значений целевой функции происходит быстрее, несмотря на хаотич-
ность. Кружок в центре рисунка 4.5. соответствует решению минимума СКО 

(4.15). Видно, что при выбранных значениях α  решение, получаемое с помо-
щью LMS-алгоритма, приближается к решению минимума CKO. 
 

 
Рисунок 4.5 – Траектория движения вектора параметров предсказателя 

 при использовании LMS-алгоритма: слева                      
 

3. Варианты заданий и программа работы 
 

3.1. Повторить теоретический материал, относящийся к архитектуре и 
правилам обучения адаптивного линейного элемента [1, 4, 5]. 

3.2. Даны четыре класса, каждый из которых представлен 2-мя точками 
(столбцами матрицы P), указанными в таблице 3.2 к лабораторной работе №3 

(используйте свой вариант). Необходимо: 

- разработать структурную схему классификатора на основе АЛЭ, распо-
знающего эти 4 класса; 

- выполнить предварительный анализ задачи, изобразив точки четырех 
классов и построив графически возможные границы решений; 

-  задать ту же целевую матрицу T, которая использовалась при выполне-

нии задания 3.3 в лабораторной работе №3, заменив все нули на -1;  
- полагая, что все входные векторы p равновероятны, написать програм-

му, вычисляющую корреляционную матрицу R , собственные числа гессиана 
целевой функции A=2R и максимальное устойчивое значение параметра αmax 

LMS-алгоритма; 
- изучить встроенные функции ann_ADALINE и ann_ADALINE_online пакета 

NeuralNetworks 2.0, реализующие блочный и последовательный варианты LMS-
алгоритма; 

-  используя указанные функции, написать программу, которая: 
- отображает диаграмму размещения входных точек из P на плоско-

сти с координатами (p1, p2); 
- обучает АЛЭ правильному распознаванию входных классов с ис-

пользованием 2-х указанных функций при разных значениях параметра α; 
- строит кривые обучения - зависимости СКО от номера эпохи для 

2-х указанных функций (для этого необходимо модифицировать встроен-

ные функции ann_ADALINE и ann_ADALINE_online); 
- накладывает на диаграмму входных точек границы решения после 

обучения слоя АЛЭ; 
- выполняет тестирование полученного решения для всех заданных 

входных данных;  
- сравнить получаемые границы решения слоя АЛЭ с границами решения 

персептрона, полученными в лабораторной работе №3, обратив внимание 
на равноудаленность границ от точек соседних классов для случая слоя 

АЛЭ; 
- сравнить кривые обучения слоя АЛЭ двумя модифицированными функ-

циями ann_ADALINE и ann_ADALINE_online для случая, когда параметр  α 
LMS-алгоритма значительно меньше αmax и когда он близок к αmax . 
3.3. Исследуйте адаптивный линейный предсказатель (рисунок 4.4): 

- сгенерируйте входной y(k)  и желаемый  t(k)=y(k) сигналы адаптивно-
го предсказателя. При этом параметры генератора выбирайте в соот-

ветствии с вариантом из таблицы 4.1. Для генерации используйте сле-
дующий программный код:  

 
//Значения параметров генератора L, F, Fc выбирайте из таблицы 4.1; 

//Параметры: 
L=      ;   
F=      ; 

Fc=     ; 
//Генератор полигармонического входного сигнала; 

td=1/(20*F); 
t=0:td:2/F; 

fi=(2*%pi*F).*t; 
Y=0; 
for i=1:L 

    Y=Y+(Fc)/(Fc+2*%pi*F*i)*sin(fi*i); 
end 

T=Y; 
 

Таблица 4.1 – Параметры генератора входного сигнала 

Вариант Число состав-

ляющих L 

Основная час-

тота F 

Частота среза 

Fc 

1 5 0.01 F*5 

2 10 0.01 F*6 

3 15 0.01 F*7 

4 20 0.01 F*8 

5 25 0.01 F*9 

6 5 0.02 F*5 

7 10 0.02 F*6 

8 15 0.02 F*7 

9 20 0.02 F*8 

10 25 0.02 F*9 

11 5 0.04 F*5 

12 10 0.04 F*6 

13 15 0.04 F*7 

14 20 0.04 F*8 

15 25 0.04 F*9 

 
- запишите развернутые выражения для всех элементов (R,h,c) целевой 

функции предсказателя, заданной в виде СКО (4.12); 
- вычислите конкретные значения матрицы R, вектора h и константы с 

для сгенерированного входного сигнала y(k) ; 
- вычислите собственные значения и собственные векторы матрицы 

Гессе целевой функции предсказателя, точку минимума целевой 
функции, постройте линии контуров равных уровней целевой функ-

ции; 
- вычислите максимальное устойчивое значение скорости обучения αmax 

для LMS-алгоритма; 
- напишите программу, обучающую предсказатель с использованием 

встроенной функции ann_ADALINE_predict пакета NeuralNetworks 2.0; 
- модифицируйте функцию ann_ADALINE_predict таким образом, чтобы 

по результатам её работы можно было построить кривую обучения 

предсказателя и траекторию движения вектора параметров предсказа-
теля на диаграмме контуров равных уровней; 

- постройте в одном графическом окне графики входного процесса и его 
предсказанных значений, кривую обучения, траекторию движения 

вектора параметров на диаграмме контуров;  

- убедитесь, что алгоритм обучения сходится, если α<αmax   и нестабилен 
когда  α>αmax ; 

- убедитесь, что при малых α траектория движения вектора параметров 
при использовании LMS алгоритма аппроксимирует в среднем траек-
торию движения вектора параметров алгоритма наискорейшего спус-

ка. 
3.4. Подготовьте и защитите отчет по работе. 

 
4. Методические рекомендации по выполнению работы 

 

4.1. Модуль NeuralNetwork 2.0 пакета Scilab содержит встроенные функ-

ции для обучения и моделирования АЛЭ:  
 

[w,b]= ann_ADALINE(P, T, alpha, itermax, initfunc) – функция обучения АЛЭ, 

реализующая блочный (пакетный, batch) LMS-алгоритм, код которой с 
комментариями приведен в приложении A; 

[w,b]= ann_ADALINE_online(P, T, alpha, itermax, initfunc) – функция обучения 
АЛЭ, реализующая последовательный (адаптивный) LMS-алгоритм, код 

которой с комментариями приведен в приложении Б; 
y= ann_ ADALINE _run(P,w,b) – функция моделирования слоя АЛЭ, код ко-
торой с комментариями приведен в приложении Г. 

 

4.2. При выполнении задания 3.2 необходимо сначала сформировать мат-

рицу T целевых выходов слоя из АЛЭ. Используйте для этого матрицу целевых 
выходов, построенную в предыдущей лабораторной работе, заменив в ней все 

нули на -1. Например, для рисунка 3.5 матрица T целевых значений выходов 
классификатора, реализованного в виде слоя из АЛЭ, будет равна: 

 

T= [   -1   -1   -1   -1   1   1   1   1; 
          -1   -1     1    1  -1   -1   1   1 ]. 
 

Так как входные векторы p равновероятны, то для вычисления корреляционной 
матрицы R используйте простую формулу 
 

          
  

    , 
 

где   
          – расширенный входной вектор слоя АЛЭ (здесь 1 соответству-

ет входу с весом, равным смещению b), Q – число входных векторов p. 

 Для вычисления собственных значений гессиана A=2R целевой функции, 
заданной в виде СКО, используйте вызов встроенной функции Scilab:  
 

evals=spec(2*R) 
 

 Для написания программы, которая выполняет обучение АЛЭ правиль-
ному распознаванию входных классов в двух режимах – блочном и последова-

тельном – используйте модифицированные функции обучения  ann_ADALINE и 

ann_ADALINE_online:  
 

[w1,b1,mse1] = ann_ADALINE1(P,T,alpha,maxiter,'zeros'); 

[w2,b2,mse2] = ann_ADALINE1_online(P,T,alpha,maxiter,'zeros'); 
 

Функции ann_ADALINE1 и ann_ADALINE1_online модифицируется таким образом, 

чтобы можно было после обучения построить кривые обучения – зависимости 
ошибки СКО от номера эпохи обучения. Для этого указанные функции, кроме 

параметров обученного слоя АЛЭ, должны возвращать массивы значений сред-
них квадратов ошибок, соответственно mse1 и mse2. Схема модификации кода 

функции ann_ADALINE: 

 
function [w, b, mse1]=ann_ADALINE1(P, T, alpha, itermax, initfunc) 

… 
itercnt = 0;  

mse1=zeros(1,itermax); 
while itercnt < itermax 

 … 
 itercnt = itercnt + 1; 
     mse1(itercnt) = mean(e.^2); 

 //disp('Epoch:…. 
end 

endfunction 

 
Схема модификации кода функции ann_ADALINE1_online: 

 
function [w, b, mse2]=ann_ADALINE1_online(P, T, alpha, itermax, initfunc) 

… 
itercnt = 0; 
mse2=zeros(1,itermax); 

while itercnt < itermax 
    for cnt = 1:size(P,2) 

          …. 
         e_all(cnt) = e.^2 
    end 

    itercnt = itercnt + 1; 
    mse2(itercnt)=mean(e_all); 

   //disp('Epoch:… 
end 
endfunction 

 

Добавленные операторы выделены красным цветом. Операторы отображения 

disp внутри функций рекомендуется исключить для уменьшения объемов дан-
ных, выводимых на экран. 

Теперь, используя значения матриц P и Т, вычисленное значение αmax, 
можно обучить слой АЛЭ, вызвав модифицированные функции обучения.  По 

результатам обучения следует построить графики кривых обучения для 2-х мо-
дифицированных функций:  

 
x_axis=1:maxiter; 
plot(x_axis,mse1(x_axis),'r',x_axis,mse2(x_axis),'g'); 

xtitle('Средний квадрат ошибки', 'Эпоха','СКО'); 

В качестве примера на рисунке 4.6 изображены кривые обучения при раз-
ных значениях параметра скорости обучения α: когда α мало и когда близко к 

максимально допустимому значению (для рассматриваемого примера αmax 
=0,19) . 

 
Рисунок 4.6 – Сравнение кривых обучения: слева α=0,02; справа α=0,19 

 

Как следует из рисунка 4.6, блочный LMS-алгоритм при малых α требует 

для обучения большего числа эпох, чем последовательный алгоритм, но в итоге 
дает более точное решение. Последовательный LMS-алгоритм при малых α 

обучается быстрее, но при значении α = αmax теряет устойчивость.  
Для отображения диаграммы расположения точек классов и границ полу-

ченных решений (рисунок 4.7) используйте код, разработанный при выполне-
нии лабораторной работы №3. 

 
 

Рисунок 4.7 – Границы решений слоя АЛЭ (блочный LMS-алгоритм) 

Сравните полученное решение с решением, изображенным на рисунке 
3.5. Обратите внимание, что LMS-алгоритм находит границы решения, равно-

удаленные от центров соседних классов, и получаемое решение не зависит от 
способа инициализации весов слоя АЛЭ. В то же время слой персептронов 
формировал различные границы решения при разных начальных значениях ве-

сов. 
  4.3. В задании 3.3 требуется обучить линейный предсказатель значений 

стационарного сигнала y(k) по двум его предыдущим значениям. Соответст-
вующая схема предсказателя с двумя элементами задержки D=2 изображена на 

рисунке 4.4. В соответствии с заданием вектор отсчетов входного сигнала Y и   
вектор целевых значений T  предсказателя генерируются с помощью про-

граммного кода, приведенного в п.3.3. Отметим, что T=Y. 
На первом этапе необходимо вычислить все элементы (R,h,c) целевой 

функции предсказателя, заданной в виде СКО (4.12). Для этого следует вос-
пользоваться общими формулами (4.26)-(4.27). Чтобы применить эти формулы, 

нужно сначала получить значения векторa p , элементы которого соответству-
ют выходам линии задержки (см. рисунок 4.3). Для формирования матрицы P, 

каждый столбец которой равен очередному входному вектору p АЛЭ, можно 
использовать начальную часть кода встроенной функции ann_ADALINE_predict 

(приложение В), предназначенной для обучения адаптивного линейного пред-

сказателя в последовательном режиме на основе LMS-алгоритма: 
 

T=Y; 
D=2; 
P = []; 
for cnt = 1:D // для каждого выхода линии задержки  

    //формируем строки  матрицы P из отсчетов Y 

    // очередная строка P – сдвинутая на один отсчет копия предыдущей строки 
    P = [P; Y(cnt:$-D+cnt-1)]; 

end 

//формируем вектор целевых значений c длиной, равной длине строки из P 
T = T(1:$-D+1); 
 

Теперь, используя формулы (4.26)-(4.27),  можно вычислить значения R,h 
и c. Для вычисления собственных векторов и собственных значений матрицы 

Гессе (A=2R) целевой функции предсказателя следует использовать вызов 
встроенной функции Scilab в форме: 

 

[evals, diagevals]=spec(2*R). 
 

Здесь переменная evals – вектор-столбец из собственных значений матри-
цы 2R, а diagevals – матрица, столбцы которой представляют собственные век-

торы матрицы 2R. При известных значениях матрицы R и вектора  h точка ми-
нимума целевой функции (x*, xstar) находится в виде решения минимума сред-

него квадрата ошибки (4.15).  
Для построения контуров равных уровней целевой функции следует по-

лучить уравнение её поверхности (4.12) в виде функции матрицы R, вектора h и 

элементов вектора х. Пусть 

      ,     
    

    
 . 

 

Тогда для  вычисления значений целевой функции (4.12) в зависимости от 

её  параметров можно определить в Scilab следующую функцию: 
 

function z=f(w1, w2, c, h, Cor) 
    x=[w1;w2]; 
    z=c-2*x'*h+x'*Cor*x; 

endfunction 
 

Для построения 3D поверхности и линий контуров равных уровней ис-

пользуйте функции feval, surf и contour2d:  
 

x=linspace(-20,20,100); 
y=linspace(-20,20,100); 
z=feval(x,y,f); //вычисляем значения высот целевой функции f на сетке x,y 
 

clf(1); 

figure(1); 
subplot(1,2,1); 
surf(x,y,z');  //строим 3D поверхность 

subplot(1,2,2); 
contour2d(x,y,z,30); //отображаем линии контуров равных уровней  
xset("fpf"," "); //подавляем отображение значений на линиях контуров 

xtitle("Контуры равных уровней квадратичной функции","w1","w2"); 
xgrid; 

plot(xstar(1),xstar(2),'*b'); //отображаем точку решения минимума СКО (4.15)  
 

В результате работы приведенного фрагмента кода будут построена 3D 

поверхность и линии контуров равных уровней (рисунок 4.8., см. ниже). 
Для обучения адаптивного предсказателя используйте встроенную функ-

цию ann_ADALINE_predict пакета NeuralNetworks 2.0. Чтобы по результатам её 
работы можно было построить кривую обучения предсказателя и траекторию 

движения вектора параметров предсказателя модифицируйте функцию по схе-
ме: 
 

function [w,b,y,ee,mse,W] = ann_ADALINE1_predict(X,T,alpha,itermax,D,initfunc) 

… 

mse=zeros(1,itermax); // создаем вектор СКО 
W=[w]; //матрица, каждая строка которой – вектор весов на очередном шаге 
itercnt = 0; // Счетчик итераций - эпох 
while itercnt < itermax 

    for cnt = 1:size(P,2)  // Цикл по всем обучающим примерам из P (1 эпоха) 

        … 
        w = (w + 2*alpha*e*P(:,cnt)'); 
        // b = b + 2*alpha*e; 

        // Вычисляем и запоминаем квадраты текущих  ошибок  
        e_all(cnt) = e.^2; 

        W=[W;w]; 
    end 
    itercnt = itercnt + 1; 

    mse(itercnt)=mean(e_all); // вычисляем СКО на шаге и запоминаем 
    //disp('Epoch: … 
     …  

end 
endfunction 

 

Теперь функция обучения будет возвращать вектор СКО ошибок mse и 
матрицу W из векторов-строк весовых коэффициентов предсказателя на каждой 

итерации. Вектор mse позволяет построить кривые обучения (см. пример на 
рисунке 4.6), а матрица W – траекторию движения вектора параметров предска-

зателя, наложив её (траекторию) на линии равных контуров (рисунок 4.8): 
 

plot2d(W(:,1),W(:,2),5); 
 

 
Рисунок 4.8 – Поверхность целевой функции, линии контуров 

и траектория движения вектора параметров для LMS-алгоритма. 
 

5. Содержание отчета 
 

5.1.  Цель работы. 

5.2.  Вариант задания.  
5.3.  Схема слоя АЛЭ, решающего задачу классификации 4-х классов; вычис-

ленное значение матрицы R и её собственных чисел, значения весов и 
смещений, максимальное значения скорости обучения; диаграмма распо-

ложения классов с границами решений, соответствующими обученному 
слою АЛЭ; графики с кривыми обучения; листинг программы с коммента-
риями. 

5.4.  Схема адаптивного линейного предсказателя; правила обучения; результа-
ты вычислений: значение  матрицы R и её собственных чисел и векторов, 

значение максимальной устойчивой скорости обучения; решение миниму-
ма СКО; кривые обучения и результирующие значения параметров пред-

сказателя; графики 3D поверхности целевой функции и линий контуров 
равных уровней, траектории движения вектора параметров; листинг про-

граммы с комментариями. 

5.5.  Выводы по результатам исследований. 
 

6. Контрольные вопросы 
 

6.1 Нарисуйте схему однослойной сети из адаптивных линейных элементов, 

объясните все обозначения, запишите формулу вычисления вектора выходных 
значений и объясните её.  

6.2 Сколько линейно-разделимых классов способна распознать сеть, содержа-
щий S адаптивных линейных нейронов? 

6.3. Изобразите АЛЭ c 2-мя входами. Запишите уравнение границы решения. 
6.6. Как связана ориентация вектора весов с границей решения АЛЭ? 

6.7. Что понимают под целевой функцией сети? 
6.8. Сформулируйте условие минимума первого порядка для целевой функции. 

6.9. Сформулируйте условие минимума второго порядка для целевой функции. 
6.10. Что такое матрица Гессе целевой функции? Каким условиям должна удов-

летворять эта матрица, чтобы целевая функция имела строгий минимум? 
6.11. Запишите выражение для квадратичной целевой функции в общей форме, 

найдите градиент и гессиан этой функции. 
6.12. Сформулируйте алгоритм наискорейшего спуска. Запишите условие схо-
ждения алгоритма. 

6.13.  Запишите выражение для целевой функции в виде среднего квадрата 
ошибки. Приведите его к квадратичной форме. 

6.14. Запишите условие минимума СКО целевой функции и получите решение 
минимума СКО. 

6.15. Что такое стохастический градиент? Сформулируйте LMS-алгоритм. 
6.16. Запишите выражения LMS-алгоритма в матричной форме, сформулируйте 

условие схождения алгоритма. 
6.17 Изобразите схему адаптивного фильтра на основе АЛЭ. Укажите области 

применения адаптивной фильтрации.  
6.18. Изобразите схему адаптивного линейного предсказателя. Запишите выра-

жения для вычисления корреляционной матрицы R и вектора кросс-корреляции 
h. 
6.19. Сформулируйте гарантирующие условие схождения LMS-алгоритма для 

трансверсального фильтра. 
6.20. Что такое кривая обучения? Как её построить? Какие выводы можно сде-

лать, анализируя кривую обучения? 
6.21. Как ориентирована траектория движения вектора параметров на графике 

линий контуров равных уровней? Чем объясняется хаотичность траектории для 
LMS-алгоритма? 

6.22. Объясните алгоритмы, реализуемые функциями ann_ADALINE, 

ann_ADALINE_online, ann_ADALINE_predict. 

Приложение A. Код функции  ann_ADALINE 
 
function [w, b]=ann_ADALINE(P, T, alpha, itermax, initfunc) 

// Обучение слоя адаптивных линейных нейронов  в блочном режиме 
// Длина блока Q  равна числу обучающих примеров  
// Примеры вызовов  

//     [w,b] = ann_ADALINE(P,T) 
//     [w,b] = ann_ADALINE(P,T,alpha,itermax,initfunc)  
//Параметры:  

//     P : Матрица обучающих примеров (RxQ) 
//     T : Матрица целевых выходных значений (SxQ) 
//     alpha : Скорость обучения (по умолчанию =0.01) 

//     itermaxs : Максимальное число итераций – эпох (по умолчанию – 100) 
//     initfunc : Функция инициализации значений w и b: 'rand', 'zeros', или 'ones' 
//                   По умолчанию используется ‘rand’.  

//     w : веса сети (SxR)) 
//     b : смещения (Sx1) 
 

// 1.===== Обработка списка входных аргументов функции ====== 
rhs=argn(2); // argn(2) – возвращает число аргументов  функции  
if rhs < 2; error("Должно быть минимум 2 аргумента: P и T"); end     

if rhs < 3; alpha = 0.01; end   
if rhs < 4; itermax = 100; end 
if rhs < 5; w = rand(size(T,1),size(P,1)); b = rand(size(T,1),1); end 

//Инициализация весов и смещений 
if  rhs == 5 then 
    select initfunc 

    case 'rand' then  
        w = rand(size(T,1),size(P,1));  
        b = rand(size(T,1),1); 

    case 'zeros' then 
        w = zeros(size(T,1),size(P,1));  
        b = zeros(size(T,1),1); 

    case 'ones' then  
        w = ones(size(T,1),size(P,1));  
        b = ones(size(T,1),1);         

    else  
        error("Неверное значение входного аргумента 5”);       
    end 

end 
 
if itermax == []; itermax = 100; end 

if alpha == []; alpha = 0.01; end     
 
//2. ====== Реализация правил обучения ========================== 

itercnt = 0; // Счетчик итераций - эпох 
while itercnt < itermax 
    // Вычисляем вектор выхода сети a и вектор ошибки сети e 

    // сразу для всего блока данных P (1 эпоха) 
    n = w*P + repmat(b,1,size(P,2));  
    a = ann_purelin_activ(n);  

    e = T - a; 
    //Реализуем блочное  правило обучения Уидроу-Хоффа 
    // вычисляется среднее обновление для w и b  в пределах блока 

    w = w + (2*alpha*e*P')./size(P,2); 
    b = b + 2*alpha*mean(e,2);  
    // Вычисляем вектор ошибки при обновленных w и b 

    n = w*P +  repmat(b,1,size(P,2));  
    a = ann_purelin_activ(n);  
    e = T - a;    

    // Вычисляем значения СКО  
    mse = mean(e.^2);  

    itercnt = itercnt + 1; 
    disp('Эпоха: ' + string(itercnt ) + ' СКО: ' + string(mean(mse)));  
end 

endfunction 
 

Приложение Б. Код функции  ann_ADALINE_online 

function [w, b]=ann_ADALINE_online(P, T, alpha, itermax, initfunc) 

// Обучение слоя адаптивных линейных нейронов  в последовательном режиме  
// Примеры вызовов  
//     [w,b] = ann_ADALINE_online(P,T)  

//     [w,b] = ann_ADALINE_online (P,T,alpha,itermax,initfunc)  
//Параметры:  
//     P : Матрица обучающих примеров (RxQ) 

//     T : Матрица целевых выходных значений (SxQ) 
//     alpha : Скорость обучения (по умолчанию =0.01)  
//     itermaxs : Максимальное число итераций – эпох (по умолчанию – 100) 

//     initfunc : Функция инициализации значений w и b: 'rand', 'zeros', или 'ones' 
//                   По умолчанию используется ‘rand’.  
//     w : веса сети (SxR)) 

//     b : смещения (Sx1) 
 
// 1.===== Обработка списка входных аргументов функции ====== 

rhs=argn(2); // argn(2) – возвращает число аргументов  функции  
if rhs < 2; error("Должно быть минимум 2 аргумента: P и T"); end     
if rhs < 3; alpha = 0.01; end   

if rhs < 4; itermax = 100; end 
if rhs < 5; w = rand(size(T,1),size(P,1)); b = rand(size(T,1),1); end 
//Инициализация весов и смещений  

if  rhs == 5 then 
    select initfunc 
    case 'rand' then  

        w = rand(size(T,1),size(P,1));  
        b = rand(size(T,1),1); 
    case 'zeros' then 

        w = zeros(size(T,1),size(P,1));  
        b = zeros(size(T,1),1); 
    case 'ones' then  

        w = ones(size(T,1),size(P,1));  
        b = ones(size(T,1),1);         
    else  

        error("Неверное значение входного аргумента 5”);       
    end 
end 

 
if itermax == []; itermax = 100; end 
if alpha == []; alpha = 0.01; end     

 
//2. ====== Реализация правил обучения ========================== 
itercnt = 0; // Счетчик итераций - эпох 

while itercnt < itermax 
    for cnt = 1:size(P,2) // Цикл по всем обучающим примерам  - 1 эпоха 
        // Вычисляем вектор текущей ошибки сети e для одного примера P(:,cnt)  

        e = T(:,cnt) - ann_purelin_activ(w*P(:,cnt)+b);  
       //Реализуем правило обучения Уидроу-Хоффа 
        //для каждого примера  P(:,cnt)  вычисляем обновление  w и b   

        w = (w + 2*alpha*e*P(:,cnt)'); 
        b = b + 2*alpha*e; 
        // Вычисляем и запоминаем квадраты текущих  ошибок  

        e_all(cnt) = e.^2; 
    end 
    itercnt = itercnt + 1; 

    disp('Epoch: ' + string(itercnt) + '  MSE: ' + string(mean(e_all))); 
end 
endfunction 

 

Приложение В. Код функции ann_ADALINE_predict 

function [w, b, y, ee]=ann_ADALINE_predict(X, T, alpha, itermax, D, initfunc) 
// Обучение адаптивного линейного предсказателя с линией задержки  
 // в последовательном режиме  

// Примеры вызовов  
//     [w,b] = ann_ADALINE__predict (P,T, alpha) 
//     [w,b] = ann_ADALINE__predict (P,T,alpha,itermax, D, initfunc)  

//Параметры:  
//     X: входные значения  предсказателя  
//     P : Матрица обучающих примеров (RxQ) 

//     T : Матрица целевых выходных значений (SxQ) 
//     alpha: Скорость обучения (по умолчанию – 0.01) 
//     itermaxs : Максимальное число итераций – эпох (по умолчанию – 100) 

//     D – число элементов задержки (по умолчанию – 1) 
//     initfunc : Функция инициализации значений w и b: 'rand', 'zeros', или 'ones' 
//                   По умолчанию используется ‘rand’.  

//     w : веса сети (SxR)) 
//     b : смещения (Sx1) 
//      y : Выход предсказанияt 

//      ee : Ошибка между  T и  y  
 
// 1.===== Обработка списка входных аргументов функции ====== 

rhs=argn(2); // argn(2) – возвращает число аргументов  функции  
if rhs < 2; error("Должно быть минимум 2 аргумента: P и T"); end     
if rhs < 3; alpha = 0.01; end 

if rhs < 4; itermax = 100; end 
if rhs < 5; D = 1; end     
if rhs < 6; w = rand(size(T,1),D); b = rand(size(T,1),1); end 

 
//Инициализация весов и смещений  
if  rhs == 6 then 

    select initfunc 
    case 'rand' then  
        w = rand(size(T,1),D);  

        b = rand(size(T,1),1); 
    case 'zeros' then 
        w = zeros(size(T,1),D);  

        b = zeros(size(T,1),1); 
    case 'ones' then  
        w = ones(size(T,1),D);  

        b = ones(size(T,1),1);         
    else  
        error("Неверное значение входного аргумента 5”);       

    end 
end 
 

if itermax == []; itermax = 100; end 
if alpha == []; alpha = 0.01; end     
if D == []; D = 1; end   

 
// Формирование выходов линии задержки: реформатирование X в матрицу P 
P = []; 

for cnt = 1:D // для каждого выхода линии задержки  
    //формируем строки  матрицы P из отсчетов X 
    // очередная строка P – сдвинутая на один отсчет копия предыдущей строки  

    P = [P; X(cnt:$-D+cnt-1)];  

end 
//формируем вектор целевых значений  
T = T(1:$-D+1);  

 
//2. ====== Реализация правил обучения ========================== 
itercnt = 0; // Счетчик итераций - эпох 

while itercnt < itermax 
    for cnt = 1:size(P,2)  // Цикл по всем обучающим примерам из P (1 эпоха) 
        n = w*P(:,cnt)+b;  // Сетевая функция АЛЭ  

        a = ann_purelin_activ(n);  
        e = T(:,cnt) - a;    // Ошибка 
        y(cnt) = a;           // Запоминаем выход АЛЭ  

        ee(cnt) = e;         // Запоминаем ошибку    
        //Реализуем правило обучения Уидроу-Хоффа 
        //для каждого примера  P(:,cnt)  вычисляем обновление  w и b 

        w = (w + 2*alpha*e*P(:,cnt)'); 
        b = b + 2*alpha*e; 
        // Вычисляем и запоминаем квадраты текущих  ошибок  

        e_all(cnt) = e.^2; 
    end 
    itercnt = itercnt + 1; 

    disp('Epoch: ' + string(itercnt) + '  MSE: ' + string(mean(e_all))); 
 
end 

 
endfunction 
 
 

Приложение Г. Код функции ann_ADALINE_run 
 
function y=ann_ADALINE_run(P, w, b) 

// Функция моделирования адаптивного линейного элемента.  
// Вызов: 
//     y = ann_ADALINE_run(P,w,b)  

// 
// Параметры:  
//     P : Матрица  входных тестовых примеров (RxQ) 

//     w : веса слоя (SxR) 
//     b : смещения слоя(Sx1) 
//     y : выход слоя (SxQ) 

y = ann_purelin_activ(w*P+repmat(b,1,size(P,2)));  
 
endfunction 

Лабораторная работа №5 

 

Исследование многослойного персептрона:  
алгоритмы обратного распространения 

 с адаптивной скоростью обучения и моментом 

1 Цель работы 

Углубление теоретических знаний в области архитектуры многослойных 
нейронных сетей прямого распространения, исследование свойств алгоритмов 
обучения многослойных нейронных сетей, приобретение практических навыков 

обучения многослойного персептрона при решении задач классификации и ап-
проксимации функций. 

2 Основные теоретические положения 

2.1 Структура многослойного персептрона и его возможности 

Многослойный персептрон (MLP – multi-layer perceptron) состоит из не-
скольких слоёв. В качестве примера на рисунке 5.1 изображена структурная 

схема 3-х слойного персептрона. Структуру сети можно описывать сокращен-
но: R-S

1
-S

2
-S

3
, где R – число входов персептрона, S

1
-S

2
-S

3
 -число нейронов в ка-

ждом слое. 
 

 

Рисунок 5.1 – Структура 3-х слойного персептрона 

 
В MLP выход одного слоя является входом следующего слоя: 
 

           (5.1) 
 

где m – номер слоя, M – число слоев MLP. На вход нейронов первого слоя по-
дается внешний вектор p: 

. 



66 
 

 

Выход последнего слоя является выходом MLP: 

 
 

Для моделирования работы МLP в модуле NeuralNetworks 2.0 пакета 

Scilab имеется функции ann_FFBP_run, код которой с комментариями приведен 
в приложении А. Функция ann_FFBP_run(P,W,af) принимает на вход матрицу P, 
столбцы которой представляют собой входные векторы p, расширенную мат-

рицу весов и смещений всех слоев W, список af имен активационных функций 
каждого слоя MLP и реализует вычисления выходного вектора a в полном со-

ответствии с формулой (5.1).  
В отличие от однослойного персептрона MLP способен решать задачи 

классификации, которые не ограничиваются случаем линейной сепарабельно-
сти. Например, MLP успешно решает задачу исключающего ИЛИ (рисунок 5.2). 

Одно из решений – использовать 2 нейрона в первом слое, которые создают две 
границы. Первая отделяет p1 от остальных образов, а вторая отделяет p4. Второй 

слой комбинирует эти две границы, используя AND операцию 
 

 
 

Рисунок 5.2 – Структура MLP для решения задачи исключающего ИЛИ 
 

МLP также широко используют для аппроксимации функций. Можно по-
казать, что 2-х слойная сеть со скрытым сигмоидальным слоем и линейным вы-
ходным слоем способна аппроксимировать любую непрерывную функцию с за-

данной точностью при надлежащем числе нейронов скрытого слоя (теорема 
универсальной аппроксимации [3, 4]). Отметим, что теорема универсальной ап-

проксимации ничего не говорит об оптимальности МLP с одним скрытым сло-
ем для решения задачи аппроксимации в отношении времени обучения, просто-

ты применения, обобщающих способностей обученного MLP.  
 

2.2 Целевая функция и BP-алгоритм обучения MLP 
 

Обучение MLP осуществляется на основе алгоритма обратного распро-
странения (BP – back  propagation). Алгоритм BP является развитием алго-

ритма LMS. Оба алгоритма минимизируют целевую функцию в виде среднего 
квадрата ошибки (СКО, MSE). Для MLP со многими выходами СКО будет ра-

вен:  
 

                                (5.2) 

Как и  LMS, алгоритм BP аппроксимирует СКО квадратами ошибки на ка-
ждом шаге k  
 

 .                      (5.3) 
 

Тогда в соответствии с алгоритмом наискорейшего спуска SDA (4.9) алгоритм 

обучения MLP запишется в виде 

                                     (5.4) 
 

С учетом того, что целевая функция F в этих формулах является стохастиче-
ской, то они подобны формулам алгоритма LMS. Сложность заключается в 

вычислении частных производных. Если ввести обозначение 
 

                                                       (5.5) 

где   ─ чувствительность целевой функции F к изменению n
m

i ─ i-го сетево-
го входа в слое m, то можно показать, что выражения для частных производных 

в формуле (5.4) запишутся в виде 
 

                                        (5.6) 
Соответственно стохастическая аппроксимация SDA (5.4) для MLP запишется в 

виде 
 

                   (5.7) 
 

Или векторно-матричной форме 
 

                                     (5.8) 
 

Чувствительности  также вычисляются с помощью рекуррентных соот-

ношений. При этом чувствительность слоя m определяется через чувствитель-
ность слоя m+1, т.е. в обратном направлении (backpropagation). Чтобы пока-

зать это, запишем рекуррентное соотношение для вычисления чувствительно-
стей, используя цепочное правило дифференцирования сложной функции 

 

                           (5.9).  

Здесь матрица частных производных от векторной функции n
m+1

 по n
m 

, назы-
ваемая якобианом (производная от векторной функции), равна 
 

                                            (5.10) 

где  ─ диагональная матрица с производными   на главной диаго-
нали. 

Таким образом, значения чувствительностей слоев распространяются 
в обратном направлении – от последнего слоя к первому. Значение чувстви-

тельности выходного слоя 
 

                  (5.11) 

 

Объединяя формулы, приведенные выше, получим алгоритм обучения MLP, 

называемый алгоритмом обратного распространения  или  алгоритмом BP 
(рисунок 5.3).  

 
Алгоритм BP 

 

1. Инициализировать веса и смещения небольшими случайными значе-
ниями, задать допустимое значение ошибки Eдоп . 

2. Распространить очередной входной вектор по сети в прямом направле-
нии: формула (5.1). 

3. Распространить чувствительности в обратном направлении, выполнив 
вычисления по формулам: 

 

                   (5.12) 
 

4. Обновить веса и смещения в соответствии с формулой (5.8). 
5. Проверить значение ошибки, если ошибка на выходе сети больше Eдоп , 

то перейти к п.2., иначе “стоп”.    
 

Рисунок 5.3 – Алгоритм обратного распространения 
 

2.3 Последовательный и блочный алгоритмы BP 
 

Рассмотренный алгоритм BP является последовательным, т.к. предпола-
гает обновление параметров сети после предъявления каждого очередного 
входного вектора. Возможен альтернативный блочный вариант алгоритма, ко-

гда полный градиент вычисляется до обновления весов и смещений (т.е. после 
подачи на вход сети всех входных векторов p). Например, если все входные 

векторы равновероятны, то CKO будет равен 

                (5.13) 
Полный градиент этой целевой функции будет равен среднему градиенту от-
дельных квадратов ошибок: 
 

          (5.14) 
 

Для реализации блочного алгоритма BP выполняются шаги 1-3 алгоритма BP 
для всех входных векторов. Затем индивидуальные градиенты усредняются, 

чтобы получить полный градиент, и параметры сети обновляются на основе 
формул, содержащих средние обновления: 

     (5.15) 

Так как рассмотренные алгоритмы BP представляют собой аппроксимацию 
алгоритма градиентного спуска (gradient descent), то будем обозначать  их как 

BP_GD. В модуле NeuralNetworks 2.0 пакета Scilab имеется функция 
ann_FFBP_gd, реализующая блочный вариант алгоритма BP_GD. Текст функ-

ции с комментариями приведен в приложении Б. В общем случае формат вызо-
ва функции требует задания следующих параметров:  

 

W = ann_FFBP_gd(P,T,N,af,lr,itermax,mse_min,gd_min) 

 

где  
P- матрица обучающих примеров входных данных; 

T – матрица обучающих целевых значений;  
N – вектор, задающий число нейронов каждого слоя, включая входной и 

выходной слои. Например, [2 30 1] – для сети с 2-мя входами, скрытым 
слоем из 30 нейронов и выходным слоем с 1-м нейроном; 

af – список  имен активационных функций слоев, например 
['ann_tansig_activ','ann_purelin_activ'] для сети сo структурой N=[2 30 1]. Чис-

ло элементов списка функций должно быть равно числу слоёв без учета 
входного слоя;  

lr – скорость обучения (параметр α в формуле 5.8), по умолчанию 0,01; 
itermax – максимальное число эпох обучения, по умолчанию 1000; 
mse_min – минимальное допустимое значение ошибки Eдоп (значение 

СКО-целевой функции), по умолчанию 1e-5; 
gd_min – минимальное значение градиента целевой функции, по умолча-

нию 1e-5.  
В отличие от алгоритма BP_GD, описанного выше, функция 

ann_FFBP_gd прекращает процесс обучения при выполнении одного из 3-
условий: 
 

mse < mse_min, itercnt > itermax , gd < gd_min, 

где mse, itercnt, gd – соответственно значения СКО, счетчика эпох и значения 
градиента целевой функции на каждой итерации алгоритма. Обратим внимание 

на то, что по мере приближения к минимуму целевой функции значения mse 
должны снижаться, а значения gd должны быть близкими к нулю. 

 
2.4 Эвристические разновидности алгоритмов BP 

 
Основной недостаток алгоритмов BP_GD – медленное схождение. Методы 

ускорения этих алгоритмов могут быть разделены на 2 группы: 
1) эвристические методы: 

- методы, основанные на изменении скорости обучения; 
- методы, основанные на использовании моментов инерции; 

2) методы, использующие алгоритмы оптимизации. 

Рассмотрим эвристические методы. Алгоритмы, относящиеся к этой 
группе, используют процедуру BP_GD в качестве базовой. Как отмечалось вы-

ше, BP_GD эквивалентен LMS, если сеть однослойная. Однако BP_GD, приме-
няемый для обучения MLP, обладает совсем другими свойствами. Это связано с 

отличием поверхностей целевых функций для однослойной и многослойной 
сетей. Если поверхность целевой функции однослойной сети имеет одну ста-

ционарную точку и постоянную кривизну, то поверхность целевой функции для 
MLP может иметь много локальных минимумов и кривизна её может суще-

ственно изменяться в различных областях пространства параметров. 
Это говорит о том, что необходимо изменять скорость обучения в про-

цессе обучения. Например, на участках, где целевая функция уменьшается 
можно скорость обучения повышать, на плоских участках – оставлять неизмен-
ной, а на участках с растущими значениями целевой функции скорость обуче-

ния следует снижать.  
Другой путь улучшения сходимости – сглаживание траектории изменения 

параметров сети. Когда алгоритм расходится, он начинает осциллировать 
(обычно поперек некоторой длинной узкой ложбины целевой функции). Если 

выполнить фильтрацию (сглаживание) траектории спуска путем усреднения 
обновления параметров, то получим более устойчивую траекторию спуска. 

Для преодоления проблемы нескольких минимумов, следует запус-
кать алгоритм с разными начальными условиями. При этом необходимо 

контролировать минимум достигнутой ошибки и выбирать те параметры сети, 
которые обеспечивают минимальный уровень CKO. 

В соответствии с указанными подходами сформулируем две эвристиче-
ские разновидности алгоритма BP для многослойной сети прямого распростра-

нения. 
2.4.1 Алгоритм BP c моментом инерции 

 

Как указывалось, схождение можно улучшить, если сглаживать траекто-
рию спуска. Это можно сделать с помощью фильтра нижних частот. Рассмот-

рим усредняющий фильтр 1-го порядка: 

                                (5.16) 
 

где w(k) – вход фильтра, y(k) – выход фильтра, γ – коэффициент момента 

(0≤γ<1). Поскольку фильтр, в общем, суммирует очередные входные значения 
w(k) с предыдущими своими выходами y(k-1), то на выходе фильтра будут фор-

мироваться значения y(k), соответствующие некоторому среднему значению 
w(k). Иными словами, фильтр добавляет в динамику обновления w(k) некото-
рый момент инерции. Чем больше γ, тем более инерционен фильтр. При γ=0 

выход фильтра y(k)= w(k) и фильтр не будет вносить никакой инерции в дина-
мику w(k). 

В соответствии с алгоритмом BP обновление параметров определяется 
выражениями 

 
 

При введении в эти уравнения момента инерции в соответствии с (5.16) полу-

чим алгоритм BP_GD с моментом инерции, который будем обозначать как 
BP_GDM 
 

                        (5.17) 
 

BP_GDM делает траекторию спуска более стабильной при большей скорости 

обучения по сравнению с алгоритмом BP_GD. 
В модуле NeuralNetworks 2.0 пакета Scilab имеется функция ann_FFBP_gdm, 

реализующая этот алгоритм. В общем случае формат вызова функции требует 

задания следующих параметров:  
 

W = ann_FFBP_gdm(P,T,N,af,lr,Mr,itermax,mse_min,gd_min),  

 

где Mr – коэффициент момента (γ), по умолчанию 0,9. Остальные параметры 
функции аналогичны параметрам функции ann_FFBP_gd. 

 
2.4.2 Алгоритм BP c адаптивной скоростью обучения 

 
Как отмечалось, скорость обучения из-за изменения кривизны поверхно-

сти целевой функции MLP следует изменять в процессе обучения. Сформули-

руем правила управления скоростью обучения α(k) для блочного алгоритма BP:  
1. Если в ходе обучения квадрат ошибки (по всему обучающему множе-

ству) увеличивается более, чем на ξ процентов ( 1-5%) после обновления весов, 
то обновление весов следует отменить, скорость обучения уменьшить, умножив 

её на значение 0<ρ<1, а коэффициент момента γ (если используется) обнулить. 
2. Если квадрат ошибки уменьшается после обновления весов, то обнов-

ление весов принимается, скорость обучения умножается η>1; если γ было об-
нулено, то его значение восстанавливается. 

3. Если квадрат ошибки увеличивается менее чем на ξ процентов, то обнов-
ление весов принимается, скорость обучения не изменяется; если γ было обну-

лено, то его значение восстанавливается.  
В модуле NeuralNetworks 2.0 пакета Scilab имеется функция ann_FFBP_gda, 

реализующая этот алгоритм. В общем случае формат вызова функции требует 

задания следующих параметров:  
 
W = ann_FFBP_gda(P,T,N,af,lr,lr_inc,lr_dec,itermax,mse_min,gd_min,mse_diff_max) , 

 

где lr_inc – коэффициент увеличения скорости обучения (η), по умолчанию 1,05; 
lr_dec – коэффициент уменьшения скорости обучения (ρ), по умолчанию 0,75; 

mse_diff_max – допустимое относительное увеличение CKO (ξ), по умолчанию 
0,01. Остальные параметры функции аналогичны параметрам функции 

ann_FFBP_gd. 
Также в составе модуля NeuralNetworks 2.0 пакета Scilab имеется функ-

ция ann_FFBP_gdх, которая комбинирует алгоритм BP c моментом и алгоритм 
BP с адаптивной скоростью обучения. Функция имеет расширенный список па-
раметров: 

 
(P,T,N,af,lr,lr_inc,lr_dec,Mr,itermax,mse_min,gd_min,mse_diff_max). 

 
Интерпретация этих параметров была рассмотрена выше. 

Эвристические алгоритмы обеспечивают более быструю сходимость для 
некоторых задач. Однако такие алгоритмы требуют установки значений допол-

нительных параметров, тогда как BP_GD требует задания только скорости обу-
чения. Часто эвристические алгоритмы оказываются чувствительными к изме-

нениям этих параметров. Выбор значений дополнительных параметров также 
зависит от задачи. 

 

3. Варианты заданий и программа работы 
 

3.1. Повторить теоретический материал, относящийся к архитектуре и алго-
ритмам обучения многослойного персептрона [1, 4, 5]. 

3.2. Для варианта из п. 4.3 лабораторной работы №1, где задана одномерная 
функция у=f(x) на некотором интервале определения: 

3.2.1.  Сформировать два множества (класса) точек данных (не менее 
200 точек в каждом множестве), которые располагаются выше и 
ниже кривой у=f(x), и отстоят от неё на расстояние d= 

                  ; 
3.2.2. Отобразить классы и кривую у=f(x) на двумерной плоскости; 
3.2.3. Разработать программу обучения многослойного персептрона с 

архитектурой R-S-1 для классификации этих классов, в качестве 
активационных функций скрытого слоя использовать функции 

ann_tansig_active или ann_logsig_active, а в качестве активацион-

ной функции выходного слоя ─ ann_purelin_active, обучение пер-
септрона выполнять с использованием функции ann_FFBP_gd;  

3.2.4. Выполнить предварительное обучение MLP с архитектурой [R-10-
1] на небольшом числе эпох itermax =200 при разных значениях 
параметра скорости обучения lr с целью определения её квазиоп-

тимального устойчивого значения.  
3.2.5. Используя полученное значение скорости обучения lr, выполнить 

обучение MLP с архитектурой [R-S-1] при разных S (10,30,40,50) 
на большом числе эпох itermax =1000. 

3.2.6. Для различных MLP, обученных в соответствии с п.3.2.5, выпол-
нить моделирование MLP и проверить качество классификации на 

тестовом множестве данных. Для этого сформировать тестовое 
множество данных аналогично п. 3.2.1, провести классификацию 

данных с помощью обученного MLP, отобразить точки получен-
ных выходных классов и кривую у=f(x) на двумерной плоскости, 

вычислить вероятности правильной классификации при разных S.  
3.3. Используя вариант из п. 4.5 лабораторной работы №1, где задана двумер-

ная функция z=f(x,y) на некотором интервале определения: 
3.3.1. Сформировать подмножества обучающих и тестовых  данных. Для 

этого выбрать на плоскости (x,y) 300 случайных точек и опреде-

лить в этих точках значение функции z=f(x,y). В качестве входного 
вектора использовать вектор p=[x;y], в качестве целевого значения 

– t=z. Полученные данные разделить на 2 подмножества: обучаю-
щее (80%) и тестовое (20%). 

3.3.2. Разработать программу обучения многослойного персептрона с 
архитектурой R-S-1для аппроксимации этой функции, в качестве 

активационных функций скрытого слоя использовать функции 
ann_tansig_active или ann_logsig_active, а в качестве активацион-

ной функции выходного слоя -- ann_purelin_active, обучение пер-
септрона выполнять с использованием функции ann_FFBP_gd; 

3.3.3. Построить кривые обучения MLP при разных значениях  S (5 10 15 
20) и фиксированной скорости обучения lr (например, lr =0.005);  
выбрать квазиоптимальное значение S для дальнейшего использо-

вания. Построение кривых выполнить при значении параметра 
itermax=300; 

3.3.4. Выполнить  обучение MLP (при квазиоптимальном S) c помощью 
функций ann_FFBP_gdm, ann_FFBP_gdа, ann_FFBP_gdх, сравнить 

получаемые кривые обучения c кривыми, полученными в п. 3.3.3; 
3.3.5. Используя тестовое подмножество данных, выполнить моделиро-

вание 4-х вариантов MLP, обученных с помощью 4-х разных 
функций, указанных выше. Построить сопоставительные графики 

значений функции z=f(x,y) и соответствующих значений на выходе 
MLP, вычислить СКО аппроксимации функции на тестовом под-

множестве, сравнить со значениями СКО, полученными при обу-
чении MLP.  

3.4. Подготовьте и защитите отчет по работе. 
 

4. Методические рекомендации по выполнению работы 
 

4.1. В задании 3.2. исследуется задача бинарной классификации. Цель задания  

– убедиться в том, что MLP способен строить нелинейные границы решения 
между классами.  

Для этого в соответствии с п.3.2.1 задания необходимо сначала сформи-
ровать два множества (класса) случайных точек, разделяемых заданной по ва-

рианту нелинейной функцией. Ниже приведен пример кода функции для фор-
мирования случайных точек, принадлежащих 2-м классам, разделяемых кривой 

y=f(x). Точки классов отстоят от разделяющей кривой на расстояние 
d=kd*abs(ymax-ymin), где ymax,ymin – максимальное и минимальное значения 
функции y=f(x) на заданном интервале Фрагменты кода, отмеченные красным 

цветом, требуют уточнения. Модернизируйте этот код для формирования мно-
жества точек 2-х классов в соответствии с заданным вариантом.  

 
function [P, T, y, x]=gendata(Q, kd) 

//формирование множеств случайных точек для 2-классов, 
// разделяемых кривой y=f(x) 
//Q - число точек в каждом множестве 

//kd - коэффициент, задающий величину отступа от границы 
 

//задание области определения функции - координата x 
xmin=0.7; 
xmax=4; 

//поиск ymax и ymin 
stepx=abs(xmax-xmin)/Q; 

x=xmin:stepx:xmax; 
y=f(x);  // функция, заданная вариантом. Определяется отдельно.! 
ymin=min(y); 

ymax=max(y); 

//диапазон изменения значений функции  

range=abs(ymax-ymin); 
//задание расстояния между классами: d=kd*|(ymax-ymin)| 
d=kd*range; 

 
//формирование случ. мн-ва точек datax вдоль x 
datax=grand(1,Q,'unf',xmin,xmax); 

 
// формирование обучающего множества {P,T} 
  P=[]; 
  T=[]; 

  for j=1:1:Q 

      //значение границы border для случайной точки datax(1,j) 
      border=f(datax(1,j)); 

      //вычисление ординаты "y" 2-х случайных точек, отстоящих от border на рас-
стояние d 

      y_class1=grand(1,1,'unf',border+d,border+range); 
      y_class2=grand(1,1,'unf',border-range,border-d); 

      //формирование x,y координат 2-х j-тых точек классов 1 и 2 в виде вектора  
      j_class1=[datax(1,j); y_class1]; 
      j_class2=[datax(1,j); y_class2]; 

      //добавление полученных j-тых точек во множество входных примеров 
      P=[P j_class1 j_class2]; 

      // формирование соответсвующих целевых выходных значений MLP  
      t_class1=1; 
      t_class2=0; 
      T=[T t_class1  t_class2]; 

  end 

endfunction 
 

Для отображения точек 2-х классов (задание п.3.2.2) и разделяющей гра-

ницы (рисунок 5.4) можно использовать следующий код:  
 

  //отображение множества точек 2-х классов и разделяющей границы 
  drawlater(); // прорисовка отображения будет позже 

  plot(P(1,T==1),P(2,T==1),'o'); //отображать точки 1-го класса знаком 'o' 
  plot(P(1,T==0),P(2,T==0),'g*'); //отображать точки 2-го класса знаком '*' 
  plot(x,y,'r'); // отображать границу красным цветом 

  drawnow(); // включаем прорисовку 
  xtitle('Классы и граница'); 

 
Рисунок 5.4 – Сгенерированные классы и нелинейная граница между классами 

 
Выполнение п. 3.2.4 заключается в предварительном обучении MLP с ис-

пользованием функции ann_FFBP_gd. При каждом запуске функция будет ото-
бражать прогресс обучения в отдельном графическом окне (рисунок 5.5) 

Рисунок 5.5   Окно прогресса, отображаемое функцией ann_FFBP_gd 

 
 Для сравнения результатов обучения при разных значениях скорости 
обучения lr контролируйте значение среднего квадрата ошибки MSE. Рекомен-

дуется значение скорости обучения выбирать из списка lr=[0.05, 0.025, 0.01, 

0.005] . В качестве квазиоптимального значения lr следует выбрать то значение, 

которое обеспечивает минимальное конечное значение MSE в окне прогресса 
(рисунок 5.5). 

 При выполнении п.п. 3.2.5-3.2.6 следует обучать MLP с разным числом 
нейронов в скрытом слое и тестировать результаты обучения. Обучение необ-

ходимо выполнять при itermax =1000 и квазиоптимальной скорости обучения, 
полученной ранее. Для тестирования результатов обучения следует использо-

вать функцию ann_FFBP_run и множество данных, на котором MLP не обу-
чался. Поэтому необходимо тестовое множество входных данных сформиро-

вать с помощью функции gendata(Q, kd): 
 

[P_test,T_test,y_test,x_test]=gendata(Q,kd); 

 
Чтобы вычислить вероятность правильной классификации необходимо сначала 

выходные значения MLP привести к бинарным уровням 0 и 1. Для этого следу-
ет их округлить, а затем посчитать вероятность правильной классификации, как 

отношение числа случаев правильной классификации к числу точек данных: 
 
 a_test = ann_FFBP_run(P_test,W,af); 
 y_bin_test=round(a_test); //делаем выход бинарным 
 //вычисляем вероятность правильной классификации 

 n_correct=sum(T_test == y_bin_test); 
 rate=n_correct/length(T_test); 

 
Для отображения результатов классификации обученным MLP используйте 

код, аналогичный коду, который применялся для построения рисунка 5.4, заме-
нив целевой вектор T, на вектор y_bin_test с результатами бинарной классифи-
кации: 

 
  plot(P(1, y_bin_test ==1),P(2, y_bin_test ==1),'o'); //отображать точки 1-го класса' 

  plot(P(1, y_bin_test ==0),P(2, y_bin_test ==0),'g*'); //отображать точки 2-го класса' 

  plot(x_test,y_test,'m'); // отображать границу малиновым цветом 

 

Пример отображения выходных классов, формируемых обученным MLP, при-
веден на рисунке 5.6. Обратите внимание, что ряд точек MLP классифицировал 

неправильно. Вероятность правильной классификации при S=10 для рассматри-
ваемого примера равна 0,945. Необходимо оценить вероятности правильной 

классификации и при других S, указанных в задании. 
 

 
Рисунок 5.6 – Результаты классификации тестирующего множества,  

вероятность правильной классификации 0,945 (S=10) 

 
4.2. В задании 3.3. исследуется задача аппроксимации двумерной функции с 

помощью MLP. Цель задания – убедиться в том, что MLP способен обучаться 
аппроксимации нелинейных функций на основе множества обучающих данных.  

Для обучения MLP в соответствии с п.3.3.1 задания требуется сформиро-
вать подмножества обучающих и тестовых данных. Тестовые данные исполь-

зуются только в ходе тестирования обученной нейросети с целью выявления 
качества её функционирования и обобщающих способностей. Ниже приведен 
фрагмент кода для генерации обучающих P_train, T_train и тестовых P_test, 

T_test подмножеств данных.  
 

Q=300; //Общее число элементов множества данных 
Q_train=floor(0.8*Q); //число эл-тов обучающего мно-ва 80% от Q 

 
x=grand(1,Q,'unf',minx,maxx); //формирование случайных координат x,y 
y=grand(1,Q,'unf',miny,maxy); 

 
x_train=x(1:Q_train);  //формирование обучающего подмно-ва 

y_train=y(1:Q_train); 
P_train=[x_train;y_train]; 
T_train=f(x_train,y_train); 

 
x_test=x(Q_train+1:$);//формирование тестового подмно-ва 

y_test=y(Q_train+1:$); 

P_test=[x_test;y_test]; 
T_test=f(x_test,y_test); 

 

В задании п. 3.3.3 требуется обучить MLP на основе алгоритма BP_gd при 
разных значениях S и построить кривые обучения. Построение кривых обуче-

ния потребует модификации кода встроенной функции ann_FFBP_gd. Необхо-
димо, чтобы функция возвращала вектор значений СКО (в программе обозна-

чен out_mse). Выполните модернизацию функции по следующей схеме:  
 

function [W, out_mse]=ann_FFBP_gd1(P, T, N, af, lr, itermax, mse_min, gd_min) 

… 
// Stopping Criteria 

        mse = mean(e.^2); 
        itercnt = itercnt + 1; 

        out_mse(itercnt)=mse; // добавленная строка программы 
        gd = mean(s(1).^2); 

… 

endfunction 

 

Изменения, внесенные в функцию, отмечены курсивом. Для запоминания зна-
чений CKO при разных S организуйте циклический вызов модифицированной 

функции ann_FFBP_gd1: 
 

af=['ann_tansig_activ','ann_purelin_activ']; 
lr=0.005 
itermax=300 

S=[5, 10,15,20]; 
for i=1:length(S) 

      N=[2 S(i) 1]; 
      [W_gd, out_mse_gd] = ann_FFBP_gd1(P_train,T_train,N,af,lr,itermax); 
      mse_gd_hist(i,:)=out_mse_gd; 

end; 

 

Векторы значений СКО будут запоминаться в строках матрицы mse_gd_hist. 

Это позволит после завершения цикла построить кривые обучения для каждого 

S (рисунок 5.7). Из анализа кривых обучения (для примера на рисунке 5.7) сле-
дует, что квазиоптимальным будет значение S=15, т.к обеспечивает наиболее 
быстрое снижение СКО и наименьшее конечное значение СКО в ходе обучения 

(контролируется по значениям матрицы mse_gd_hist).  
Для сравнения рассмотренных в п.2.4 эвристических алгоритмов обуче-

ния с алгоритмом BP_gd (п. 3.3.4 задания) следует обучить MLP одной и той же 
структуры, используя функции обучения, указанные ниже: 

 
[W_gd, out_mse_gd] = ann_FFBP_gd1(P_train,T_train,N,af,lr,itermax,mse_min,gd_min);  

[W_gda,out_mse_gda] = 
ann_FFBP_gda1(P_train,T_train,N,af,lr,lr_inc,lr_dec,itermax,mse_min,gd_min,mse_diff_max ); 
[W_gdm, out_mse_gdm]=ann_FFBP_gdm1(P_train,T_train,N,af,lr,Mr,itermax,mse_min,gd_min);  

[W_gdx, out_mse_gdx] = ann_FFBP_gdx1(P_train,T_train,N,af,lr,lr_inc,lr_dec,Mr,itermax,mse_min,gd_min);  

Код всех этих функций модифицируется аналогично функции ann_FFBP_gd1. 

Результаты обучения MLP разными алгоритмами запоминаются в матрицах ве-

сов и векторах ошибок (хранят значения СКО), возвращаемых функциями. За-
поминание весов обученных MLP необходимо для тестирования результатов 
обучения, а запоминание векторов значений СКО – для построения кривых 

обучения  (рисунок 5.7).  
 

 
Рисунок 5.7. Кривые обучения MLP  

 

Тестирование результатов обучения (п. 3.3.5 задания) выполняется исключи-
тельно на данных, которые не использовались для обучения. Это позволяет 

оценить обобщающие свойства полученных моделей нейросетей. Для тестиро-
вания набора полученных нейросетей используйте функцию ann_FFBP_run: 

 
y_test_gd = ann_FFBP_run(P_test,W_gd,af);    // сеть, обученная алг-м BP_gd 

y_test_gda = ann_FFBP_run(P_test,W_gda,af); // сеть, обученная алг-м BP_gda 
y_test_gdm = ann_FFBP_run(P_test,W_gdm,af); // сеть, обученная алг-м BP_gdm 
y_test_gdx = ann_FFBP_run(P_test,W_gdx,af); // сеть, обученная алг-м BP_gdx 

 
// вычисление СКО на тестовых данных 

mse_gd=((T_test-y_test_gd)*(T_test-y_test_gd)')/Q_test 
mse_gda=((T_test-y_test_gda)*(T_test-y_test_gda)')/Q_test 
mse_gdm=((T_test-y_test_gdm)*(T_test-y_test_gdm)')/Q_test 

mse_gdx=((T_test-y_test_gdx)*(T_test-y_test_gdx)')/Q_test 

 

По результатам тестирования необходимо построить сопоставительные графи-
ки функции z=f(x,y), определенной на тестовом множестве входных данных, и 

графики выходных сигналов MLP, обученных разными алгоритмами (рисунок 
5.8), а также сравнить вычисленные выше СКО аппроксимации функции на 
тестовом подмножестве со значениями СКО, полученными на этапе обучения 

соответствующих MLP. 

Рисунок 5.8 – Сопоставительные  графики тестовых значений функции z=f(x,y) 

и значений на выходе MLP, обученных разными функциями 
 

Из рисунка 5.8 следует, что алгоритмы BP_gd и BP_gdm дают неудовле-
творительные результаты на выбранном числе эпох обучения. Эти алгоритмы 

медленно сходятся и требуют значительно большего числа эпох обучения по 
сравнению с алгоритмами, основанными на управлении скоростью обучения.  

Наименьшее значение СКО получается при обучении MLP с помощью алго-
ритма BP_gdx. 

 
5. Содержание отчета 

 

5.1.  Цель работы. 
5.2.  Вариант задания.  

5.3.  Схема MLP, решающего задачу классификации 2-х классов; программа 
обучения MLP бинарной классификации с использованием функции 

ann_BP_gd; результаты предварительного обучения MLP при разных зна-
чениях параметра lr  и выбор квазиоптимального lr; результаты обучения 
MLP при разных S и большом числе эпох;  результаты тестировании обу-

ченного бинарного классификатора на основе MLP и вероятности пра-
вильной классификации при разных S.  

5.4.  Схема MLP для решения задачи аппроксимации функций; программа обу-
чения MLP;  кривые обучения MLP при разных значениях  S и  выбор ква-

зиоптимального S; кривые обучения MLP (при квазиоптимальном S) c по-
мощью функций ann_FFBP_gd, ann_FFBP_gdm, ann_FFBP_gdа, 

ann_FFBP_gdх; результаты тестирования MLP, обученных указанными 

функциями (сравнительные графики),  значения СКО на этапе обучения и 
на этапе тестирования.  

5.5.  Выводы по результатам исследований. 
 

6. Контрольные вопросы 

 
6.1. Нарисуйте схему многослойного персептрона и запишите  выражения для 

вычисления его выходных значений в векторно-матричной форме. 
6.2. Объясните алгоритм, положенный в основу функции ann_FFBP_run(P,W,af), 

и   назначение её параметров. 
6.3. Нарисуйте структуру MLP для решения задачи исключающего ИЛИ. 

6.4. Сформулируйте общий смысл теоремы об универсальной аппроксимации в 
отношении MLP. 

6.5. Запишите выражение целевой функции, используемой в алгоритме BP. 
6.6. Запишите общие выражения алгоритма наискорейшего спуска примени-

тельно к MLP. 
6.7. Что понимают под чувствительностью целевой функции MLP? 

6.8. Запишите выражения для производных целевой функции MLP с использо-
ванием чувствительности. 
6.9. Запишите векторно-матричные выражения стохастической аппроксимации 

SDA  для MLP. 
6.10. Запишите рекуррентное соотношение для вычисления чувствительностей, 

используя цепочное правило дифференцирования. 
6.11. Чему равен  якобиан применительно к MLP? 

6.12. Сформулируйте последовательный алгоритм BP. 
6.13 Сформулируйте и запишите выражения для блочного алгоритма BP. 

6.14. Объясните параметры функции ann_FFBP_gd, реализующей блочный ва-
риант алгоритма BP_GD.  

6.15. При каких условиях функция ann_FFBP_gd прекращает процесс обуче-
ния? 

6.16. Какой основной недостаток алгоритма BP_GD?  
6.17. На какие группы подразделяются эвристические методы улучшения 
BP_GD? Дайте их краткую характеристику. 

6.18.  Сформулируйте алгоритм BP c моментом инерции. 
6.19. Объясните параметры функции ann_FFBP_gdm. 

6.20. Сформулируйте алгоритм BP c адаптивной скоростью обучения. 
6.21. Объясните параметры функции ann_FFBP_gda. 

6.22. Объясните назначение и параметры функции ann_FFBP_gdх. 
6.23. Как формируются обучающее и тестовые подмножества данных? Зачем 

нужно тестовое подмножество? 
6.24. Почему отличаются СКО обучения и тестирования? Какая из ошибок 

обычно меньше?  

Приложение А. Функция ann_FFBP_run 
 
function y=ann_FFBP_run(P, W, af) 

// Функция моделирования сети прямого распространения 

// Пример вызова: 

//     [y] = ann_FFBP_run(W,P,af)  

// Параметры: 
//     P : Тестовые входные значенияt 

//     W : Веса и смещения  

//     af : Список активационных функций 

//     y : Выходные значения  

 
//1. ================Обработка списка аргументов функции========================  

rhs=argn(2); 

if rhs < 3; af = ['ann_tansig_activ','ann_purelin_activ']; end 

 

if size(W) ~= size(af,2) then 
    error('Numbers of activation functions must match numbers of layers (N-1)'); 

end 

 

//2.================= Фаза моделирования================================== 

// Определяем число слоев сети по размерам W 
layers = size(W); 

//Создаем списки для хранения сетевых значений и активностей слоев  

n = list(0); 

a = list(0); 

//Вычисляем сетевые значения 1-го слоя: n=Wp+b 
n(1) = W(1)(:,1:$-1)*P + repmat(W(1)(:,$),1,size(P,2)); // Можно временно хранить в n для экономии памяти 

// Оцениваем значения на выходе 1-го слоя: a=af(n), где af – нелинейность слоя 

a(1) = evstr(af(1)+'(n('+string(1)+'))'); 

// Повторяем вычисления для каждого следующего слоя 

for cnt = 2:layers 
    n(cnt) = W(cnt)(:,1:$-1)*a(cnt-1) + repmat(W(cnt)(:,$),1,size(P,2)); // Можно временно хранить в n для экономии памяти 

    a(cnt) = evstr(af(cnt)+'(n('+string(cnt)+'))'); 

end 

// Результирующие значения соответствуют выходам последнего слоя  

y = a($); 
endfunction 

 

Приложение Б. Функция ann_FFBP_gd 
 

function W=ann_FFBP_gd(P, T, N, af, lr, itermax, mse_min, gd_min) 

// Обучение на основе блочного  алгоритма нисходящего градиента с обратным распространением. 

// 
// Примеры вызова:  

//     W = ann_FFBP_gd(P,T,N)  

//     W = ann_FFBP_gd(P,T,N,af,lr,itermax,mse_min,gd_min) 

// 

// Параметры 
//     P : Обучающий вход  

//     T : Обучающие целевые значения 

//     N : Вектор, задающий число нейронов каждого слоя, включая входной и выходной слои  

//     af : Список имен активационных функций от 1-го до выходного слоев 

//     lr : скорость обучения 
//     itermax : Максимальное число эпох обучения  

//     mse_min : Минимальная ошибка  (Значение целевой функции)  

//     gd_min : Минимальное значение градиента 

//     W : Выходные веса и смещения 

 
//1. ================Обработка списка аргументов функции========================  

rhs=argn(2); 

 

// Проверка ошибки списка аргументов 

if rhs < 3; error("Expect at least 3 arguments, P, T and N");end 
// Выбор значений аргументов по умолчанию 

if rhs < 4; af = ['ann_tansig_activ','ann_purelin_activ']; end 

if rhs < 5; lr = 0.01; end 

if rhs < 6; itermax = 1000; end 

if rhs < 7; mse_min = 1e-5; end 
if rhs < 8; gd_min =  1e-5; end 

 

if af == []; af = ['ann_tansig_activ','ann_purelin_activ']; end 

if lr == []; lr = 0.01; end 

if itermax == []; itermax = 1000; end 
if mse_min == []; mse_min = 1e-5; end 

if gd_min == []; gd_min =  1e-5; end 

 

// Проверка ошибки списка активационных функций 

if size(N,2)-1~= size(af,2) then 
    error('Numbers of activation functions must match numbers of layers (N-1)'); 

end 

 

//2.=======================Инициализация====================  

 
// Инициализация сети и формирование списка имен производных активационных функций 

format(8); 

W = ann_ffbp_init(N); 

itercnt = 0; 

af_d = strsubst(af,'ann_','ann_d_'); 
mse = %inf; 

gd = %inf; 

 

//Инициализация GUI отображения прогресса обучения 

handles = ann_training_process(); 
handles.itermax.string = string(itermax); 

handles.msemin.string = string(mse_min); 

handles.gdmax.string = 'inf'; 

handles.gdmin.string = string(gd_min); 

 
// Задание числа слоев и списков промежуточных переменных  

layers = size(N,2)-1; // слои считаются от 1-го скрытого слоя до выходного слояr 

n = list(0); 

a = list(0); 

m = list(0); 
s = list(0); 

 

// 3. ===================Реализация цикла эпох обучения==========================  

while mse > mse_min & itercnt < itermax & gd > gd_min 

        // Прямое распространение – моделирование (аналогично функции ann_FFBP_run (формула 5.1)) 
        n(1) = W(1)(:,1:$-1)*P + repmat(W(1)(:,$),1,size(T,2));  

        a(1) = evstr(af(1)+'(n('+string(1)+'))'); 

        for cnt = 2:layers 

            n(cnt) = W(cnt)(:,1:$-1)*a(cnt-1) + repmat(W(cnt)(:,$),1,size(T,2));  

            a(cnt) = evstr(af(cnt)+'(n('+string(cnt)+'))'); 
             

        end 

        // Вычисление ошибки  

        e = T - a($); 

        // Обратное распространение значений чувствительности  
        m(layers) = evstr(af_d(layers)+'(a('+string(layers)+'))'); // Вычисление производных актив. функций последнего слоя 

        s(layers) = (-2*m(layers).*e); // Вычисление чувствительности выходного слоя  (формула 5.12) 

        for cnt = layers-1:-1:1  //Вычисляем производные актив. функций слоев и распространяем обратно чувствительности.  

            m(cnt) = evstr(af_d(cnt)+'(a('+string(cnt)+'))'); 

            s(cnt) = m(cnt).*(W(cnt+1)(:,1:$-1)'*s(cnt+1));   (формула 5.12) 
        end 

         

        // Обновление весов сети (формула 5.15)  

        W(1)(:,1:$-1) = W(1)(:,1:$-1) - (lr*s(1)*P')./size(P,2); // Обновление весов 1-го скрытого слоя 

        W(1)(:,$) = W(1)(:,$) - lr*mean(s(1),2); // Обновление смещений 1-го скрытого слоя 
        for cnt = 2:layers // Обновление весов и смещений следующих слоев  

            W(cnt)(:,1:$-1) = W(cnt)(:,1:$-1) - (lr*s(cnt)*a(cnt-1)')./size(P,2); 

            W(cnt)(:,$) = W(cnt)(:,$) - lr*mean(s(cnt),2);                 

        end 

         
       // Вычисление критериев остановки алгоритма 

        mse = mean(e.^2);  // СКО  

        itercnt = itercnt + 1; //Число эпох 

        gd = mean(s(1).^2); // Средний квадрат значения градиента целевой функции  

       // программирование GUI отображения прогресса обучения  

       if  itercnt == 1 then 

           mse_max = mse; 

           handles.msemax.string = string(mse_max); 

           gd_max = gd; 
           handles.gdmax.string = string(gd_max);            

           mse_span = log(mse) - log(mse_min); 

           iter_span = itermax; 

           gd_span =  log(gd) - log(gd_min); 

       end 
         

        // для версии выше Scilab 5.5  

        handles.iter.value = round((itercnt/iter_span)*100); 

        handles.mse.value = -(log(mse)-log(mse_max))/mse_span * 100;// round(((log(mse) - log(mse_min))/mse_span)*100); 

        handles.gd.value = -(log(gd)-log(gd_max))/gd_span * 100; //round(((log(gd) - log(gd_min))/gd_span)*100);  
         

        handles.itercurrent.string = string(itercnt); 

        handles.msecurrent.string = string(mse);      

        handles.gdcurrent.string = string(gd);       

 
end 

endfunction 

Лабораторная работа №6 
 

Исследование многослойного персептрона: 
 алгоритм Левенберга-Марквардта 

1 Цель работы 

Углубление теоретических знаний в области алгоритмов обучения много-
слойных нейронных сетей прямого распространения, исследование свойств ал-

горитмов обучения второго порядка, приобретение практических навыков обу-
чения многослойного персептрона при решении задач классификации и ап-

проксимации функций. 

2 Основные теоретические положения 

2.1 Метод Ньютона  
 

Метод Ньютона основан на поиске стационарной точки квадратичной аппрок-
симации целевой функции F(x): 
 

                        (6.1) 

Найдем градиент (6.1) по отношению к Δxk  и приравняем  его нулю 
 

                                                (6.2) 
Отсюда оптимальный шаг будет равен 

 

                                                     (6.3) 

Тогда правило обучения в соответствии с методом Ньютона запишется в виде 
 

,                                               (6.4) 
где Ak – матрица Гессе целевой функции (матрица вторых производных), 

           –  значение вектора градиента на k–ой итерации алгоритма . 

Достоинством метод Ньютона, является то, что он находит минимум 

квадратичной функции за один шаг. В общем случае метод Ньютона при дру-
гих видах F(x) не сходится за один шаг и не гарантирует схождение. Это зави-
сит от вида F(x) и начальных условий поиска. 

Метод Ньютона, как и SDA, полагается только на локальные свойства по-
верхности целевой функции (первая и вторая производные). Он «не знает» гло-

бальных свойств целевой функции. В действительности метод Ньютона может 
давать непредсказуемые результаты. В то время как SDA гарантирует схожде-

ние при малых значениях скорости обучения. Исключить расхождение метода 
Ньютона можно путем включения SDA шагов, когда происходит расхождение. 

Другим недостатком метода Ньютона является необходимость вычисления об-
ратной матрицы Гессе. 

Метод Ньютона совпадает с SDA когда гессиан – это единичная матрица. 
Это наблюдение приводит к квазиньютоновским алгоритмам. В этих методах 
обратную матрицу Гессе заменяют положительной определённой матрицей Hk, 

которая обновляется на каждом шаге без использования инверсии. Эти алго-
ритмы разрабатываются так, чтобы Hk сходилась к A

-1
 .    

 
2.2 Алгоритм Левенберга-Марквардта  (LM)  

 
2.2.1 Основной алгоритм 

 

LM алгоритм – разновидность метода Ньютона (6.4), для которого ЦФ 

F(x) представляется суммой квадратов: 

,                                    (6.5) 
где v(x) – векторная  функция параметров сети, N – число элементов v. 

Для этой целевой функции j- ый элемент градиента будет равен 

                            (6.6) 
Градиент (6.6) можно представить в матричной форме  

                                       (6.7) 
где 

                               (6.8) 
якобиан векторной функции v(x). 

Элементы матрицы Гессе и сама матрица будут равны: 

  ,              (6.9) 

                                    (6.10) 

где 

Если S(x) мало, то матрицу Гессе можно аппроксимировать 
 

                                           (6.11) 
Выполняя подстановки (6.7) и (6.11) в формулу метода Ньютона, получим фор-

мулы метода Гаусса-Ньютона 
 

 (6.12) 
Достоинство метода Гаусса-Ньютона: не требуется вычислять вторые 

производные. Проблема: матрица   

                                                     (6.13) 
может быть неинвертируемой. Эту проблему преодолевают введением аппрок-

симации матрицы Гессе: 

 
 

Отсюда получаем основной алгоритм Левенберга –Марквардта   
 

                    (6.14) 
LM алгоритм обладает полезным свойством: если μk увеличивается, то 

LM алгоритм трансформируется в SDA с малой скоростью обучения: 
 

 
Если μk →0, то LM алгоритм трансформируется в алгоритм Гаусса-
Ньютона. 

Алгоритм начинают выполнять при малом μk =0.01. Если шаг не приво-
дит к уменьшению целевой функции, то его повторяют с μk увеличенным в θ>1 

раз (например, θ=10). Если шаг дает уменьшение целевой функции, то μk 
уменьшают в θ раз. При уменьшении μk  алгоритм будет приближаться к алго-

ритму Гаусса-Ньютона, что обеспечит быстрое схождение. 
LM алгоритм можно применить для обучения MLP. Если целевые значе-

ния обучающей выборки равновероятны, то СКО будет пропорционален сумме 
квадратов ошибок и  

(6.15) 
 

где e j,q  j-ый элемент ошибки для q-ой обучающей пары. Таким образом, целе-
вая функция MLP будет эквивалентна той, которая использовалась при выводе 

LM алгоритма. 

2.2.2. Вычисление Якобиана для LM алгоритма 
 

Алгоритм (6.14) требует предварительного вычисления Якобиана. Для его 
вычисления необходимо вычислять производные ошибок. 
Вектор ошибок MLP применительно к LMA запишется в виде:  

 
Вектор параметров:  

. 

При этом 

. 
Для вычисления якобиана (6.8) необходимо вычислять элементы матрицы: 

                                       (6.16) 
Определим чувствительность Марквардта: 

                                      (6.17) 

где   

Теперь мы можем вычислить элементы якобиана. Для весов MLP 

(6.18) 
или, если параметр – это смещение, то 

        (6.19) 
Чувствительности Марквардта могут вычисляться в соответствии с рекуррент-
ными соотношениями для стандартных чувствительностей алгоритма BP, за 

исключением последнего слоя MLP. Для последнего слоя 

   (6.20). 

Следовательно, когда на вход MLP подан вектор pq, а на выходе сформирован 

вектор   LMA  инициализируется значениями 

,                                              (6.21) 

где  ─ диагональная матрица с производными на главной диагонали. 

Каждый столбец матрицы (6.19) должен быть распространен в обратном на-
правлении, используя (5.12), чтобы сформировать одну строку якобиана. 

Столбцы также могут обратно распространяться совместно 

                              (6.22) 
Общая матрица чувствительностей Марквардта для каждого слоя создается пу-

тем дополнения матрицами, вычисляемыми для каждого входного вектора p: 

                                           (6.23) 

Обратите внимание, что для каждого входного вектора должно выпол-
няться обратное распространение чувствительностей. Это связано с тем, что 

вычисляются производные для каждой отдельной ошибки, а не производные от 
суммы квадратов ошибок. Для каждого входного вектора, поданного на вход 

сети, вычисляется вектор ошибки и для каждого элемента вектора ошибки вы-
числяется одна строка матрицы Якоби. После того, как чувствительности будут 

распространены обратно, якобиан рассчитывается по формулам (6.18-6.19). 
 

2.2.3. LM алгоритм обучения MLP 
 

Объединяя формулы, приведенные выше, получим LM-алгоритм обучения 

MLP (рисунок 6.1).  
 

LM алгоритм 
 

1. Подать на вход MLP очередной входной вектор и вычислить реакцию 
MLP (5.1) и ошибку 

. 

Вычислить сумму квадратов ошибок для всех входных векторов, исполь-

зуя (6.15). 
2. Вычислить якобиан (6.8). Для этого вычислить чувствительности в 

соответствии с рекуррентным соотношением (6.22), выполнив инициали-
зацию (6.21). Составить общую матрицу чувствительностей Марквардта 

(6.23). Вычислить элементы матрицы Якоби (6.18-6.19). 
3. Вычислить новые значения параметров с помощью (6.14)  

4. Заново вычислить сумму квадратов ошибок при новых параметрах 
MLP. Если сумма квадратов меньше, чем значение, вычисленное на шаге 
1, то поделить  μk на  θ. Запомнить обновленные значения параметров и 

перейти к  шагу 1. Если сумма квадратов не уменьшилась, то умножить μk 
на θ и перейти к шагу 3. 

 

Рисунок 6.1 – LM алгоритм  

 
LM алгоритм сходится быстрее, чем любой ранее рассмотренный алго-

ритм. Конечно, он требует большего числа вычислений на каждой итерации, 

т.к. включает инверсию матрицы. Тем не менее, LMА является самым быстрым 
при среднем числе параметров сети.  

Основной недостаток LM алгоритма – это большой объем требуемой па-
мяти (матрица Гессе имеет размер nхn, где n – число параметров). Когда число 
параметров очень большое, то использовать LMA непрактично (верхний предел 

использования LMA – несколько тысяч параметров). 
В модуле NeuralNetworks 2.0 пакета Scilab имеется функция ann_FFBP_lm, 

реализующая LM алгоритм. Текст функции с комментариями приведен в при-
ложении A. В общем случае формат вызова функции требует задания следую-

щих параметров:  
 

W = ann_FFBP_lm(P, T, N, af, mu, mumax, theta, itermax, mse_min, gd_min) 
 

где mu -  значение параметра μk LM алгоритма, по умолчанию mu=0.001; mumax 

– максимально возможное значение μk, по умолчанию mumax=100000000; theta – 
значение параметра θ LM алгоритма, по умолчанию theta=10. Остальные пара-

метры функции аналогичны параметрам функции ann_FFBP_gd, которые были 
рассмотрены в лабораторной работе №5.  

 
3. Варианты заданий и программа работы 

 

3.1 Повторить теоретический материал, относящийся к алгоритмам обучения 
многослойного персептрона, основанных на методах оптимизации [4, 5]. 

3.2 Выполнить сравнительное исследование MLP, обученного на основе алго-
ритмов BP_gd и BP_lm, при решении задачи классификации. Для этого по 

аналогии с заданием п.3.2 лабораторной работы №5 выполнить следующее: 
3.2.1  Сформировать два множества (класса) точек данных (не менее 200 точек 

в каждом множестве), которые располагаются выше и ниже кривой у=f(x), 

и отстоят от неё на расстояние d=                   ; 
3.2.2 Отобразить классы и кривую у=f(x) на двумерной плоскости; 
3.2.3 Разработать программу обучения многослойного персептрона с архитек-

турой R-S-1 для классификации этих классов, в качестве активационных 
функций скрытого слоя использовать функции ann_tansig_active или 

ann_logsig_active, а в качестве активационной функции выходного слоя ─ 
ann_purelin_active, обучение персептрона выполнять с использованием 

функций ann_FFBP_gd и  ann_FFBP_lm; 
3.2.4 Используя значение скорости обучения lr=0.01, выполнить сравнительное 

обучение MLP с архитектурой [R-S-1] двумя алгоритмами BP_gd и BP_lm 
при разных S (10,20,30,40) и при itermax =100. Построить и сравнить кри-
вые обучения для 2-х алгоритмов при разных S. 

3.2.5 Для различных MLP (при разных S), обученных в соответствии с LM ал-
горитмом, выполнить моделирование этих MLP и проверить качество 

классификации на тестовом множестве данных. Для этого сформировать 
тестовое множество данных аналогично п. 3.2.1, провести классифика-

цию данных с помощью обученного MLP, отобразить точки полученных 
выходных классов и кривую у=f(x) на двумерной плоскости, вычислить 

вероятности правильной классификации при разных S. Сравнить вероят-
ности правильной классификации для MLP, обученного на основе LM ал-

горитма с аналогичными вероятностями, полученными в предыдущей ла-
бораторной работе, обратив внимание на значение itermax. 

3.3 Выполнить сравнительное исследование MLP, обученного на основе алго-

ритмов BP_gdx  и BP_lm, при решении задачи аппроксимации функции 2-х 
переменных z=f(x,y). Для этого по аналогии с заданием п.3.3 лабораторной 

работы №5 выполнить следующее:  
3.3.1 Сформировать подмножества обучающих и тестовых  данных. Для этого 

выбрать на плоскости (x,y) 300 случайных точек и определить в этих точ-
ках значение функции z=f(x,y). В качестве входного вектора использовать 

вектор p=[x;y], в качестве целевого значения – t=z. Полученные данные 
разделить на 2 подмножества: обучающее (80%) и тестовое (20%). 

3.3.2 Разработать программу обучения многослойного персептрона с архитек-
турой  R-S-1для аппроксимации функции z=f(x,y), в качестве активацион-

ных функций скрытого слоя использовать функции ann_tansig_active или 
ann_logsig_active, а в качестве активационной функции выходного слоя -- 

ann_purelin_active, обучение персептрона выполнять с использованием 
функции ann_FFBP_gdx и ann_FFBP_lm; 

3.3.3 Выполнить обучение MLP (при квазиоптимальном S, itermax=100) c по-

мощью функций ann_FFBP_gdх и ann_FFBP_lm, построить и сравнить 
кривые обучения; 

3.3.4 Используя тестовое подмножество данных, выполнить моделирование 2-
х вариантов MLP, обученных с помощью 2-х разных функций, указанных 

выше. Построить сопоставительные графики значений функции z=f(x,y) и 
соответствующих значений на выходе MLP, вычислить СКО аппрокси-

мации функции на тестовом подмножестве, сравнить со значениями 
СКО, полученными на обучающем подмножестве.  

3.4 Подготовьте и защитите отчет по работе. 
 

4 Методические рекомендации по выполнению работы 
 

4.1 В задании п.3.2. исследуется задача бинарной классификации. Цель зада-

ния  – убедиться в том, что MLP, обучаемый на основе LM алгоритма, обучает-
ся быстрее, чем BP_gd и строит более точную нелинейную границу решения 

между классами.  
Это задание практически повторяет задание, сформулированное п. 3.2 ла-

бораторной работы №5. Поэтому все, что Вам необходимо сделать – это адап-
тировать соответствующую часть программного кода лабораторной работы №5.  

 Поскольку в п.3.2.4. требуется сравнить кривые обучения для функций 
ann_BP_gd и ann_BP_lm, то необходимо модифицировать эти функции обуче-
ния, таким образом, чтобы они возвращали, кроме весов, значение СКО на каж-

дой эпохе. Схема такой модернизации была указана в предыдущей лаборатор-
ной работе. Кроме этого, необходимо запоминать веса обученных MLP при 

разных S, так как они понадобятся при выполнении моделирования и тестиро-



92 
 

вания MLP (п.3.2.5). Для этого удобно использовать клеточные массивы. При-
мер соответствующего кода, выполняющего обучение на основе модифициро-

ванной функции ann_FFBP_gd1 и запоминающего СКО и веса при разных S, 
приведен ниже: 
… 

     //Обучаем при разных S  
    S=[10 20 30 40]; 

… 
    MSE_gd=[]; 
    W_all_gd=cell(); // клеточный массив для хранения всех весов 

    for k=1:length(S) 
 N=[2 S(k) 1]; 

  [W_gd,out_mse_gd] = ann_FFBP_gd1(P,T,N,af,lr,itermax); 
 MSE_gd=[MSE_gd;out_mse_gd']; 
 W_all_gd{k}= W_gd; 

   end; 

 

В качестве примера на рисунке 6.2 изображены кривые обучения MLP для 2-х 
исследуемых алгоримов обучения BP_gd и ВP_lm, построенные с помощью по-

добного кода при itermax=100. 

 
Рисунок 6.2 – Кривые обучения MLP для алгоритмов BP_gd и BP_lm 

 
 Из рисунка видно, что LM алгоритм сходится значительно быстрее и 

обеспечивает уровень СКО обучения почти в 6 раз ниже, чем алгоритм BP_gd. 
 Для тестирования MLP (задание п.3.2.5), обученного на основе LM ал-

горитма, используйте запомненные значения весов и функцию ann_FFBP_run. 
Тестирование выполняйте на тестовом подмножестве данных, на котором MLP 

не обучался. Используйте код тестирования, который Вы разработали в преды-
дущей лаборатоной работе. 

 Пример выходных классов, формируемых обученным MLP на основе 
LM алгоритма, приведен на рисунке 6.3. Вероятности правильной классифика-

ции при S=[10,20,30,40] для примера были соответственно равны 0.995, 0.9975, 

1, 1. Т.е. при S=30 и 40 MLP не допускает ошибок классификации. Это намного 
лучше по сравнению с  результатами обучения на основе алгоритма BP_gd, по-

лученными в предыдущей лабораторной работе.  
 

 
Рисунок 6.3 – Результаты классификации тестирующего множества,  

вероятность правильной классификации 1,0 (S=30, LM алгоритм) 

 
4.1 В задании 3.3 исследуется задача аппроксимации двумерной функции с 

помощью MLP. Цель задания – убедиться в том, что MLP, обучаемый на основе 
LM алгоритма, превосходит по скорости обучения и точности получаемой ап-
проксимации лучший из ранее рассмотреных алгоритмов обучения BP_gdx.  

Это задание практически повторяет задание, сформулированное п. 3.3 ла-
бораторной работы №5. Поэтому все, что Вам необходимо сделать – это адап-

тировать соответствующую часть программного кода лабораторной работы №5. 
При выполнении задания в соответствии с п.3.3.3 отбразите кривые обу-

чения для 2-х исследуемых алгоритмов BP_gdx и BP_lm на одном графике (ри-
сунок 6.4). 

 
Рисунок 6.4 – Кривые обучения MLP (алгоритмы BP_gdx и BP_lm) 

Как следует из рисунка 6.4, алгоритм BP_lm завершил обучение за 7 эпох. 
При этом обучение завершилось в результате достижения минимального значе-

ния СКО mse_min=1e-5. 
Для тестирования набора обученных нейросетей используйте функцию 

ann_FFBP_run (задание п. 3.3.4). По результатам тестирования необходимо по-

строить сопоставительные графики функции z=f(x,y), определенной на тестовом 
множестве входных данных, и графики выходных сигналов MLP, обученных 

разными алгоритмами BP_gdx и BP_lm . В качестве примера на рисунок 6.5 
изображены графики тестовых значений функции z=f(x,y) и значений на выходе 

MLP, обученных функциями ann_BP_gdx и ann_BP_lm. 
 

 
Рисунок 6.5 –  Тестовые значения функции z=f(x,y) и значения на выходе 

MLP, обученного с помощью функций ann_BP_gdx и ann_BP_lm 

 
Из рисунка 6.5 следует, что алгоритм BP_gdх характериризуется 

бȯльшими значениями CКО, чем BP_lm при обучении на 100 эпохах.  В частно-
сти, для рассматриваемого примера указанные ошибки на тестовом подмноже-

стве данных равны: для BP_gdх –  0.01378, для BP_lm –  0.00008.  

 
5. Содержание отчета 

 

5.1.  Цель работы. 

5.2.  Вариант задания.  
5.3.  Схема MLP, решающего задачу классификации 2-х классов; программа 

обучения MLP бинарной классификации с использованием функций ann_BP_gd 
и ann_BP_lm; результаты обучения MLP при разных S; результаты тестирова-
нии обученного бинарного классификатора на основе MLP и вероятности пра-

вильной классификации при разных S.  
5.4.  Схема MLP для решения задачи аппроксимации функций; программа обу-

чения MLP; кривые обучения MLP для 2-х функций ann_BP_gdx и ann_BP_lm; 
результаты тестирования MLP, обученных указанными функциями (сравни-

тельные графики), значения СКО на этапе обучения и на этапе тестирования.  

5.5.  Выводы по результатам исследований. 
 

6. Контрольные вопросы 
 

6.1. Сформулируйте решение задачи оптимизации целевой функции на основе 

метода Ньютона.  
6.2. Сформулируйте достоинства и недостатки метода Ньютона. 

6.3. Сформулируйте условия, при которых метод Ньютона совпадает с SDA. 
6.4. Запишите выражение для целевой функции алгоритма Левенберга-

Марквардта.   
6.5. Запишите выражение для градиента целевой функции алгоритма Левен-

берга-Марквардта.   
6.6. Запишите общее выражение для якобиана векторной функции v(x) в ал-

горитме Левенберга-Марквардта.   
6.7. Запишите выражение, аппроксимирующее матрицу Гессе в алгоритме 

Левенберга-Марквардта. 
6.8. Запишите выражение основного алгоритма Левенберга-Марквардта. 

6.9. При каком условии алгоритм Левенберга-Марквардта трансформируется 
в SDA. 
6.10. При каком условии алгоритм Левенберга-Марквардта трансформируется 

в алгоритм Гаусса-Ньютона. 
6.11. Определите чувствительность Марквардта. 

6.12. Как вычислить элементы якобиана в алгоритме Левенберга-Марквардта, 
используя чувствительсность? 

6.13. Как вычислить чувствительсность выходного слоя в алгоритме Левенбер-
га-Марквардта? 

6.14. Запишите рекуррентное соотношение для вычисления чувствительностей 
в алгоритме Левенберга-Марквардта. 

6.15.  Сформулируйте LM алгоритм. 
6.16. Объясните назначение параметров функции ann_FFBP_lm. 

6.17. При каких условиях функция ann_FFBP_lm прекращает процесс обуче-
ния? 
6.18. Объясните код функции ann_FFBP_lm. 

 
 

Приложение А. Функция ann_FFBP_lm 
 

function W=ann_FFBP_lm(P, T, N, af, mu, mumax, theta, itermax, mse_min, gd_min) 

///Обучение нейросети прямого распространения на основе алгоритма Левенберга-Марквардта. 
// Примеры вызова:  

//     W = ann_FFBP_lm(P,T,N)  

//     W = ann_FFBP_lm(P,T,N,af,mu,mumax,theta,itermax,mse_min,gd_min) 

// Параметры 

//     P : Обучающий вход  
//     T : Обучающие целевые значения 

//     N : Вектор, задающий число нейронов каждого слоя, включая входной и выходной слои  

//     af : Список имен активационных функций от 1-го до выходного слоев 

//    mu : значение “мю” ЛМ алгоритма 

//     mumax : максимально возможное “мю”  

//     theta : множитель “тета” для “мю” 
//     itermax : Максимальное число эпох обучения  

//     mse_min : Минимальная ошибка  (Значение целевой функции)  

//     gd_min : Минимальное значение градиента 

//     W : Выходные веса и смещения 

//     
//1. ================Обработка списка аргументов функции========================  

    rhs=argn(2); 

 

/ Проверка ошибки списка аргументов 

    if rhs < 3; error("Expect at least 3 arguments, P, T and N");end 
// Выбор значений аргументов по умолчанию 

    if rhs < 4; af = ['ann_tansig_activ','ann_purelin_activ']; end 

    if rhs < 5; mu = 0.001; end 

    if rhs < 6; mumax = 100000000; end 

    if rhs < 7; theta = 10; end         
    if rhs < 8; itermax = 1000; end 

    if rhs < 9; mse_min = 1e-5; end 

    if rhs < 10; gd_min =  1e-5; end 

 

    if af == []; af = ['ann_tansig_activ','ann_purelin_activ']; end 
    if mu == []; mu = 0.001; end 

    if mumax == []; mumax = 100000000; end 

    if theta == []; theta = 10; end     

    if itermax == []; itermax = 1000; end 

    if mse_min == []; mse_min = 1e-5; end 
    if gd_min == []; gd_min =  1e-5; end 

 

//2.=======================Инициализация====================  

// Инициализация сети и формирование списка имен производных активационных функций 

    format(8);warning('off'); 
    W = ann_ffbp_init(N,[-1 1]); // Каждый элемент W – матрица весов одного из слоев 

    itercnt = 0; 

    af_d = strsubst(af,'ann_','ann_d_'); 

    mse = %inf; 

    gd = %inf; 
    A = ann_ffbp_init(N,[0 0]); 

    tempW = A; 

    train_N = size(P,2); // Размер обучающего множества 

 

    // Инициалиазация GUI окна прогресса 
    handles = ann_training_process(); 

    handles.itermax.string = string(itermax); 

    handles.msemin.string = string(mse_min); 

    handles.gdmax.string = 'inf'; 

    handles.gdmin.string = string(gd_min); 
 

// Задание числа слоев и списков промежуточных переменных  

layers = size(N,2)-1; // слои считаются от 1-го скрытого слоя до выходного слояr 

n = list(0); 

a = list(0); 
m = list(0); 

s = list(0); 

 

// 3. ===================Реализация цикла эпох обучения==========================  

    while mse > mse_min & itercnt < itermax & mu <= mumax & gd > gd_min 
 

        mucnt = 0; 

        // Прямое распространение – моделирование (аналогично функции ann_FFBP_run (формула 5.1)) 

        n(1) = W(1)(:,1:$-1)*P + repmat(W(1)(:,$),1,size(P,2)); // Вычисляем и запоминаем сетевые выходы 1го слоя 

        a(1) = evstr(af(1)+'(n('+string(1)+'))'); // Вычисляем и запоминаем активности на выходе 1го слоя 
        for cnt = 2:layers 

            n(cnt) = W(cnt)(:,1:$-1)*a(cnt-1) + repmat(W(cnt)(:,$),1,size(P,2)); // Вычисляем и запоминаем сетевые выходы  

            a(cnt) = evstr(af(cnt)+'(n('+string(cnt)+'))'); // Вычисляем и запоминаем активности на выходе каждого слоя  

        end 

        // Вычисление ошибки  
        e = T - a($); 

        [r,c] = size(a(layers)); //запоминаем размерность выходного слоя 

         

        m(layers) = evstr(af_d(layers)+'(a('+string(layers)+'))'); Вычисление производных актив. функций выходного слоя (6.20) 

        // s(layers) = -(m(layers).*.eye(N($),N($)));  
        s(layers) = -(m(layers).*.ones(1,r)).*(ones(1,c).*.eye(r,r));  // Чувствительности выходного слоя –формула (6.21) 

     

    for cnt = layers-1:-1:1 //Цикл для вычисления производных слоев и обратного  распространения чувствительностей  

        Wpre = W(cnt+1)(:,1:$-1);     //Извлекаем из W и запоминаем веса следующего слоя    

        a(cnt) = a(cnt).*.ones(1,N($));//Формируем матрицу активности слоя 
        m(cnt) = evstr(af_d(cnt)+'(a('+string(cnt)+'))'); //Матрица производных F’(n) слоя  

        s(cnt) = m(cnt).*(Wpre'*s(cnt+1)); // Матрица чувствительности слоя – формула (6.22)  

 

    end 

        
        // Вычисление элементов Якобиана – формула (6.18) и формирование матрицы Якоби 

        Jj = []; 

        jac = ann_calcjac(kron(P,ones(1,N($))),s(1)); // Для первого слоя 

        Jj = [Jj jac s(1)']; 

        for cnt = 2:layers 
            jac = ann_calcjac(a(cnt-1),s(cnt)); // Для последующих слоев 

            Jj=[Jj jac s(cnt)']; // добавление элементов в матрицу Якоби  

        end 

         

        mse = (mean(e.^2)) 
        mse2 = %inf; 

        J = Jj; 

        J2 = (J' * J);         // Формула (6.13)  

        Je = J'*e(:);  // часть формулы (6.14)  

        //Обновление параметров  
        while  mse2 >= mse & mu <= mumax // пока новый СКО не станет меньше текущего СKO 

            dx = -(J2 + (eye(J2)*mu)) \ (Je); // Вычисление обновления (формула 6.14) 

            szpre = 0; 

            for cnt = 1:layers       //Цикл для послойного обновления параметров 

                sz = N(cnt)*N(cnt+1) + N(cnt+1); // Число параметров текущего слоя 
                dx_part = dx(szpre+1:szpre+sz)  //Выделение части обновлений для текущего слоя 

                A(cnt) = [matrix(dx_part(1:$-N(cnt+1)),N(cnt+1),N(cnt)) dx_part($-N(cnt+1)+1:$)]; // Реформатирование 

                tempW(cnt) = W(cnt) + A(cnt); // обновление параметров слоя  

 

                szpre = szpre + sz; // Начало индексов следующего слоя 
            end 

            //Вычисление выхода при новых значениях параметров  

            y = ann_FFBP_run(P,tempW,af); 

            e2 = T - y; // Вычисление новой ошибки 

            mse2 = (mean(e2.^2)); //Вычисление нового СКО 
            if  mse2 >= mse //Если СКО увеличился, то увеличить mu 

                mu = mu*theta; 

            end 

 

        end 
 

        mu = mu/theta;  // Т.к. новый СКО меньше текущего, то уменьшить mu  

        if (mu < 1e-20)    

             mu = 1e-20; 

                    // break 
        end 

        W = tempW; //Сохранить обновленные значения параметров 

 

        //Критерий остановки 

        mse = mean(e.^2); 
        itercnt = itercnt + 1; 

        //средний градиент  

        gd = 2*sqrt(Je'*Je)/train_N; 

 

       // Отображение  GUI  прогресса обучения 
       if  itercnt == 1 then 

           mse_max = mse; 

           handles.msemax.string = string(mse_max); 

           gd_max = gd; 

           handles.gdmax.string = string(gd_max);            
           mse_span = log(mse) - log(mse_min); 

           iter_span = itermax; 

           gd_span =  log(gd) - log(gd_min); 

       end 

        // Scilab 5.5  и выше 

        handles.iter.value = round((itercnt/iter_span)*100); 

        handles.mse.value = -(log(mse)-log(mse_max))/mse_span * 100;// round(((log(mse) - log(mse_min))/mse_span)*100); 

        handles.gd.value = -(log(gd)-log(gd_max))/gd_span * 100; //round(((log(gd) - log(gd_min))/gd_span)*100);  

 
        handles.itercurrent.string = string(itercnt); 

        handles.msecurrent.string = string(mse);      

        handles.gdcurrent.string = string(gd);       

 

    end 
endfunction 

Лабораторная работа №7 
 

Исследование состязательных сетей и сетей 
векторного квантования 

1 Цель работы 

Углубление теоретических знаний в области обучения нейросетей без 
учителя,  исследование свойств алгоритмов обучения состязательных сетей на 

основе правил Кохонена, приобретение практических навыков обучения само-
организующихся карт Кохонена и сетей векторного квантования при решении 

задач классификации.  

2 Основные теоретические положения 

2.1 Обучение без учителя и самоорганизующиеся ИНС 
 

При обучении без учителя отсутствует информация о правильности 
реакции ИНС на тот или иной входной вектор. Нейронная сеть самостоятельно 

обнаруживает взаимосвязь входных векторов и преобразует их в ассоциирован-
ную с ними выходную реакцию, т.е. в ИНС происходят процессы самооргани-

зации. 
 Самоорганизующиеся ИНС способны открывать отношения подобия во 

входных данных, обеспечивая тем самым выделение во входном пространст-

ве классов.  
Существует два основных типа самоорганизующихся ИНС:  

1) ИНС, основанные на правиле обучения Хебба; 
2) состязательные ИНС. 

 
2.2 Правило обучения Хебба 

 

Правило обучения Хебба заимствовано из биологических нейронных се-
тей. В соответствии с этим правилом веса синаптических связей изменяются 

пропорционально активности нейронов. Правило обучения Хебба можно за-
писать в виде 

),()()1()( qjpqiaq
ij

wq
ij

w 
                           (7.1) 

где  ai(q) – выход  i- го нейрона; pj(q) – входное воздействие на  j-ом входе  ней-
рона. Изменение веса связи в соответствии с правилом не требует привлечения 

сведений о желаемой реакции нейрона.  
Правило Хебба может быть записано в векторной форме 

).()()1()( qqqq  paWW                            (7.2) 
 

Как в любом правиле обучения без учителя, обучение выполняется по реакциям 

на последовательность предъявляемых входных образов p(1), p(2),…,p(Q). 

Приведенная форма правила Хебба не ограничивает рост весов связей 
по мере обучения. Поэтому в правило Хебба вводят затухание 

 

    ),()()1()1()1()()()1()( qqqqqqqq  paWWpaWW     (7.3) 
 

где γ – скорость затухания. В этом случае максимальный вес связи ограничива-

ется значением wij =α/ γ (получается, если все a и p равны 1). 
 

2.3 Простая распознающая сеть Instar 
 

Рассмотрим простую сеть из одного нейрона (рисунок 7.1). 
 

 
Рисунок 7.1 – Сеть Instar 

 

Нейрон будет активен, когда   Скалярное произведение будет 
максимальным при                 Т.е. сеть активна, когда p близко к 1w. Подбирая 

b, мы можем регулировать степень схожести p и 1w.  
Если  установить              , то нейрон  будет  активен, когда p точно 

соответствует направлению 1w. В этом случае нейрон будет распознавать толь-
ко образ, соответсвующий 1w.  

 Если мы хотим, чтобы сеть распознавала образы, близкие к 1w, надо 
уменьшить b. Чем меньше b, тем большое число входных образов будут акти-

вировать сеть.  
 

2.4 Состязательные сети 
 

Рассмотрим слой нейронов, в котором на выходе нейрона с наибольшим 

значением сетевой функции устанавливается единичная активность. Такая сеть 
называется состязательной  или соревновательной. Нейрон, имеющий наи-

большее значение сетевой функции, называют нейроном-победителем, выиг-
равшим “состязание”.   

Состязательную  функцию преобразования обозначают как a= compet(n) 
(рисунок 7.2). Она обеспечивает нахождение индекса i* нейрона-победителя и 

устанавливает на его выходе 1:  

Рисунок 7.1 – Состязательная сеть 

 

В состязательной сети  устанавливается 1 на выходе того, нейрона, чей 
вектор весов по направлению “ближе” всего к входному вектору p.  

 
2.5 Правило обучения Кохонена 

 

Отдельные нейроны состязательной сети на уровне сетевой функции со-
ответствуют Instar-нейронам. Для решения задачи классификации с помощью 

состязательной сети можно строкам матрицы весов связей W присвоить значе-
ния желаемых векторов-прототипов классов. Т.к. векторы-прототипы нам зара-

нее неизвестны, то для обучения сети можно использовать ассоциирующее In-
star правило (7.3). 

Поскольку для состязательной сети ai(q)≠0 только для нейрона-
победителя i*, то 

 

.*,)()1()1())1()(()1()( iiqqiqiqqiqi  pwwpww 
  (7.4) 

.*,)1()( iiqiqi  ww
                                                                (7.5) 

 

Правило (7.4-7.5) называют правилом Кохонена. В 

соответствии с этим правилом на каждой итерации вектор-
строка iw(q-1) матрицы весов, ближайшая к входному век-
тору p(q) (или имеющая наибольшее значение скалярного 

произведения с p(q)), смещается  по направлению к вход-
ному вектору (рисунок 7.2), формируя новое значение век-

тора весов iw(q). 
 

 
2.6 Самоорганизующиеся карты признаков (SOM) 

 

SOM  (self-organizing feature maps ), предложенные Кохоненым, сначала 
определяют нейрон-победитель i* (как в состязательной сети), а затем веса всех 

нейронов в окрестности нейрона-победителя  Ni*  обновляются на основе пра-
вила Кохонена 
 

.
*

,)()1()1())1()(()1()(
i

Niqqiqiqqiqi  pwwpww 
    (7.6) 

 

Окрестность Ni*  содержит индексы всех нейронов, которые находятся на рас-

стоянии радиуса d  от нейрона–победителя i*: Ni (d)={j, dij  ≤   d} . Когда предъ-
является входной вектор p(q)  веса нейрона-победителя и нейронов в его окре-

стности смещаются в сторону вектора p(q). После многократных предъявлений  
соседние нейроны будут обучены распознаванию похожих входных векторов. 

Карта признаков отображает образы из входного n-мерного пространства 

в пространство меньшей размерности. Кроме сокращения размерности про-
странства, карта признаков обеспечивает сохранение отношений топологиче-

ского соседства входных образов. При этом геометрическая близость выходных 
классов на карте признаков означает близость соответствующих образов во 

входном пространстве признаков.  
Примеры круговых окрестностей  c радиусами d=1 и  d=2  для двумерной 

карты изображены на рисунке 7.4. 
 

 
Рисунок 7.4 – Примеры круговых окрестностей 

 

Нейроны могут быть также организованы  в 1-мерные и 3-х мерные кар-

ты. Кохоненым были предложены прямоугольные и гексагональные окрест-
ности с целью повышения эффективности реализации.     
  

 
Рисунок 7.5 – Прямоугольная и гексогональные окрестности 

 
На практике множество  Ni*(d) и коэффициент обучения изменяют свои значе-

ния динамически в процессе обучения. Процесс обучения начинается при рас-

ширенном множестве  Ni*(d)  и больших значениях α . По мере обучения коли-
чество НЭ, включаемых в Ni*(d) уменьшается, одновременно уменьшается и 

значение α . При обучении учёт нейронов, образующих окрестность Ni*(d), 
осуществляется с помощью функции соседства Λ(i,i*), которая равна 1 при i=i* 
и уменьшается по мере увеличения расстояния от нейрона  i* до нейрона  i  в 

выходном топологическом пространстве. В этом случае правило  обучения за-
писывается в виде: 

))1()()(,()()( *  qqiiqq ii wpw  .                     (7.7) 
Например, функция  Λ(i,i*)  может быть задана выражением 
 

)2/exp(),( 22
* drii ii 

,                                (7.8) 
где  rii* – расстояние  между нейронами  i  и i*   в выходном двумерном про-
странстве;  d – параметр, задающий “диаметр” функции соседства. Диаметр d  и 

скорость обучения α  уменьшаются в зависимости от q , например по экспонен-
те. 

 
2.7 Обучение на основе векторного квантования (LVQ) 

 

LVQ (learning vector quantization) сети являются гибридными (рисунок 
7.6).  

 

 
Рисунок 7.6 – Сеть векторного квантования 

 
Как и в состязательной сети, 1-ый слой определяет подклассы  входного 

пространства. При этом вместо скалярного произведения вычисляется прямое 
расстояние между векторами. Это позволяет не выполнять нормализацию 
входных векторов.  

Второй слой комбинирует подклассы в  один  класс. Для второго слоя 
столбцы матрицы  W

2 
 представляют подклассы, а строки – классы. W

2 
 содер-

жит в каждом столбце по одной 1. Единицы в строках отображают класс под-
класса: 

.                (7.9) 

Процесс комбинации подклассов в классы позволяет сетям LVQ  создавать 
сложные границы между классами. 

 
2.8 Правило обучения сетей векторного квантования 

 

Правило комбинирует обучение без учителя и с учителем. Оно ис-
пользует множество обучающих примеров {p1,t1},  {p2,t2},…,{pQ,tQ}. Каждый 

целевой вектор содержит только одну единицу, которая обозначает класс при-
надлежности входного вектора.  Перед началом обучения формируется матрица 

W
2   

и нейронам 1-го слоя назначается класс принадлежности (выходной ней-
рон). Обычно равное количество нейронов 1-го слоя связывается с каждым вы-

ходным нейроном. 
Затем веса 1-го слоя обучаются на основе модифицированного правила 

Кохонена. Если p классифицируется корректно (i*классу k), то  
 

          . (7.10) 
 

Если p классифицируется  не корректно (i* классу k), то 
 

.    (7.11) 
 

Таким образом, нейроны скрытого слоя смещаются в строну векторов, 

которые попадают в класс, для которого они формируют подкласс, и удаляются 
от векторов, которые относятся к другому классу.  

Правило LVQ чувствительно к инициализации весов 1-го первого 
слоя. Из-за этого некоторые нейроны не могут «попасть» в свои подклассы. 

Имеется усовершенствованное правило LVQ2. Если возникает ситуация не-
верной классификации, то дополнительно  подстраиваются со знаком «+» веса 

ближайшего нейрона к входному вектору, который даёт верную классифика-
цию. 

 
2.7 Функции Scilab  Neural Network 2.0 для обучения состязательных сетей 

и сетей векторного квантования 
 

 В модуле Scilab Neural Network 2.0 реализованы следующие функции для 

обучения без учителя: 
 ann_COMPET — функция обучения состязательной сети; 

 ann_COMPET_run — функция моделироавания  состязательной сети; 
 ann_COMPET_visualize2d — состязательная сеть с 2D анимацией;  
 ann_COMPET_visualize3d — состязательная сеть с 3D анимацией; 

 ann_SOM — ИНС в виде самоорганизующейся карты  (блочное обуче-
ние);  

 ann_SOM_online — ИНС в виде самоорганизующейся карты  (последова-
тельное обучение; 

 ann_SOM_run — функция моделирования ИНС в виде самоорганизую-
щейся карты; 

 ann_SOM_visualize2d — ИНС в виде самоорганизующейся карты с 2D ви-
зуализацией;  

 ann_SOM_visualize3d — ИНС в виде самоорганизующейся карты с 3D ви-
зуализацией. 

Для обучения и моделирования сетей векторного квантования в модуле 

Scilab Neural Network 2.0 имеются следующие функции: 
 ann_LVQ1 — ИНС векторного квантования ;  

 ann_LVQ_run — функция моделирования ИНС векторного квантования. 
Рассмотрим функцию для обучения состязательной сети ann_COMPET (код 

функции с комментариями приведен в приложении A). Синтаксис вызова 
функции: 

 

[W,b] =  ann_COMPET(P,N) 
[W,b] =  ann_ COMPET(P,N,lr,lr_c,itermax), 

 

где P  – матрица обучающих примеров; N – число нейронов в состязательном 

слое (число классов); lr – скорость обучения (по умолчанию 0.1); lr_c – скорость 
обучения для смещения (по умолчанию 0.1); itermax – максимальное число эпох 
обучения (по умолчанию 100); W – выходная матрица весов связей; b – вектор 

смещений.  
Ниже приведен пример применения функции:  
 

x = rand(2,10);  
x(:,1:5) = x(:,1:5) + 1; 

plot(x(1,:),x(2,:),'.'); 
[W,b] =  ann_COMPET(x,4);close; 

plot(W(:,1), W(:,2), 'or'); 
 

В примере формируется двумерный случайный массив чисел x  в диапазо-

не от 0 до 1 с равномерным распределением. Затем первая часть массива со 
столбцами с 1 по 5 увеличивается на 1. Массив x представляет собой матрицу 

обучающих примеров, которые на рисунке 7.7 обозначены голобыми точками. 
Вызов ann_COMPET(x,4) обеспечивает обучение состязательной сети  для разде-

ления обучающих примеров на 4 класса. Результаты обучения (значения весов 
нейронов) представлены красными кружочками на рисунке 7.7. 

 

Рисунок 7.7 – Пример обучения состязательной сети 

Функции ann_COMPET_visualize2d  и ann_COMPET_visualize3d вызываются 
аналогично. Эти функции позволяют визуализировать несложные примеры 

обучения либо на плоскости, либо в 3-мерном пространстве. 
Функция ann_SOM предназначена для блочного обучения состязательной 

сети, реализуемой в виде самоорганизующейся карты. Синтаксис вызова функ-

ции: 
 

W = ann_SOM(P) 
W = ann_SOM(P,N,itermax,steps,NS,topfcn,distfcn),  
 

где P – матрица обучающих примеров; N – структура карты признаков (по 
умолчанию - 8х8); itermax – максимальное число эпох обучения (по умолчанию 

- 200); steps – число шагов сокращения NS (по умолчанию - 100); NS – началь-
ный размер окрестности (по умолчанию - 3); topfcn – функция, определяющая 

топологию сети (по умолчанию ann_som_gridtop – прямоугольная сетка); distfcn 
– функция, определяющая способ вычисления расстояния соседства (по умол-

чанию ann_som_linkdist). 
Ниже приведен пример применения функции ann_SOM:  

 

x = rand(2,10); 
x(:,1:5) = x(:,1:5) + 1; 
W =  ann_SOM(x,[2 2]); 

[y,classes] = ann_SOM_run(W,x) 
 

В примере формируется массив обучающих данных и создается сеть в виде-
самоорганизующейся карты признаков Кохонена размером 2х2 с прямоуголь-
ной топологией. В результате обучения функция ann_SOM вернет матрицу ве-

сов W. Вызов функции моделирования ann_SOM_run обеспечивает проверку 
функционирования обученной сети. Т.к. сеть содержит 4 нейрона, то обучаю-

щие примеры будут отнесены к одному из 4-х классов, распознаваемых сетью. 
При этом значением переменной y для каждого примера из x являются бинар-

ные векторы с 1 в позиции нейрона-победителя, а значением переменной classes 
– номер класса каждого входного примера. 

Функция ann_SOM_online имеет сходный набор и вызывается аналогично    
ann_SOM. Отличие заключается в том, что функция осуществляет обучение в 

последовательном режиме, т.е. матрица весов корректриуется после предъявле-
ния каждого обучающего примера. Функции ann_SOM_visualize2d и 

ann_SOM_visualize3d осуществляют визуализацию процесса обучения в блоч-
ном режиме (вызываются с тем же набором параметров, что и ann_SOM). 

В состав модуля Scilab Neural Network 2.0 для работы с самоорганизующи-
мися картами входят следующие функции, определяющие топологию карт и 
способ вычисления расстояний соседства: 

 ann_som_boxdist — функция блочного расстояния;  
 ann_som_eudist — функция евклидова расстояния;  

 ann_som_gridtop — функция прямоугольной топологии;  
 ann_som_hextop — функция гексагональной топологии;   

 ann_som_linkdist — функция расстояния связи;  

 ann_som_mandist — функция манхэттенского расстояния;  
 ann_som_randtop — функция случайной топологии.  

Для визулизации топологии самоорганизующей сети можно использовать 
функции ann_som_plot2d и ann_som_plot3d. Примеры использования этих 
функций для визулизации топологии сетей в 2- и 3-х мерном пространствах: 

 

N = [5,5]; 

y = ann_som_hextop(N); 
ann_som_plot2d(y); 
 

N = [5,5,5]; 
y = ann_som_hextop(N); 

ann_som_plot3d(y); 
 

Результаты визуализации изображены на рисунке 7.8. 

  
Рисунок 7.8 – 2D и 3D гексагональные топологии 

 
Для обучения и моделирования сетей векторного квантования применяются 

соответственно функции ann_LVQ1 и ann_LVQ_run. Форматы вызова функций:  
 

[W,b] =  ann_LVQ1(P,T,N2); 
[W,b] =  ann_LVQ1(P,T,N2,lr,itermax); 

[y,classes] = ann_LVQ_run(W,P), 
 

где T – целевые классы, задаваемые в виде бинарных векторов-состояний вы-

ходов сети; N2 – количество нейронов состязательного слоя (количество под-
классов). Ниже приведен пример применения этих функций для разделения 

множества входных данных на 4 класса: 
 

x = rand(2,10); 
x(:,1:5) = x(:,1:5) + 1; 

T = [1 1 1 1 1 0 0 0 0 0; 0 0 0 0 0 1 1 1 1 1] 
[W,b] =  ann_LVQ1(x,T,4); 
[y,classes] = ann_LVQ_run(W,x). 

3. Варианты заданий и программа работы 
 

3.1 Повторить теоретический материал, относящийся к алгоритмам обучения 
состязательных нейронных сетей и сетей векторного квантования [1, 5]. 

3.2  Сформировать в соответствии с таблицей 7.1 множество входных дан-

ных для обучения SOM. Для генерации данных использовать генератор 
случайных чисел с равномерным распределением в диапазоне, указанном 

в таблице 7.1. 
 

Таблица 7.1 – Варианты заданий 

Вари-
ант 

Про-
странство 

Функции топо-
логии и расстоя-

ния  

Кол-во 
классов 

Диапазон 
значений 

вх.данных 

1 2D gridtop, eudist 16 0.0-4.0 

2 2D hextop, linkdist 16 0.0-4.0 

3 2D randtop, mandist 16 0.0-4.0 

4 2D gridtop, linkdist 36 0.0-6.0 

5 2D hextop, mandist 8 0.0-8.0 

6 2D randtop, eudist 8 0.0-8.0 

7 2D gridtop, mandist 25 0.0-5.0 

8 3D hextop, mandist 27 0.0-3.0 

9 3D randtop, linkdist 27 0.0-3.0 

10 3D gridtop, eudist 27 0.0-3.0 

11 3D hextop, eudist 8 0.0-2.0 

12 3D randtop, linkdist 8 0.0-2.0 

13 3D gridtop, mandist 8 0.0-2.0 

14 3D hextop, linkdist 64 0.0-4.0 

15 3D randtop, linkdist 64 0.0-4.0 

 
3.3 Написать программу, создающую и обучающую SOM, заданной тополо-

гии в соответствии с таблицей 7.1. на сгенерированных данных. 
3.4  Визуализировать топологию SOM на начальном этапе и после обучения. 

3.5  Сгенерировать множество векторов, представляющих центры классов, 
равномерно распределив их в заданных диапазонах входного пространст-

ва. Подать на вход сети эти векторы и определить номера нейронов-
победителей, распознающих эти векторы.  

3.6  Используя данные, сгенерированные в п. 3.2. и п.3.5 создать и обучить 

LVQ сеть, которая: 
- содержит в скрытом слое такое же число нейронов как SOM, заданная 

по варианту в соответствии с таблицей 7.1; 
- группирует каждые 2 (при четном числе подклассов) или 3 (при нечет-

ном числе подклассов) подкласса, формируемые скрытым слоем, в один 
класс. 

   3.7 Оценить корректность классификации векторов, сгенреированных в п.3.5 

   3.8. Подготовить и защитить отчет. 
 

4. Методические рекомендации по выполнению работы 
 

4.1  При выполнении работы используйте примеры, приведенные в подпунк-

те 2.8.  
 

5. Содержание отчета 
 

5.1      Цель работы. 
5.2 Вариант задания.  

5.3  Схемы SOM и LVQ сетей заданной архитектуры, топология SOM до 
обучения и после обучения, листинги программ с комментариями, таблица с 
векторами-прототипами и номерами нейронов-победителей для SOM, таблица с 

векторами- прототипами и номерами классов для LVQ сети.  
5.4 Выводы  
 

6. Контрольные вопросы 

 
6.1  Что понимается под термином «обучение без учителя»?  

6.2  Назовите основные типы самоорганизующихся ИНС. 
6.3 Сформулируйте и запишите правило обучения Хебба в его исходном ви-

де. 
6.4 Запишите правило обучения Хебба с затуханием. 
6.5 Нарисуйте схему сети Instar и запишите правило её обучения. 

6.6  Нарисуте схему состязательной сети и определите  функцию её преобра-
зования. 

6.7 Запишите правило обучения Кохонена для состязательной сети. Как гра-
фически интепретируется правило? 

6.8 Запишите правило обучения Кохонена для самоорганизующейся карты. 
6.9  Объясните понятие окрестности для нейрона-победителя. Приведите 

примеры разных видов окрестностей. 
6.10 Сформулируйте правило обучения SOM с использованием функции 

соседства. Определите функцию соседства. 
6.11 Нарисуйте схему сети векторного квантования. Объясните процесс 

вычислений в этой схеме. 
6.12 Как назначаются веса второго слоя сети векторного квантования? 

6.13 Сформулируйте модифицированное правило Кохонена для сетивек-
торного квантования. 

6.14 Какие  проблемы возникают при обучении состязательных сетей и 

как их решают на практике? 
6.15 В чем особенность усовершенстовованного правила обучения 

LVQ2? 
6.16 Объясните  и приведите пример вызова функции ann_COMPET.  

6.17 Объясните  и приведите пример вызова функции ann_SOM. 

6.18 Объясните  и приведите пример вызова функции ann_LVQ1. 
 

Приложение А. Функция обучения состязательной сети 
 
function [W, b]=ann_COMPET(P, N, lr, lr_c, itermax) 

// Функция обучения состязательной сети. 

// 

// Обработка входных аргументов функции 

rhs=argn(2); 
 

if rhs < 2; error("Expect at least 2 arguments, P and N");end 

if rhs < 3; lr = 0.1; end  // Скорость обучения по умолчанию 0.1  

if rhs < 4; lr_c = 0.1; end // Скорость обучения смещения по умолчанию-  0.1  

if rhs < 5; itermax = 100; end // Число эпох обучения по умолчанию - 100  
 

if lr == []; lr = 0.1; end 

if lr_c == []; lr_c = 0.1; end 

if itermax == []; itermax = 100; end 

 
[W, b] = ann_compet_init(P, N); // инициализация сети  

iter_span = itermax; 

 

// Инициализация GUI прогресса обучения 

handles = ann_training_process(); 
handles.itermax.string = string(itermax); 

handles.msecurrent.visible = 'off'; 

handles.msemin.visible = 'off'; 

handles.msemax.visible = 'off'; 

handles.mse.visible = 'off'; 
handles.msetitle.visible = 'off'; 

handles.gdcurrent.visible = 'off'; 

handles.gdmin.visible = 'off'; 

handles.gdmax.visible = 'off'; 

handles.gd.visible = 'off'; 
handles.gdtitle.visible = 'off'; 

 Q = size(P,2); // Определение размера обучающего множества примеров  

 

for itercnt = 1:itermax // цикл эпох обучения  

    for Pcnt = 1:size(P,2) // Цикл по всем  обучающим примерам из P 
        //n = W*P(:,Pcnt);  

        q = fix(rand(1, 'uniform')*Q)+1; // q – формируем случайный номер примера  

        p = P(:,q);        // Выбираем случайный пример из P 

        n = ann_negdist(W,p);  // вычисление отрицательного расстояния между примером и весами нейронов сети  

        a = ann_compet_activ(n + b); // вычисление состязательной функции  
        W(find(a),:) = W(find(a),:) +lr*(p'-W(find(a),:)); // Правило обучения Кохонена (формула 7.4)  

        b(find(a)) = b(find(a)) - 0.2; // Обновление смещений активного нейрона 

        b(find(~a)) = lr_c.*b(find(~a));  // Обновление смещений проигравших нейронов   

    end 

     
    handles.iter.value = round((itercnt/iter_span)*100); 

    handles.itercurrent.string = string(itercnt); 

     

end 

 
endfunction 

